

한 권으로 끝내는
시나공
컴활

길벗알앤디 지음

2급

필기+실기



이장 컴활 함수 사전

01 날짜 / 시간 함수

출제 001	날짜의 일련번호 구하기 - DATE	6
출제 002	날짜에서 일만 추출하기 - DAY	7
출제 003	회계상 근무일 수 계산하기 - DAYS	8
출제 004	날짜에서 연도만 추출하기 - YEAR	9
출제 005	날짜에서 요일 알아내기 - WEEKDAY	10
출제 006	현재 날짜 표시하기 - TODAY	11
출제 007	시간의 일련번호 계산하기 - TIME	12
출제 008	시간에서 시만 추출하기 - HOUR	13
출제 009	시간에서 분만 추출하기 - MINUTE	14
출제 010	날짜에서 월만 추출하기 - MONTH	15
출제 011	시간에서 초만 추출하기 - SECOND	16
012	현재 날짜와 시간 표시하기 - NOW	17
013	3개월 전/후 오늘의 일련번호 구하기 - EDATE	17
014	3개월 전/후 달의 마지막 날짜에 대한 일련번호 구하기 - EOMONTH	18
출제 015	주말을 제외한 휴가 마지막 날 계산하기 - WORKDAY	19

02 논리 함수

출제 016	1월 실적이 평균 이상이면 "우수", 평균 미만이면 "미달" 표시하기 - IF	20
017	과목번호가 1이면 "영어", 2이면 "수학", 3이면 "국어" 표시하기 - IFS	21
018	요일이 "토"나 "일"이면 "주말", "월"~"금"이면 "평일" 표시하기 - SWITCH	21
출제 019	모두 참(TRUE)일 때 참(TRUE) 반환하기 - AND	22
출제 020	하나라도 참(TRUE)이면 참(TRUE) 반환하기 - OR	23
021	논리식의 결과 부정하기 - NOT	24
출제 022	수식의 결과가 오류일 경우 "오류" 표시하기 - IFERROR	25

03 데이터베이스 함수

출제 023	과일의 평균 판매 금액 계산하기 - DAVERAGE	26
출제 024	가장 적게 팔린 과일의 금액 찾기 - DMIN	27
출제 025	가장 많이 팔린 과일의 금액 찾기 - DMAX	28
출제 026	과일의 판매 수량 합계 계산하기 - DSUM	29
출제 027	과일의 판매 건수 계산하기 - DCOUNTA	30
출제 028	과일의 판매 건수 계산하기 - DCOUNT	31

04 수학 / 삼각 함수

출제 029	1에 가까운 방향으로 자리 올림하기 - ROUNDUP	32
출제 030	조건에 맞는 품목의 합계 구하기 - SUMIF	33
출제 031	절대값 계산하기 - ABS	34
출제 032	나머지 계산하기 - MOD	35
출제 033	반올림하기 - ROUND	36
출제 034	실수를 정수로 변경하기 - INT	37
출제 035	합계 구하기 - SUM	38
출제 036	소수 이하 잘라내기 - TRUNC	39
037	로또 번호 예상하기 1 - RAND	40
038	로또 번호 예상하기 2 - RANDBETWEEN	40
출제 039	자리 내림하기 - ROUNDDOWN	41
출제 040	부서별 직급별 기본급의 합계 계산하기 - SUMIFS	42
출제 041	거듭 제곱 계산하기 - POWER	43

05 찾기 / 참조 함수

출제 042	직위별 상여금 계산하기 - HLOOKUP	44
출제 043	직위별 상여금 계산하기 - VLOOKUP	45
출제 044	요일 번호로 요일 표시하기 - CHOOSE	46
출제 045	직급별, 호봉별 급여 기준액 계산하기 - INDEX	47
출제 046	자료가 기록된 위치 찾기 - MATCH	48
047	현재 셀의 열 번호 알아내기 - COLUMN	49
048	셀 범위에 포함된 열의 개수 알아내기 - COLUMNS	49
049	현재 셀의 행 번호 알아내기 - ROW	50
050	셀 범위에 포함된 행의 개수 알아내기 - ROWS	50

06 텍스트 함수

출제 051	시작위치부터 지정한 수만큼 추출하기 - MID	51
출제 052	왼쪽에서 지정한 수만큼 추출하기 - LEFT	52
출제 053	오른쪽에서 지정한 수만큼 추출하기 - RIGHT	53
출제 054	대·소문자 알맞게 변환하기 - PROPER	54
출제 055	소문자를 대문자로 변환하기 - UPPER	55
출제 056	문자열의 양쪽 끝 공백 제거하기 - TRIM	56
출제 057	대문자를 소문자로 변환하기 - LOWER	57
058	문자열에서 특정 문자의 위치 찾기(글자 단위) - FIND	58
059	문자열의 길이 알아내기 - LEN	58
출제 060	문자열에서 특정 문자의 위치 찾기(글자 단위) - SEARCH	59

07 통계 함수

출제 061	부서별 직급별 인원수 파악하기 - COUNTIFS	60
출제 062	품목의 판매 건수 구하기 - COUNTIF	61
출제 063	순위 계산하기 1 - RANK.EQ	62
출제 064	N번째로 큰 수 찾기 - LARGE	63
출제 065	평균 계산하기 - AVERAGE	64
066	문자도 포함하여 평균 계산하기 - AVERAGEA	65
067	분산 계산하기 - VARS	65
출제 068	자료가 없는 셀의 개수 세기 - COUNTBLANK	66
출제 069	가장 큰 수 찾기 - MAX	67
출제 070	가장 작은 수 찾기 - MIN	68
출제 071	숫자가 들어 있는 셀의 개수 세기 - COUNT	69
출제 072	자료가 입력되어 있는 모든 셀의 개수 세기 - COUNTA	70
출제 073	중간에 위치한 값 찾아내기 - MEDIAN	71
출제 074	가장 많이 나오는 수(최빈수) 찾아내기 - MODE.SNGL	72
출제 075	N번째로 작은 수 찾기 - SMALL	73
출제 076	표준편차 계산하기 - STDEV.S	74
077	과목별 최대 가산점 구하기 - MAXA	75
078	과목별 최소 가산점 구하기 - MINA	75
출제 079	부서별 기본급의 평균 계산하기 - AVERAGEIF	76
출제 080	부서별 직급별 기본급의 평균 계산하기 - AVERAGEIFS	77

02장 계산작업 문제 모음

2025년 상시01 계산작업	80
2025년 상시02 계산작업	82
2025년 상시03 계산작업	84
2025년 상시04 계산작업	86
2024년 상시01 계산작업	88
2024년 상시02 계산작업	90
2024년 상시03 계산작업	92
2024년 상시04 계산작업	94
2023년 상시01 계산작업	96
2023년 상시02 계산작업	98

컴활 함수 사전

컴퓨터활용능력 2급 실기

001 날짜의 일련번호 구하기 — DATE

DATE 함수는 1900년 1월 1일을 기준으로 하여 특정 날짜에 대한 일련번호를 반환하는 함수입니다. 예를 들어 1900년 2월 1일은 32(31+1)로 반환됩니다. DATE 함수는 '년', '월', '일'이 상수가 아닌 수식으로 지정되어야 하는 수식에서 유용하게 사용됩니다.

형식 DATE(년, 월, 일) : '년', '월', '일'에 대한 일련번호를 구합니다.

준비하세요! : '길벗컴퓨터2급통합\부록\함수사전' 폴더의 'DATE.xlsm' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C	D
1	날짜의 일련번호 구하기			
2				
3	년	월	일	일련번호
4	2025	9	17	45917
5	1545	4	28	564419
6	0	1	1	1
7	1900	7	31	213
8	1950	6	25	18439
9	2025	13	45	46067

- ① =DATE(A4, B4, C4) : [A4], [B4], [C4] 셀의 '년', '월', '일'에 대한 일련번호 45917이 [D4] 셀에 입력됩니다.
- ② =DATE(A6, B6, C6) : [A6], [B6], [C6] 셀의 '년', '월', '일'에 대한 일련번호 1이 [D6] 셀에 입력됩니다.
- ③ =DATE(A8, B8, C8) : [A8], [B8], [C8] 셀의 '년', '월', '일'에 대한 일련번호 18439가 [D8] 셀에 입력됩니다.

결과가 날짜 서식으로 표시되므로 일반 숫자로 보려면 '셀 서식' 대화상자의 '표시 형식' 탭에서 범주를 '일반'으로 지정해야 합니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

주민등록번호를 이용하여 생년월일을 표시하세요.

▶ 표시 예 : 1988-10-12 → 1988년 10월 12일

정답 [D3] : =DATE(MID(C3, 1, 2), MID(C3, 3, 2), MID(C3, 5, 2))

	A	B	C	D
1		동호회 회원 명단		
2		성명	주민등록번호	생년월일
3		이정우	881012-1*****	1988년 10월 12일
4		남국현	861123-1*****	1986년 11월 23일
5		연제식	970203-1*****	1997년 2월 3일
6		송달호	960420-1*****	1996년 4월 20일
7		원정균	890804-2*****	1989년 8월 4일

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선 순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=DATE(MID(C3, 1, 2), MID(C3, 3, 2), MID(C3, 5, 2))

① ② ③

- ① MID(C3, 1, 2) : "881012-1*****"의 1 번째 위치부터 2 글자만 추출한 "88"을 반환합니다.
- ② MID(C3, 3, 2) : "881012-1*****"의 3 번째 위치부터 2 글자만 추출한 "10"을 반환합니다.
- ③ MID(C3, 5, 2) : "881012-1*****"의 5 번째 위치부터 2 글자만 추출한 "12"를 반환합니다.
- "88", "10", "12"를 ①, ②, ③에 대입하면 다음과 같습니다.
- =DATE("88", "10", "12") : "88"을 연도로, "10"을 월로, "12"를 일로 적용한 날짜 "1988년 10월 12일"이 [D3] 셀에 입력됩니다.



전문가의 조언

MID 함수는 지정된 문자 수에 따라 텍스트 문자열의 특정 위치에서 원하는 수 만큼의 문자를 추출하는 함수입니다. 자세한 설명은 51쪽을 참고하세요.

002 날짜에서 일만 추출하기 — DAY

DAY 함수는 날짜에서 일(Day)만을 추출하여 반환하는 함수로, 일(Day)은 1에서 31까지의 정수로 표시됩니다. 날짜는 DATE 함수를 사용하여 입력하거나 다른 수식 또는 함수의 결과값으로 입력해야 합니다. 예를 들어, 2025년 5월 23일을 직접 입력하여 일(Day)만을 추출하려면 DAY(DATE(2025, 5, 23))과 같이 DATE 함수를 사용해야 합니다. 날짜를 텍스트로 입력해도 되지만 오류가 발생할 수 있습니다.

형식 DAY(날짜) : '날짜'에서 일을 추출합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'DAY.xlsm' 파일을 열어 '기본' 시트에서 실험하세요.

	A	B
1	날짜에서 일만 추출하기	
2		
3	날짜	일
4	2025/9/17	17
5	2025년 09월 17일	17
6	45,917	17
7	2025/09/17 0:00	17
8	17-Sep-25	17
9	2025-09-31	#VALUE!

① =DAY(A4) : [A4] 셀에서 일만 추출한 17이 [B4] 셀에 입력됩니다.

② =DAY(A6) : 45,917은 "2025/09/17"에 대한 날짜 일련번호로 45,917에서 일만 추출한 17이 [B6] 셀에 입력됩니다.

③ =DAY(A9) : "2025/09/31"이라는 날짜는 없으므로 오류값(#VALUE!)이 [B9] 셀에 입력됩니다. 9월은 30일까지만 있습니다.

기출문제 따라잡기

'기출' 시트에서 실험하세요.

거래일자를 이용하여 상품발송 여부를 표시하세요.

- ▶ 거래일자에서 일이 10일에서 20일 사이이면 "발송", 그 외에는 공백으로 표시

정답 [E3] : IF(AND(DAY(D3)>=10, DAY(D3)<=20), "발송", " ")

	A	B	C	D	E
1		상품거래 일자			
2		거래처	연락처	거래일자	상품발송
3		대한상회	2758-9645	05월 12일	발송
4		상공물산	3547-9852	10월 20일	발송
5		현대수산	4253-1452	08월 23일	
6		고려상사	2563-7412	11월 30일	
7		조선무역	2583-7458	10월 19일	발송

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선 순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=IF(AND(DAY(D3)>=10, DAY(D3)<=20), "발송", " ")

① ② ③

조건(①)이 참(TRUE)이면 ②를 표시하고, 거짓(FALSE)이면 ③을 표시합니다.

• AND(DAY(D3)>=10, DAY(D3)<=20)

④ ⑤

– ④ DAY(D3) : [D3] 셀의 날짜에서 일만 추출한 12가 반환됩니다. 12를 ⑤에 대입하면 다음과 같습니다.

– AND(12>=10, 12<=20) : 12가 10보다 크고, 20보다 작으므로 참(TRUE)을 반환합니다. 참을 ①에 대입하면 다음과 같습니다.

• =IF(참, "발송", " ") : 조건이 참이므로 [E3] 셀에는 "발송"이 입력됩니다.



전문가의 조언

- AND 함수는 인수가 모두 참이면 참(TRUE)을 반환하는 함수입니다. 자세한 설명은 22쪽을 참고하세요.
- IF 함수는 참과 거짓에 관한 논리식을 판별하여 참일 때와 거짓일 때 서로 다른 값을 반환하기 위해 사용하는 함수입니다. 자세한 설명은 20쪽을 참고하세요.

003 근무일 수 계산하기 — DAYS

23.상시, 22.상시, 21.상시

DAYS 함수는 두 날짜 사이의 일수를 반환하는 함수입니다. 날짜는 DATE 함수를 사용하여 입력하거나 다른 수식 또는 함수의 결과값으로 입력해야 합니다. 예를 들어, 2025년 5월 23일을 직접 함수의 인수로 입력하려면 DAYS(DATE(2025, 5, 23), ...)와 같이 DATE 함수를 사용해야 합니다.

형식 DAYS(끝 날짜, 시작 날짜) : '끝 날짜'에서 '시작 날짜'를 뺀 일 수를 계산합니다.

준비하세요! : '길벗컴퓨터2급통합\부록\함수사전' 폴더의 'DAYS.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C
1	근무일수 계산하기		
2			
3	입사일	퇴사일	근무일수
4	2014-12-04	2025-03-05	3,744
5	2017-08-31	2023-05-06	2,074
6	2020-09-22	2024-09-22	1,461
7	2022-11-21	2025-03-08	838
8	2015-05-31	2023-07-31	2,983
9	2016-04-29	2024-03-20	2,882

① =DAYS(B4, A4) : [B4] 셀의 “2025-03-05”에서 [A4] 셀의 “2014-12-04”를 뺀 일 수 3,744가 [C4] 셀에 입력됩니다.

② =DAYS(B6, A6) : [B6] 셀의 “2024-09-22”에서 [A6] 셀의 “2020-09-22”를 뺀 일 수 1,461이 [C6] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

기준일부터 휴가출발일까지 남은일수를 계산하세요.

- ▶ 남은일수 = 휴가출발일 - 기준일
- ▶ 남은일수 뒤에 “일”을 표시 [표시 예 : 3일]

정답 [C4] : =DAYS(B4, \$E\$4) & “일”

	A	B	C	D	E
1	사원별 휴가현황				
2					
3	사원명	휴가출발일	남은일수	기준일	
4	김태현	2025-06-07	13일	2025-05-25	
5	이나연	2025-06-07	13일		
6	한동석	2025-06-10	16일		
7	유영미	2025-06-10	16일		
8	김지윤	2025-06-18	24일		
9	조경원	2025-06-18	24일		

수식의 이해

=DAYS(B4, \$E\$4) & “일”

①

- ① DAYS(B4, \$E\$4) : [B4] 셀의 “2025-06-07”에서 [E4] 셀의 “2025-05-25”를 뺀 일 수 13을 반환합니다. 13을 ①에 대입하면 다음과 같습니다.
- =13 & “일” : 문자열 결합 연산자(&)에 의해 13과 “일”이 합쳐진 “13일”이 [C4] 셀에 입력됩니다.

004 날짜에서 연도만 추출하기 — YEAR

YEAR 함수는 날짜에서 연도(Year)만 추출하여 표시하는 함수로 연도는 1900에서 9999까지의 정수로 표시됩니다.

형식 YEAR(날짜) : '날짜'에서 연도를 추출합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'YEAR.xlsm' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B
1	날짜에서 년만 추출하기	
2		
3	날짜	년
4	2025/9/17	2025
5	2025년 09월 17일	2025
6	45,917	2025
7	2025/09/17 0:00	2025
8	17-Sep-25	2025
9	2025-09-31	#VALUE!

- ① =YEAR(A4) : [A4] 셀의 값 "2025/09/17"에서 연도만 추출한 2025가 [B4] 셀에 입력됩니다.
- ② =YEAR(A6) : 45,917은 "2025/09/17"에 대한 날짜 일련번호로, 45,917에서 연도만 추출한 2025가 [B6] 셀에 입력됩니다.
- ③ =YEAR(A9) : "2025/09/31"이라는 날짜는 없으므로 오류값(#VALUE!)이 [B9] 셀에 입력됩니다. 9월은 30일까지만 있습니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

현재연도와 결혼기념일을 이용해서 결혼주년을 계산하고, 결혼기념일표에서 결혼주년에 해당되는 결혼기념 명칭을 찾아 결혼기념에 표시하세요.

▶ 결혼주년 = 현재연도 - 결혼연도

정답 [E3] : =VLOOKUP(YEAR(TODAY()) - YEAR(D3), \$G\$3:\$H\$9, 2)

	A	B	C	D	E	F	G	H
1								
2				회원 관리 현황				<결혼기념일표>
3		회원번호	이름	결혼기념일	결혼기념		주년	결혼기념명칭
4		95-034	이동권	2021-05-08	혁혼식(革婚式)		1	지혼식(紙婚式)
5		95-035	이동기	2019-02-01	목혼식(木婚式)		2	고혼식(藁婚式)
6		95-036	이동률	2023-01-13	고혼식(藁婚式)		3	과혼식(菓婚式)
7		95-037	이동목	2018-04-03	화혼식(花婚式)		4	혁혼식(革婚式)
8		95-038	이동욱	2020-03-15	목혼식(木婚式)		5	목혼식(木婚式)
9		95-039	이동주	2022-11-10	과혼식(菓婚式)		7	화혼식(花婚式)
10		95-040	이동희	2023-08-18	고혼식(藁婚式)		10	석혼식(錫婚式)

※ 결혼기념의 결과값은 현재 날짜의 연도에 따라 달라집니다.

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

VLOOKUP(YEAR(TODAY()) - YEAR(D3), G3:H9, 2)

• ① TODAY() : 오늘 날짜(오늘이 2025-01-15인 경우) "2025-01-15"를 반환합니다. "2025-01-15"를 ①에 대입하면 다음과 같습니다.

=VLOOKUP(YEAR("2025-01-15") - YEAR(D3), G3:H9, 2)

• ② YEAR("2025-01-15") : "2025-01-15"에서 연도만 추출한 2025를 반환합니다.

• ③ YEAR(D3) : [D3] 셀의 값 "2021-05-08"에서 연도만 추출한 2021을 반환합니다. 2025와 2021을 ②와 ③에 대입하면 다음과 같습니다.

• =VLOOKUP(2025 - 2021, G3:H9, 2) : "2025 - 2021"은 4이므로 계산 순서는 다음과 같습니다.

① [G3:H9] 영역의 맨 왼쪽 열에서 4를 넘지 않는 근사값을 찾습니다. 6행에 있는 4를 찾습니다.

② 4가 있는 행에서 2열에 있는 값 "혁혼식(革婚式)"을 찾아서 [E3] 셀에 입력합니다.

	F	G	H
1			
2			<결혼기념일표>
3		주년	결혼기념명칭
4		1	지혼식(紙婚式)
5		2	고혼식(藁婚式)
6		3	과혼식(菓婚式)
7		4	혁혼식(革婚式)
8		5	목혼식(木婚式)
9		7	화혼식(花婚式)
10		10	석혼식(錫婚式)



전문가의 조언

- VLOOKUP 함수는 범위로 정한 영역의 맨 왼쪽 열에서 특정 기준값으로 자료를 찾고, 그 자료가 속한 행 중에서 필요한 값이 있는 열의 위치를 지정하여 값을 반환하는 함수입니다. 자세한 설명은 45쪽을 참고하세요.
- TODAY 함수는 현재 날짜를 반환하는 함수입니다. 자세한 설명은 11쪽을 참고하세요.

005 날짜에서 요일 알아내기 — WEEKDAY

WEEKDAY 함수는 날짜에서 요일을 추출하는 함수입니다. 기본적으로 요일은 1(일요일)에서 7(토요일)까지의 정수로 표시합니다. 날짜는 DATE 함수를 사용하여 입력하거나 다른 수식 또는 함수의 결과로 입력해야 합니다.

형식 WEEKDAY(날짜, 옵션) : '날짜'에 해당하는 요일번호를 추출합니다. '옵션'은 계산 방법을 지정합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'WEEKDAY.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B
1	요일 알아내기	
2		
3	날짜	요일번호
4	1979-04-28	7
5	1990-03-24	7
6	2013-06-21	6
7	2016-12-01	5
8	2023-08-21	2

- ① =WEEKDAY(A4, 1) : [A4] 셀에 입력된 날짜의 요일 번호를 옵션에 맞게 추출한 7(토요일)이 [B4] 셀에 입력됩니다.
- ② =WEEKDAY(A6, 1) : [A6] 셀에 입력된 날짜의 요일 번호를 옵션에 맞게 추출한 6(금요일)이 [B6] 셀에 입력됩니다.
- ③ =WEEKDAY(A8, 1) : [A8] 셀에 입력된 날짜의 요일 번호를 옵션에 맞게 추출한 2(월요일)이 [B8] 셀에 입력됩니다.

WEEKDAY 함수에서 옵션 값의 종류

- 1 또는 생략 : 1(일요일)에서 7(토요일)까지의 숫자를 사용합니다.
- 2 : 1(월요일)에서 7(일요일)까지의 숫자를 사용합니다.
- 3 : 0(일요일)에서 6(일요일)까지의 숫자를 사용합니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

판매일을 이용하여 해당되는 요일을 판매요일에 표시하세요.

- ▶ 단, 요일 계산 시 일요일이 1인 유형으로 지정
- ▶ '일요일'과 '토요일'만 표시되도록 하고, 나머지 요일은 빈 칸으로 할 것

정답 [E3] : =CHOOSE(WEEKDAY(C3, 1), "일요일", "", "", "", "", "", "토요일")

	A	B	C	D	E
1	제품별 판매단가				
2	제품코드	판매일	단가	판매요일	
3	A-101	2025-03-09	13,500	일요일	
4	A-104	2025-02-03	3,000		
5	A-105	2025-01-04	7,000	토요일	
6	A-107	2025-03-14	9,000		
7	A-104	2025-03-21	3,000		

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

CHOOSE(WEEKDAY(C3, 1), "일요일", "", "", "", "", "", "토요일")

①

① WEEKDAY(C3, 1) : [C3] 셀에 입력된 날짜의 요일번호 1을 반환합니다. 1을 ①에 대입하면 다음과 같습니다.

=CHOOSE(1, "일요일", "", "", "", "", "", "토요일") : 1번째에 있는 일요일이 [E3] 셀에 입력됩니다.



전문가의 조언

- 오늘의 요일을 표시하려면 'WEEKDAY(TODAY())'를 입력하면 됩니다.
- CHOOSE 함수는 인덱스 번호를 이용하여 특정 번째에 있는 값을 반환하는 함수입니다. 자세한 내용은 46쪽을 참고하세요.

006 현재 날짜 표시하기 — TODAY

TODAY 함수는 현재 시스템의 날짜를 반환하는 함수입니다. 함수가 입력되기 전에 셀이 일반 서식을 가지고 있어도 결과 값은 날짜 서식으로 표시됩니다. 현재의 날짜와 시간이 같이 표시되게 하려면 NOW 함수를 사용하세요.

형식 TODAY() : TODAY 함수는 인수 없이 사용합니다. 현재 시스템의 날짜를 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'TODAY.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

	A
1	오늘의 날짜 표시하기
2	
3	2025-05-10 ①
4	

① = TODAY() : 현재의 날짜인 "2025-05-10"이 [A3] 셀에 입력됩니다.

TODAY 함수의 결과값은 현재 날짜에 따라 다르게 표시됩니다.

기출문제 따라잡기

'기술' 시트에서 실행하세요.

현재년도와 결혼기념일을 이용해서 결혼주년을 계산하고, 결혼기념일표에서 결혼주년에 해당되는 결혼기념 명칭을 찾아 결혼기념에 표시하세요.

▶ 결혼주년 = 현재년도 - 결혼년도

정답 [E3] : =VLOOKUP(YEAR(TODAY()) - YEAR(D3), \$G\$3:\$H\$9, 2)

	A	B	C	D	E	F	G	H
1					회원 관리 현황			<결혼기념일표>
2		회원번호	이름	결혼기념일	결혼기념		주년	결혼기념명칭
3		95-034	이동권	2021-05-08	혁혼식(革婚式)		1	지혼식(紙婚式)
4		95-035	이동기	2019-02-01	목혼식(木婚式)		2	고혼식(藁婚式)
5		95-036	이동률	2023-01-13	고혼식(藁婚式)		3	과혼식(菓婚式)
6		95-037	이동목	2018-04-03	화혼식(花婚式)		4	혁혼식(革婚式)
7		95-038	이동옥	2020-03-15	목혼식(木婚式)		5	목혼식(木婚式)
8		95-039	이동주	2022-11-10	과혼식(菓婚式)		7	화혼식(花婚式)
9		95-040	이동희	2023-08-18	고혼식(藁婚式)		10	석혼식(錫婚式)

※ 결혼기념의 결과값은 현재 날짜의 연도에 따라 달라집니다.

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

VLOOKUP(YEAR(TODAY()) - YEAR(D3), G3:H9, 2)

• ① TODAY() : 오늘 날짜(오늘이 2025-01-15인 경우) "2025-01-15"를 반환합니다. "2025-01-15"를 ①에 대입하면 다음과 같습니다.

=VLOOKUP(YEAR("2025-01-15") - YEAR(D3), G3:H9, 2)

• ② YEAR("2025-01-15") : "2025-01-15"에서 연도만 추출한 2025를 반환합니다.

• ③ YEAR(D3) : [D3] 셀의 값 "2021-05-08"에서 연도만 추출한 2021을 반환합니다. 2025와 2021을 ②와 ③에 대입하면 다음과 같습니다.

• =VLOOKUP(2025 - 2021, G3:H9, 2) : "2025 - 2021"은 4이므로 계산 순서는 다음과 같습니다.

④ [G3:H9] 영역의 맨 왼쪽 열에서 4를 넘지 않는 근사값을 찾습니다. 6행에 있는 4를 찾습니다.

⑤ 4가 있는 행에서 2열에 있는 값 "혁혼식(革婚式)"을 찾아서 [E3] 셀에 입력합니다.

	F	G	H
1			<결혼기념일표>
2		주년	결혼기념명칭
3		1	지혼식(紙婚式)
4		2	고혼식(藁婚式)
5		3	과혼식(菓婚式)
6		4	혁혼식(革婚式)
7		5	목혼식(木婚式)
8		7	화혼식(花婚式)
9		10	석혼식(錫婚式)



전문가의 조언

- VLOOKUP 함수는 범위로 정한 영역의 맨 왼쪽 열에서 특정 기준값으로 자료를 찾고, 그 자료가 속한 행 중에서 필요한 값이 있는 열의 위치를 지정하여 값을 반환하는 함수입니다. 자세한 설명은 45쪽을 참고하세요.
- YEAR 함수는 날짜에서 연도(Year)만 추출하여 표시하는 함수입니다. 자세한 설명은 9쪽을 참고하세요.

007 시간의 일련번호 계산하기 — TIME

23.상시, 22.상시, 21.상시, 20.상시, 11.1

TIME 함수는 특정 시간에 대한 일련번호를 실수값으로 반환하는 함수로, 시간은 0에서 0.999999999까지의 실수로 표시되는데, 이는 0:00:00(오전12:00:00)부터 23:59:59(오후11:59:59)까지의 시간을 나타냅니다. 함수가 입력되기 전에 셀이 일반 서식을 가지고 있어도 결과 값은 시간 서식으로 지정됩니다.

형식 TIME(시, 분, 초) : '시', '분', '초'에 대한 시간의 일련번호를 구합니다.

준비하세요! : '길벗컴퓨터2급통합\부록\함수사전' 폴더의 'TIME.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C	D
1	시간의 일련번호 구하기			
2				
3	시	분	초	일련번호
4	0	0	65	0.000752315
5	0	1	5	0.000752315
6	1	1	1	0.042372685
7	25	1	1	0.042372685
8	12	1	1	0.500706019
9	23	59	59	0.999988426

① =TIME(A4, B4, C4) : [A4], [B4], [C4] 셀의 '시', '분', '초'에 대한 일련번호 0.000752315가 [D4] 셀에 입력됩니다.

② =TIME(A6, B6, C6) : [A6], [B6], [C6] 셀의 '시', '분', '초'에 대한 일련번호 0.042372685가 [D6] 셀에 입력됩니다.

③ =TIME(A9, B9, C9) : [A9], [B9], [C9] 셀의 '시', '분', '초'에 대한 일련번호 0.999988426이 [D9] 셀에 입력됩니다.

결과가 시간 서식으로 표시되므로 일련번호 형태로 보려면 '셀 서식' 대화상자의 '표시 형식' 탭에서 범주를 '일반'으로 지정해야 합니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

출발시간과 정류장수를 이용하여 도착예정시간을 구하세요.

▶ 도착예정시간 = 출발시간 + 정류장수 × 정류장 당 소요 시간(4분)

정답 [D4] : =TIME(HOUR(B3), MINUTE(B3)+C3*4, 0)

	A	B	C	D
1	버스운행시간표			
2	도착지	출발시간	정류장수	도착예정시간
3	시청	10:15	5	10:35
4	망원동	9:30	6	9:54
5	서교동	10:05	4	10:21
6	상암동	9:45	5	10:05
7	영등포역	9:45	6	10:09
8	신촌역	10:15	4	10:31
9	남대문	9:30	3	9:42

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=TIME(HOUR(B3), MINUTE(B3)+C3*4, 0)

① ② ③

- ① HOUR(B3) : [B3] 셀에서 시간 추출한 값인 10을 반환합니다.
- ② MINUTE(B3) : [B3] 셀에서 분만 추출한 값인 15를 반환합니다.
- ③ C3*4 : [C3] 셀에 4를 곱한 값인 20을 반환합니다. 10, 15, 20을 ①, ②, ③에 대입하면 다음과 같습니다.
- TIME(10, 15+20, 0) : 10(시), 35(분), 0(초)에 대한 일련번호인 0.440972222가 [D3] 셀에 입력됩니다.

※ 셀 서식이 '시간'으로 지정되어 있어 '10:35'로 표시됩니다.



전문가의 조언

- HOUR 함수는 시간에서 '시'만 추출하는 함수입니다. 자세한 내용은 13쪽을 참고하세요.
- MINUTE 함수는 시간에서 '분'만 추출하는 함수입니다. 자세한 내용은 14쪽을 참고하세요.

008 시간에서 시만 추출하기 — HOUR

23.상시, 20.상시, 19.상시, 11.1, 05.1, 04.4

HOUR 함수는 시간값에서 시(Hour)만을 추출합니다. 시간은 0(오전 12:00)부터 23(오후 11:00)까지의 정수로 표시됩니다. 시간은 큰따옴표로 묶은 텍스트 문자열("6:45 PM")이나 실수(6:45 PM을 의미하는 0.78125) 또는 다른 수식이나 함수의 결과(TIMEVALUE("6:45 PM"))로 입력할 수 있습니다.

형식 HOUR(시간) : '시간'에서 시만 추출합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'HOUR.xlsm' 파일을 열어 '기본' 시트에서 실험하세요.

	A	B
1	시간에서 시만 추출하기	
2		
3	시간	시
4	1:34:55 PM	13
5	1:34:00 AM	1
6	11:01:47 PM	23
7	12:46:48 AM	0
8	2025-09-23 2:25:05 AM	2
9	25-01-Fri 12:58:54 PM	12

① =HOUR(A4) : [A4] 셀에서 시만 추출한 13이 [B4] 셀에 입력됩니다.

② =HOUR(A6) : [A6] 셀에서 시만 추출한 23이 [B6] 셀에 입력됩니다.

③ =HOUR(A9) : [A9] 셀에서 시만 추출한 12가 [B9] 셀에 입력됩니다.

기술문제 따라잡기

'기술' 시트에서 실험하세요.

도착시간과 출발시간의 차이를 이용하여 요금을 계산하세요.

▶ 요금은 10분당 65원임

정답 [D3] : =(HOUR(C3-B3)*60 + MINUTE(C3-B3)) / 10*65

	A	B	C	D
1	고속 버스 요금 계산			
2	버스번호	출발시간	도착시간	요금
3	BS331	10:00	12:30	975
4	BS332	9:10	13:20	1,625
5	BS333	8:30	18:00	3,705
6	BS334	9:00	11:20	910
7	BS335	11:00	12:50	715
8	BS336	12:30	15:20	1,105
9	BS337	15:20	16:00	260
10	BS338	13:20	14:50	585
11	BS339	15:50	18:30	1,040

수식의 이해

$$=(\text{HOUR}(\text{C3-B3}) * 60 + \text{MINUTE}(\text{C3-B3})) / 10 * 65$$

- ① HOUR(C3-B3) : [C3] 셀과 [B3] 셀에서 시만 추출한 후 두 값(12 - 10)의 차이인 2를 반환합니다.
- ② MINUTE(C3-B3) : [C3] 셀과 [B3] 셀에서 분만 추출한 후 두 값(30 - 0)의 차이인 30을 반환합니다. 2와 30을 ①과 ②에 대입하면 다음과 같습니다.
- (2*60+30) / 10*65 : 계산된 결과인 975가 [D3] 셀에 입력됩니다.



전문가의 조언

MINUTE 함수는 시간 값에서 분(Minute)을 추출하는 함수입니다. 자세한 내용은 14쪽을 참고하세요.

009 시간에서 분만 추출하기 — MINUTE

23.상시, 20.상시, 11.1, 04.4

MINUTE 함수는 시간값에서 분(Minute)만을 추출합니다. 분은 0부터 59까지의 정수로 표시됩니다. 시간은 큰따옴표로 묶은 텍스트 문자열("6:45 PM")이나 실수(6:45 PM을 나타내는 0.78125) 또는 다른 수식이나 함수의 결과(TIMEVALUE("6:45 PM"))로 입력할 수 있습니다.

형식 MINUTE(시간) : '시간'에서 분만 추출합니다.

준비하세요! : '길벗컴퓨터2급통합\부록\함수사전' 폴더의 'MINUTE.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

	A	B
1	시간에서 분만 추출하기	
2		
3	시간	분
4	1:34:55 PM	34
5	1:34:00 AM	34
6	11:01:47 PM	1
7	12:46:48 AM	46
8	2025-09-23 2:25:05 AM	25
9	25-01-Fri 12:58:54 PM	58

- ① =MINUTE(A4) : [A4] 셀에서 분만 추출한 34가 [B4] 셀에 입력됩니다.
- ② =MINUTE(A6) : [A6] 셀에서 분만 추출한 1이 [B6] 셀에 입력됩니다.
- ③ =MINUTE(A9) : [A9] 셀에서 분만 추출한 58이 [B9] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실행하세요.

도착시간과 출발시간의 차이를 이용하여 요금을 계산하세요.

▶ 요금은 10분당 65원임

정답 [D3] : =(HOUR(C3-B3)*60 + MINUTE(C3-B3)) / 10*65

	A	B	C	D
1	고속 버스 요금 계산			
2	버스번호	출발시간	도착시간	요금
3	BS331	10:00	12:30	975
4	BS332	9:10	13:20	1,625
5	BS333	8:30	18:00	3,705
6	BS334	9:00	11:20	910
7	BS335	11:00	12:50	715
8	BS336	12:30	15:20	1,105
9	BS337	15:20	16:00	260
10	BS338	13:20	14:50	585
11	BS339	15:50	18:30	1,040

수식의 이해

$$=(\text{HOUR}(\text{C3-B3}) * 60 + \text{MINUTE}(\text{C3-B3})) / 10 * 65$$

- ① HOUR(C3-B3) : [C3] 셀과 [B3] 셀에서 시간 추출한 후 두 값(12 - 10)의 차이인 2를 반환합니다.
- ② MINUTE(C3-B3) : [C3] 셀과 [B3] 셀에서 분만 추출한 후 두 값(30 - 0)의 차이인 30을 반환합니다. 2와 30을 ①과 ②에 대입하면 다음과 같습니다.
- (2*60+30) / 10*65 : 계산된 결과인 975가 [D3] 셀에 입력됩니다.



전문가의 조언

HOUR 함수는 시간 값에서 시(Hour)를 추출하는 함수입니다. 자세한 내용은 13쪽을 참고하세요.

010 날짜에서 월만 추출하기 — MONTH

24.상시, 23.상시, 22.상시, 21.상시, 08.3, 04.4, 04.3

MONTH 함수는 주어진 날짜에서 월(Month)만을 표시하는 함수로, 월은 1에서 12까지의 정수로 표시됩니다.

형식 MONTH(날짜) : '날짜'에서 월만 추출합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'MONTH.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B
1	날짜에서 월만 추출하기	
2		
3	날짜	월
4	2025/9/17	9
5	2025년 09월 17일	9
6	45,917	9
7	2025/09/17 0:00	9
8	17-Sep-25	9
9	2025-09-31	#VALUE!

- ① =MONTH(A4) : [A4] 셀에서 월만 추출한 9가 [B4] 셀에 입력됩니다.
- ② =MONTH(A6) : 45,917은 "2025/09/17"에 대한 날짜 일련번호로 45,917에서 월만 추출한 9가 [B6] 셀에 입력됩니다.
- ③ =MONTH(A9) : "2025/09/31"이라는 날짜는 없으므로 오류값(#VALUE!)이 [B9] 셀에 입력됩니다. 9월은 30일까지만 있습니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

코드, 생산일자, 인식표를 이용하여 제품코드를 표시하세요.

- ▶ 제품코드는 코드 뒤에 "-", 생산일자 중 월 뒤에 "-", 인식표를 연결한 후 대문자로 변환한 것임
- ▶ 표시 예 : 코드가 jh, 생산일자가 2025-10-2, 인식표가 ek이면
→ JH-10-EK

정답 [D3] : =UPPER(A3) & "-" & MONTH(B3) & "-" & UPPER(C3)

	A	B	C	D
1	생산품목 현황			
2	코드	생산일자	인식표	제품코드
3	ag	2025-11-11	w	AG-11-W
4	rf	2025-08-30	e	RF-8-E
5	dk	2025-12-30	f	DK-12-F
6	ik	2025-10-15	d	IK-10-D
7	wd	2025-11-22	e	WD-11-E
8	od	2025-12-10	w	OD-12-W
9	uf	2025-09-03	h	UF-9-H
10	dh	2025-12-29	d	DH-12-D

수식의 이해

=UPPER(A3) & "-" & MONTH(B3) & "-" & UPPER(C3)

- ① UPPER(A3) : [A3] 셀의 문자열을 대문자로 변환한 "AG"를 반환합니다.
 - ② MONTH(B3) : [B3] 셀에서 월만 추출한 11을 반환합니다.
 - ③ UPPER(C3) : [C3] 셀의 문자열을 대문자로 변환한 "W"를 반환합니다. "AG", 11, "W"를 ①, ②, ③에 대입하면 다음과 같습니다.
- "AG" & "-" & 11 & "-" & "W" : & 연산자에 의해 모든 문자열이 합쳐진 "AG-11-W"가 [D3] 셀에 입력됩니다.



전문가의 조언

- UPPER 함수는 문자열에 포함된 소문자를 모두 대문자로 변환하는 함수입니다. 자세한 내용은 55쪽을 참고하세요.
- &는 2개 이상의 문자열을 합쳐 하나의 문자열로 만들 때 사용하는 문자열 결합 연산자입니다.

011 시간에서 초만 추출하기 — SECOND

SECOND 함수는 시간값에서 초(Second)만을 추출합니다. 초는 0부터 59까지의 정수로 표시됩니다. 시간은 큰따옴표로 묶은 텍스트 문자열("6:45PM")이나 실수(6:45 PM)를 나타내는 0.78125) 또는 다른 수식이나 함수의 결과(TIMEVALUE("6:45 PM"))로 입력할 수 있습니다.

형식 SECOND(시간) : '시간'에서 초만 추출합니다.

준비하세요! : '길벗컴퓨터2급통합\부록\함수사전' 폴더의 'SECOND.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

	A	B
1	시간에서 초만 추출하기	
2		
3	시간	초
4	1:34:55 PM	55
5	1:34:00 AM	0
6	11:01:47 PM	47
7	12:46:48 AM	48
8	2025-09-23 2:25:05 AM	5
9	25-01-Fri 12:58:54 PM	54

① =SECOND(A4) : [A4] 셀에서 초만 추출한 55가 [B4] 셀에 입력됩니다.

② =SECOND(A6) : [A6] 셀에서 초만 추출한 47이 [B6] 셀에 입력됩니다.

③ =SECOND(A9) : [A9] 셀에서 초만 추출한 54가 [B9] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실행하세요.

기록이 가장 빠른 선수의 기록을 찾아 표시하세요.

▶ 표시 예 : 1:23:34 → 1시간23분34초

정답 [C9] : =HOUR(SMALL(D3:D8, 1)) & "시간" &
MINUTE(SMALL(D3:D8, 1)) & "분" &
SECOND(SMALL(D3:D8, 1)) & "초"

	A	B	C	D
1	하프마라톤 결과			
2	선수번호	소속	나이	기록
3	168001	춘천	35	1:49:27
4	168002	영월	42	1:45:51
5	168003	강릉	29	2:03:26
6	168004	평창	38	1:51:15
7	168005	영월	44	1:32:08
8	168006	고성	51	1:41:53
9	가장 빠른 기록		1시간32분8초	

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=HOUR(SMALL(D3:D8, 1)) & "시간" & MINUTE(SMALL(D3:D8, 1)) & "분" & SECOND(SMALL(D3:D8, 1)) & "초"

- ① SMALL(D3:D8, 1) : [D3:D8] 영역에서 가장 작은 기록인 1:32:08을 반환합니다. 1:32:08을 ①에 대입하면 다음과 같습니다.
- ② HOUR(1:32:08) : 1:32:08에서 시간 추출한 값인 1을 반환합니다.
- ③ MINUTE(1:32:08) : 1:32:08에서 분만 추출한 값인 32를 반환합니다.
- ④ SECOND(1:32:08) : 1:32:08에서 초만 추출한 값인 08을 반환합니다.
- 1 & "시간" & 32 & "분" & 8 & "초" : & 연산자에 의해 모든 문자열이 합쳐진 "1시간32분8초"가 [C9] 셀에 입력됩니다.

012 현재 날짜와 시간 표시하기 — NOW

NOW 함수는 현재 날짜와 시간을 반환하는 함수입니다. 함수가 입력되기 전에 셀이 일반 서식을 가지고 있어도 결과 값은 날짜 서식으로 지정됩니다. 오늘의 날짜만 나타나도록 하려면 TODAY() 함수를 사용하세요.

형식 NOW() : NOW 함수는 인수 없이 사용합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'NOW.xlsm' 파일을 열어 '기본' 시트에서 실험하세요.

	A	
1	날짜와 시간 표시하기	
2		
3	2025-05-10 19:40	1
4		

① =NOW() : 현재의 날짜와 시간인 '2025-05-10 19:40'이 [A3] 셀에 입력됩니다.

Now 함수의 결과값은 현재 날짜와 시간에 따라 다르게 입력됩니다.

013 3개월 전/후 오늘의 일련번호 구하기 — EDATE

EDATE 함수는 특정 날짜로부터 지정한 개월이 더해진 날짜에 대한 일련번호를 반환하는 함수입니다. 날짜는 DATE 함수를 사용하여 입력하거나 다른 수식 또는 함수의 결과값으로 입력해야 합니다. 예를 들어 2025년 10월 14일을 직접 입력하여 한 달 전 날짜를 추출하려면 EDATE(DATE(2025, 10, 14), -1) 과 같이 DATE 함수를 사용해야 합니다. 날짜를 텍스트로 입력해도 되지만 오류가 발생할 수 있습니다.

형식 EDATE(시작 날짜, 개월 수) : '시작 날짜'에서 '개월 수'를 더한 날짜를 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'EDATE.xlsm' 파일을 열어 '기본' 시트에서 실험하세요.

	A	B	C
1	3개월 전/후 오늘의 일련번호 구하기		
2			
3	날짜	전/후 개월 수	일련번호
4	2025-08-20	-3	45797
5	2025-12-01	-2	45931
6	2025-07-25	-1	45833
7	2025-08-23	1	45923
8	2025-06-21	2	45890
9	2025-09-31	3	#VALUE!

① =EDATE(A4, B4) : 더해진 개월 수가 음수이므로 [A4] 셀의 날짜에서 3개월 전의 날짜에 대한 일련번호 45797가 [C4] 셀에 입력됩니다.

② =EDATE(A9, B9) : "2025-09-31"이라는 날짜는 없으므로 오류값 (#VALUE!)이 [C9] 셀에 입력됩니다. 9월은 30일까지만 있습니다.

- 특정 날짜에 지정한 개월 수를 더했을 때 결과에 해당하는 날짜가 없는 경우에는 가장 마지막 날짜가 표시됩니다. 예를 들어 3월 31일에 1개월을 더하면 4월 31일 표시되어야 하지만 4월은 30일까지만 있으므로 4월 30일이 표시됩니다.
- 더해지는 개월 수는 정수만 가능하므로, 소수점 이하 자릿수는 무시됩니다.

014 3개월 전/후 달의 마지막 날짜에 대한 일련번호 구하기 — EOMONTH

EOMONTH 함수는 특정 날짜로부터 지정한 개월이 더해진 달의 마지막 날짜에 대한 일련번호를 반환하는 함수입니다. 날짜는 DATE 함수를 사용하여 입력하거나 다른 수식 또는 함수의 결과값으로 입력해야 합니다. 예를 들어 2025년 10월 14일을 직접 입력하여 한 달 전 마지막 날짜를 추출하려면 EOMONTH(DATE(2025, 10, 14), -1)과 같이 DATE 함수를 사용해야 합니다. 날짜를 텍스트로 입력해도 되지만 오류가 발생할 수 있습니다.

형식 EOMONTH(시작 날짜, 개월 수) : '시작 날짜'에서 '개월 수'를 더한 달의 마지막 날짜를 반환합니다.

준비하세요! : '값벡터\2급통합\부록\함수사전' 폴더의 'EOMONTH.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C
1	전/후 달 마지막 날짜의 일련번호 구하기		
2			
3	날짜	전/후 개월 수	일련번호
4	2025-08-20	-3	45808
5	2025-12-01	-2	45961
6	2025-07-25	-1	45838
7	2025-08-23	1	45930
8	2025-06-21	2	45900
9	2025-09-31	3	#VALUE!

① =EOMONTH(A4, B4) : 더해진 개월 수가 음수이므로 [A4] 셀의 날짜에서 3개월 전 달의 마지막 날짜에 대한 일련번호 45808이 [C4] 셀에 입력됩니다.

② =EOMONTH(A9, B9) : “2025-09-31”이라는 날짜는 없으므로 오류값 (#VALUE!)이 [C9] 셀에 입력됩니다. 9월은 30일까지만 있습니다.

더해지는 개월 수는 정수만 가능하므로 소수점 이하 자릿수는 무시됩니다.

015 주말을 제외한 휴가 마지막 날 계산하기 — WORKDAY

23.상시, 19.상시, 14.3

WORKDAY 함수는 특정일을 기준으로 해서 토, 일요일과 휴일 날짜를 제외하고 지정한 근무일 수만큼 지난 날이 며칠인지를 계산하는 함수입니다. 날짜는 DATE 함수를 사용하여 입력하거나 다른 수식 또는 함수의 결과값으로 입력해야 합니다. 예를 들어 2025년 10월 14일을 직접 입력하여 3일째 근무하는 날을 계산하려면 WORKDAY(DATE(2025, 10, 14), 3)과 같이 DATE 함수를 사용해야 합니다. 날짜를 텍스트로 입력해도 되지만 오류가 발생할 수 있습니다.

형식 WORKDAY(시작 날짜, 일 수, 휴일 날짜) : '시작 날짜'에 토, 일요일과 '휴일 날짜'를 제외하고 '일수'만큼 지난 날짜를 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'WORKDAY.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

WORKDAY 함수를 이용하여 토, 일요일과 휴일 날짜를 제외한 휴가 마지막 날을 계산해 보겠습니다.

	A	B	C	D
1	여름 휴가 일정			
2				
3	근무 마지막 날	휴가일수	휴일날짜	휴가 마지막날
4	2025-08-21	7	2025-08-25	2025-09-02
5	2025-12-01	8		2025-12-11
6	2025-07-25	7	2025-07-28	2025-08-06
7	2025-08-23	5		2025-08-29
8	2025-06-21	3		2025-06-25
9	2025-09-31	7	2025-10-02	#VALUE!

① =WORKDAY(A4, B4, C4) : [A4] 셀의 날짜에서 토, 일요일과 휴일 날짜를 제외하고 7일째가 되는 "2025-09-02"가 [D4] 셀에 입력됩니다.

② =WORKDAY(A9, B9, C9) : "2025-09-31"이라는 날짜는 없으므로 오류값(#VALUE!)이 [D9] 셀에 입력됩니다. 9월은 30일까지만 있습니다.

더해지는 날짜 수는 정수만 가능하므로 소수점 이하 자릿수는 무시됩니다.

기출문제 따라잡기

'기출' 시트에서 실행하세요.

휴가출발일과 휴가일수를 이용하여 회사출근일을 표시하세요.

▶ 주말(토, 일요일)은 제외

정답 [D4] : =WORKDAY(B4, C4)

	A	B	C	D
1	휴가 일정표			
2				
3	성명	휴가출발일	휴가일수	회사출근일
4	성소민	2025-04-01	4	2025-04-07
5	이수양	2025-04-01	8	2025-04-11
6	김성찬	2025-04-10	6	2025-04-18
7	장선옥	2025-04-10	7	2025-04-21
8	이수아	2025-04-16	8	2025-04-28
9	최수현	2025-04-16	5	2025-04-23

수식의 이해

=WORKDAY(B4, C4)

[B4] 셀의 날짜(2025-04-01)에서 토, 일요일과 휴일 날짜를 제외하고 4일째가 되는 "2025-04-07"이 [D4] 셀에 입력됩니다.

016 1월 실적이 평균 이상이면 “우수”, 평균 미만이면 “미달” 표시하기 — IF

IF 함수는 참과 거짓에 관한 논리식을 판별하여 참일 때와 거짓일 때 서로 다른 값을 반환하기 위해 사용하는 함수입니다. 예를 들어 1월 실적이 평균 이상이면 “우수”를 반환하고 평균 미만이면 “미달”을 반환하는 수식은 ‘=IF(1월실적 >= 평균, “우수”, “미달”)’과 같이 입력하여 사용할 수 있습니다.

형식 IF(조건, 인수1, 인수2) : ‘조건’이 참이면 ‘인수1’, 거짓이면 ‘인수2’를 반환합니다.

준비하세요! : ‘길벗컴활2급통합\부록\함수사전’ 폴더의 ‘IF.xlsm’ 파일을 열어 ‘기본’ 시트에서 실습하세요.

IF 함수를 이용하여 1월 실적이 평균보다 크면 평가에 “우수”를, 그렇지 않으면 공백을 표시해 보겠습니다.

	A	B	C	D	E
1		개인별 영업 실적 현황			
2					
3		영업소	사원이름	1월 실적	평가
4		서울	김정식	137,000	우수
5		경기	박기수	78,900	
6		강원	한송희	57,900	
7		충북	장영철	103,400	우수
8		대구	김만호	117,800	우수
9		경북	최수정	78,900	
10		부산	서용식	114,000	우수
11		평균		98,271	

- ① =IF(D4>D11, “우수”, “ ”) : [D4] 셀의 1월 실적(137,000)이 평균 실적(98,271)을 초과하므로 “우수”가 [E4] 셀에 표시됩니다.
- ② =IF(D6>D11, “우수”, “ ”) : [D6] 셀의 1월 실적(57,900)이 평균 실적(98,271) 미만이므로 공백이 [E6] 셀에 표시됩니다.
- ③ =IF(D9>D11, “우수”, “ ”) : [D9] 셀의 1월 실적(78,900)이 평균 실적(98,271) 미만이므로 공백이 [E9] 셀에 표시됩니다.

[E4] 셀에 수식을 입력한 후 [E10] 셀까지 수식을 복사하여 나머지 셀을 계산하려면 ‘=IF(D4>D\$11, “우수”, “ ”)’로 입력해야 합니다.

기출문제 따라잡기

‘기출’ 시트에서 실습하세요.

근태점수, 실적점수, 연수점수 모두가 70 이상이면 “승진”, 그렇지 않으면 공백을 승진여부에 표시하세요.

정답 [F3] : =IF(AND(C3>=70, D3>=70, E3>=70), “승진”, “ ”)

	A	B	C	D	E	F
1		사원 승진 심사표				
2		성명	근태점수	실적점수	연수점수	승진여부
3		김선우	78	86	67	
4		유세준	90	88	70	승진
5		손상훈	62	76	84	
6		김승완	83	56	78	
7		박진수	89	74	96	승진

수식의 이해

=IF(AND(C3>=70, D3>=70, E3>=70), “승진”, “ ”)

① ② ③

• 조건①이 참(TRUE)이면 ②를, 거짓(FALSE)이면 ③을 표시합니다.

• ① AND(C3>=70, D3>=70, E3>=70)

④ ⑤ ⑥

— ④, ⑤, ⑥이 모두 참이면 참을, 셋 중에 하나라도 거짓이면 거짓을 반환합니다.

— ④ : [C3] 셀의 값 78이 70보다 크므로 참

— ⑤ : [D3] 셀의 값 86이 70보다 크므로 참

— ⑥ : [E3] 셀의 값 67이 70보다 작으므로 거짓

— ④와 ⑤는 참이나 ⑥이 거짓이므로 거짓을 반환합니다.

• 조건①이 거짓이므로 ③을 수행하여 [F3] 셀에 “ ”(공백)이 표시됩니다.



전문가의 조언

AND 함수는 인수가 모두 참이면 참(TRUE)을 반환하는 함수입니다. 자세한 설명은 22쪽을 참고하세요.

017 과목번호가 10이면 “영어”, 20이면 “수학”, 30이면 “국어” 표시하기 — IFS

IFS 함수는 조건이 여러 개일 때 조건에 만족하는 값을 반환하는 함수입니다. 예를 들어 과목번호가 10이면 “영어”, 20이면 “수학”, 30이면 “국어”를 반환하는 수식은 ‘=IFS(과목번호=1, “영어”, 과목번호=2, “수학”, 과목번호=3, “국어”)’와 같이 입력하여 사용할 수 있습니다.

형식 IFS(조건1, 인수1, 조건2, 인수2, ... 조건n, 인수n) : 조건1이 ‘참’이면 인수1, 조건2가 ‘참’이면 인수2, ... 조건n이 ‘참’이면 인수n을 반환합니다.

준비하세요! ‘길벗컴활2급통합\부록\함수사전’ 폴더의 ‘IFS.xlsx’ 파일을 열어 ‘기본’ 시트에서 실험하세요.

IFS 함수를 이용하여 과목번호가 1이면 “영어”, 2이면 “수학”, 3이면 “국어”를 표시해 보겠습니다.

	A	B	C	D	E
1					
2					
3					
4		이름	성별	과목번호	담당과목
5		신민서	여	1	영어
6		이한열	남	2	수학
7		박성훈	남	3	국어
8		최영선	여	1	영어
9		유현숙	여	2	수학

① =IFS(D4=1, “영어”, D4=2, “수학”, D4=3, “국어”) : [D4] 셀의 과목번호가 1이므로 “영어”가 [E4] 셀에 표시됩니다.

② =IFS(D6=1, “영어”, D6=2, “수학”, D6=3, “국어”) : [D6] 셀의 과목번호가 3이므로 “국어”가 [E6] 셀에 표시됩니다.

마지막 조건인 D4=3 대신 TRUE를 입력해도 됩니다.

=IFS(D4=1, “영어”, D4=2, “수학”, TRUE, “국어”)

018 요일이 “토”나 “일”이면 “주말”, “월”~“금”이면 “평일” 표시하기 — SWITCH

SWITCH 함수는 IFS와 마찬가지로 조건이 여러 개일 때 조건에 만족하는 값을 반환하는 함수입니다. 예를 들어 요일이 “토”나 “일”이면 “주말”을 반환하고, “월”~“금”이면 “평일”을 반환하는 수식은 ‘=SWITCH(요일, “토”, “주말”, “일”, “주말”, “평일”)’과 같이 입력하여 사용할 수 있습니다.

형식 SWITCH(인수, 비교값1, 반환값1, 비교값2, 반환값2, ..., 일치하는 비교값 없을 때 반환값) : ‘인수’가 ‘비교값1’과 같으면 ‘반환값1’, ‘비교값2’와 같으면 ‘반환값2’, ... ‘인수’와 일치하는 비교값이 없을 경우 ‘일치하는 비교값이 없을 때 반환값’을 반환합니다.

준비하세요! ‘길벗컴활2급통합\부록\함수사전’ 폴더의 ‘SWITCH.xlsx’ 파일을 열어 ‘기본’ 시트에서 실험하세요.

SWITCH 함수를 이용하여 요일이 “토”나 “일”이면 “주말”, “월”~“금”이면 “평일”을 표시해 보겠습니다.

	A	B	C	D
1				
2				
3		날짜	요일	구분
4		2025-07-04	금	평일
5		2025-07-05	토	주말
6		2025-07-06	일	주말
7		2025-07-07	월	평일
8		2025-07-08	화	평일

① =SWITCH(C4, “토”, “주말”, “일”, “주말”, “평일”) : [C4] 셀의 요일이 “금”이므로 “평일”이 [D4] 셀에 표시됩니다.

② =SWITCH(C6, “토”, “주말”, “일”, “주말”, “평일”) : [C6] 셀의 요일이 “일”이므로 “주말”이 [D6] 셀에 표시됩니다.

019 모두 참(TRUE)일 때 참(TRUE) 반환하기 — AND

AND 함수는 여러 개의 논리식 결과가 모두 참(TRUE)일 때만 참(TRUE)을 반환하는 함수입니다. 예를 들어 임의의 값 X가 5보다 크고, 10보다 작은 조건에 맞는지를 판별할 때는 '=AND(X>5, X<10)'와 같이 함수식을 입력하면 됩니다. 그러면 X가 6~9 사이에 있을 때만 논리식이 모두 참이 되어 TRUE를 반환하고, 6~9 사이를 벗어나면 거짓이 되어 FALSE를 반환합니다. AND 함수 안에 표시하는 인수는 1개에서 255개까지 지정할 수 있습니다.

형식 AND(인수1, 인수2, ...): 주어진 '인수'가 모두 참이면 참입니다.

준비하세요! : '길벗컴퓨터2급통합' 부록 '함수사전' 폴더의 'AND.xlsm' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C	D
1				
2				
3				
4				
5				
6				
7				

- ① =AND(B3:C3) : [B3] 셀과 [C3] 셀의 값이 모두 거짓(FALSE)이므로 거짓(FALSE)이 [D3] 셀에 입력됩니다.
- ② =AND(B4:C4) : [B4] 셀은 참(TRUE), [C4] 셀은 거짓(FALSE)이므로 거짓(FALSE)이 [D4] 셀에 입력됩니다.
- ③ =AND(B6:C6) : [B6] 셀과 [C6] 셀의 값이 모두 참(TRUE)이므로 참(TRUE)이 [D6] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

근태점수, 실적점수, 연수점수 모두가 70 이상이면 "승진", 그렇지 않으면 공백을 승진여부에 표시하세요.

정답 [F3] : =IF(AND(C3>=70, D3>=70, E3>=70), "승진", "")

	A	B	C	D	E	F
1						
2						
3						
4						
5						
6						
7						

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=IF(AND(C3>=70, D3>=70, E3>=70), "승진", "")

• 조건①이 참(TRUE)이면 ②를, 거짓(FALSE)이면 ③을 표시합니다.

• ① AND(C3>=70, D3>=70, E3>=70)

— ④, ⑤, ⑥이 모두 참이면 참을, 셋 중에 하나라도 거짓이면 거짓을 반환합니다.

— ④ : [C3] 셀의 값 78이 70보다 크므로 참

— ⑤ : [D3] 셀의 값 86이 70보다 크므로 참

— ⑥ : [E3] 셀의 값 67이 70보다 작으므로 거짓

— ④와 ⑤는 참이나 ⑥이 거짓이므로 거짓을 반환합니다.

• 조건①이 거짓이므로 ③을 수행하여 [F3] 셀에 " "(공백)이 표시됩니다.



전문가의 조언

IF 함수는 참과 거짓에 관한 논리식을 판별하여 참일 때와 거짓일 때 서로 다른 값을 반환하기 위해 사용하는 함수입니다. 자세한 설명은 20쪽을 참고하세요.

020 하나라도 참(TRUE)이면 참(TRUE) 반환하기 — OR

OR 함수는 여러 개의 논리식 결과 중 하나라도 참(TRUE)이면 참(TRUE)을 반환하는 함수입니다. 예를 들어 임의의 값 X가 10보다 크거나, 5보다 작은 조건에 맞는지를 판별할 때는 $=OR(X>10, X<5)$ 와 같이 함수식을 입력하면 됩니다. 그러면 X가 5~10 사이에 있을 때만 모두 거짓이 되어 FALSE를 반환하고, 그 이외의 값은 모두 TRUE를 반환합니다. OR 함수 안에 표시하는 인수는 1개에서 255개까지 지정할 수 있습니다.

형식 OR(인수1, 인수2, ...): '인수' 중 하나라도 참이면 참입니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'OR.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

	A	B	C	D
1		OR 진리표		
2		A	B	X
3		FALSE	FALSE	FALSE
4		TRUE	FALSE	TRUE
5		FALSE	TRUE	TRUE
6		TRUE	TRUE	TRUE

- ① $=OR(B3:C3)$: [B3] 셀과 [C3] 셀이 모두 거짓(FALSE)이므로 거짓(FALSE)이 [D3] 셀에 입력됩니다.
- ② $=OR(B4:C4)$: [B4] 셀은 참(TRUE), [C4] 셀은 거짓(FALSE)이므로 참(TRUE)이 [D4] 셀에 입력됩니다.
- ③ $=OR(B6:C6)$: [B6] 셀과 [C6] 셀이 모두 참(TRUE)이므로 참(TRUE)이 [D6] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실행하세요.

개인별 계획서, 작품성, 평점의 점수 중 한 항목이라도 점수가 70 미만이면 "탈락", 그 외는 "통과"로 평가에 표시하세요.

정답 [F3] : $=IF(OR(C3<70, D3<70, E3<70), "탈락", "통과")$

	A	B	C	D	E	F
1		1차 대회 성적				
2		성명	계획서	작품성	평점	평가
3		이구연	90	78	84	통과
4		나잘해	78	80	79	통과
5		정민철	92	70	81	통과
6		최고다	45	34	39.5	탈락
7		한가위	56	45	50.5	탈락

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

$=IF(OR(C3<70, D3<70, E3<70), "탈락", "통과")$

- 조건 ①이 참(TRUE) ②를 표시하고, 거짓(FALSE)이면 ③을 표시합니다.
- ① $OR(C3<70, D3<70, E3<70)$
 - ④, ⑤, ⑥ 중 하나라도 참이면 참을, 셋 모두 거짓이면 거짓을 반환합니다.
 - ④ : [C3] 셀의 값 90이 70보다 크므로 거짓
 - ⑤ : [D3] 셀의 값 78이 70보다 크므로 거짓
 - ⑥ : [E3] 셀의 값 84가 70보다 크므로 거짓
 - ④, ⑤, ⑥이 모두 거짓이므로 거짓을 반환합니다.
 - 조건 ①이 거짓이므로 ③을 수행하여 [F3] 셀에 "통과"가 표시됩니다.



전문가의 조언

IF 함수는 참과 거짓에 관한 논리식을 판별하여 참일 때와 거짓일 때 서로 다른 값을 반환하기 위해 사용하는 함수입니다. 자세한 설명은 20쪽을 참고하세요.

021 논리식의 결과 부정하기 — NOT

NOT 함수는 논리식의 결과를 부정하는 함수입니다. 즉, 논리식의 결과가 참(TRUE)이면 거짓(FALSE)을 반환하고, 거짓(FALSE)이면 참(TRUE)을 반환하는 함수입니다.

형식 NOT(인수) : '인수'의 반대 값을 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'NOT.xlsm' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C
1		NOT 진리표	
2		A	X
3		FALSE	TRUE
4		TRUE	FALSE

- ① =NOT(B3) : [B3] 셀의 값이 거짓(FALSE)이므로 반대 값인 참(TRUE)이 [C3] 셀에 입력됩니다.
- ② =NOT(B4) : [B4] 셀의 값이 참(TRUE)이므로 반대 값인 거짓(FALSE)이 [C4] 셀에 입력됩니다.

022 수식의 결과가 오류일 경우 “오류” 표시하기 — IFERROR

IFERROR 함수는 인수로 지정한 수식에서 오류가 발생할 경우 사용자가 지정한 오류 값을 반환하고, 그렇지 않으면 수식의 결과를 반환하는 함수입니다.

형식 IFERROR(인수, 오류 시 표시할 값) : '인수'로 지정한 수식이나 셀에서 오류가 발생했으면 '오류 시 표시할 값'을 반환하고, 그렇지 않으면 결과값을 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'IFERROR.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C
1		오류 여부 확인하기	
2			
3		200	100
4		200	100
5		0	5
6	1	오류	40
7			

① =IFERROR((B3+B4)/B5, “오류”) : (B3+B4)/B5의 결과가 오류(#DIV/0!)이므로 [B6] 셀에 “오류”가 입력됩니다.

② =IFERROR((C3+C4)/C5, “오류”) : (C3+C4)/C5의 결과 40이 [C6] 셀에 입력됩니다.

#DIV/0! : 나누는 수가 빈 셀이나 0이 있는 셀을 참조한 경우 나타나는 오류 메시지입니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

내신등급과 등급표를 이용하여 등급을 구하시요. 단, 내신등급이 등급표에 존재하지 않는 경우 등급에 “등급오류”라고 표시하세요.

▶ 등급표의 의미 : 내신등급이 1~3이면 “A”, 4~6이면 “B”, 7~10이면 “C”, 11~13이면 “D”, 14 이상이면 “E”를 적용함

정답 [F3] : =IFERROR(HLOOKUP(E3, \$B\$13:\$F\$14, 2), “등급오류”)

	A	B	C	D	E	F
1	입학 지원자 현황					
2	학교명	성명	결석일수	자격증	내신등급	등급
3	대한고	서유민	0	유	2	A
4	망원고	엄진아	10	무	14	E
5	명유고	표현진	6	무	0	등급오류
6	군자고	전수식	4	유	1	A
7	강서고	김정린	5	유	8	C
8	영생고	강남원	1	유	4	B
9	수영고	이진국	2	유	12	D
10	명천고	안현정	5	무	6	B
11						
12	<등급표>					
13	내신등급	1	4	7	11	14
14	등급	A	B	C	D	E

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=IFERROR(HLOOKUP(E3, \$B\$13:\$F\$14, 2), “등급오류”)

①

- ① HLOOKUP(E3, \$B\$13:\$F\$14, 2) : [E3] 셀의 값이 20이므로 계산 순서는 다음과 같습니다.
- ② : [B13:F14] 영역의 첫 번째 행에서 2를 넘지 않는 근사값을 찾습니다. B열에 있는 1을 찾습니다.
- ③ : 1이 있는 열에서 2행에 있는 값 “A”를 찾아서 반환합니다.

	A	B	C	D	E	F
12			<등급표>			
13	내신등급	1	4	7	11	14
14	등급	A	B	C	D	E

- =IFERROR(①, “등급오류”) : ①의 결과가 오류가 아니고 “A”를 반환하므로 [F3] 셀에 “A”가 입력됩니다.



전문가의 조언

HLOOKUP 함수는 범위로 정한 영역의 첫 번째 행에서 특정 기준값으로 자료를 찾고, 그 자료가 속한 열 중에서 필요한 값이 있는 행의 위치를 지정하여 값을 반환하는 함수입니다. 자세한 내용은 44쪽을 참고하세요.

023 과일의 평균 판매 금액 계산하기 — DAVERAGE

DAVERAGE 함수는 데이터 목록에서 조건에 맞는 자료만의 평균을 계산할 때 사용하는 함수입니다. 데이터 목록은 행(레코드)과 열(필드)로 이루어진 관련 데이터의 모임으로 목록의 첫째 행에는 반드시 각 열의 제목(필드명)이 있어야 합니다. 그리고 DAVERAGE 함수와 같은 데이터베이스 함수는 데이터 목록과는 별도의 셀에 조건을 지정한 후 함수의 인수로 사용하는 것이 일반적입니다.

형식 DAVERAGE(범위, 열 번호, 조건) : 해당 '범위'에서 '조건'에 맞는 자료를 대상으로 지정된 '열 번호'에서 평균을 계산합니다.

준비하세요! : '갈빗컴활2급통합\부록\함수사전' 폴더의 'DAVERAGE.xsm' 파일을 열어 '기본' 시트에서 실험하세요.

DAVERAGE 함수를 이용하여 품목별로 평균 판매 금액을 계산해 보겠습니다.

	A	B	C	D	E	F
1		과일 판매 현황				
2		날짜	품명	수량	단가	금액
3		1	사과	10	100	1,000
4		1	배	20	200	4,000
5		1	감	30	300	9,000
6		2	사과	10	100	1,000
7		2	배	20	200	4,000
8		2	감	10	300	3,000
9			<품목별 금액의 평균>			
10			품명	품명	품명	
11			사과	배	감	
12		1	1,000	4,000	6,000	2
13						

① =DAVERAGE(B2:F8, 5, C11:C12) : [B2:F8] 영역에서 [C11:C12] 영역의 조건, 즉 품명이 "사과"인 과일의 금액들을 5열(F)에서 찾은 후 금액들(1,000, 1,000)의 평균인 1,000이 [C13] 셀에 입력됩니다.

② =DAVERAGE(B2:F8, 5, E11:E12) : [B2:F8] 영역에서 [E11:E12] 영역의 조건, 즉 품명이 "감"인 과일의 금액들을 5열(F)에서 찾은 후 금액들(9,000, 3,000)의 평균인 6,000이 [E13] 셀에 입력됩니다.

조건을 지정할 때는 반드시 데이터 목록의 필드명과 같은 필드명을 사용해야 합니다.

기출문제 따라잡기

'기출' 시트에서 실험하세요.

"출신고"가 "우주고"인 학생들의 종합 점수 평균을 구하여 표시하세요.

▶ 우주고 종합 평균은 소수점 이하 둘째 자리에서 반올림하여 첫째 자리까지 표시 [표시 예 : 64.66 → 64.7]

정답 [D10] : =ROUND(DAVERAGE(B2:F7, 5, C2:C3), 1)

	A	B	C	D	E	F
1		경시대회 성적				
2		성명	출신고	필기	실기	종합
3		고영인	우주고	75	95	85
4		성수영	대한고	77	89	83
5		구정철	우주고	88	93	90.5
6		박대철	우주고	91	67	79
7		전소영	상공고	85	56	70.5
8						
9				우주고 종합 평균		
10				84.8		

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=ROUND(DAVERAGE(B2:F7, 5, C2:C3), 1)

①

- ① DAVERAGE(B2:F7, 5, C2:C3) : [B2:F7] 영역에서 [C2:C3] 영역의 조건, 즉 출신고가 "우주고"인 학생들([C3], [C5], [C6])의 종합 점수를 5열(F)에서 찾은 후 찾은 종합 점수들(85, 90.5, 79)의 평균인 84.833...을 반환합니다. 84.833...을 ①에 대입하면 다음과 같습니다.
- =ROUND(84.833..., 1) : 84.833...을 소수점 이하 둘째 자리에서 반올림하여 소수점 이하 첫째 자리까지 표시한 84.80이 [D10] 셀에 입력됩니다.



전문가의 조언

- ROUND 함수는 숫자를 지정한 자리수로 반올림하여 표시하는 함수입니다. 자세한 내용은 36쪽을 참고하세요.
- 데이터베이스 함수는 데이터 목록과는 별도의 셀에 조건을 지정한 후 함수의 인수로 사용하는 것이 일반적이지만 위 문제처럼 데이터 목록 내의 값(C2:C3)을 함수의 조건 인수로 사용할 수도 있습니다.

024 가장 적게 팔린 과일의 금액 찾기 — DMIN

DMIN 함수는 데이터 목록에서 조건에 맞는 자료 중 가장 작은 값을 찾을 때 사용하는 함수입니다. 데이터 목록은 행(레코드)과 열(필드)로 이루어진 관련 데이터의 모임으로, 목록의 첫째 행에는 반드시 각 열의 제목(필드명)이 있어야 합니다. 그리고 DMIN 함수와 같은 데이터베이스 함수는 데이터 목록과는 별도의 셀에 조건을 지정한 후 함수의 인수로 사용하는 것이 일반적입니다.

형식 DMIN(범위, 열 번호, 조건) : 해당 '범위'에서 '조건'에 맞는 자료를 대상으로 지정된 '열 번호'에서 가장 작은 값을 찾습니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'DMIN.xsm' 파일을 열어 '기본' 시트에서 실습하세요.

DMIN 함수를 이용하여 품목별로 최소 판매 금액을 찾아 보겠습니다.

	A	B	C	D	E	F
1		과일 판매 현황				
2		날짜	품명	수량	단가	금액
3		1	사과	10	100	1,000
4		1	배	20	200	4,000
5		1	배	30	200	6,000
6		2	사과	20	100	2,000
7		2	배	20	200	4,000
8		2	감	30	300	9,000
9						
10			<품목별 최소 판매 금액>			
11			품명	품명	품명	
12			사과	배	감	
13			1,000	4,000	9,000	

- DMIN(B2:F8, 5, C11:C12) : [B2:F8] 영역에서 [C11:C12] 영역의 조건, 즉 품명이 "사과"인 과일의 금액을 5열(F)에서 찾은 후 그 금액들(1,000, 2,000) 중 가장 작은 금액인 1,000이 [C13] 셀에 입력됩니다.
- =DMIN(B2:F8, 5, D11:D12) : [B2:F8] 영역에서 [D11:D12] 영역의 조건, 즉 품명이 "배"인 과일의 금액을 5열(F)에서 찾은 후 그 금액들(4,000, 6,000, 4,000) 중 가장 작은 금액인 4,000이 [D13] 셀에 입력됩니다.

조건을 지정할 때는 반드시 데이터 목록의 필드명과 같은 필드명을 사용해야 합니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

구분이 "사탐"인 과목 중에서 표준편차의 최고점수와 최저점수의 차이값을 구하세요.

▶ [C11:C12] 영역에 조건을 입력하여 함수 적용

- 정답** • 먼저 [C11] 셀에 구분, [C12] 셀에 사탐을 입력하여 조건을 만든 후 [D12] 셀에 수식을 입력합니다.
- [D12] : =DMAX(B2:E9, 4, C11:C12) - DMIN(B2:E9, 4, C11:C12)

	A	B	C	D	E
1		선택과목 원점수 평균 및 표준편차			
2		구분	과목명	평균	표준편차
3		직탐	프로그래밍	21.19	10.56
4		사탐	한국지리	14.05	7.32
5		사탐	국사	23.34	12.33
6		사탐	윤리	23.22	8.31
7		과탐	생물	15.03	9.86
8		사탐	세계지리	15.31	8.97
9		직탐	공업입문	27.05	14.02
10					
11			구분	차이값	
12			사탐	5.01	

수식의 이해

=DMAX(B2:E9, 4, C11:C12) - DMIN(B2:E9, 4, C11:C12)

- ① DMAX(B2:E9, 4, C11:C12) : [B2:E9] 영역에서 [C11:C12] 영역의 조건, 즉 구분이 "사탐"인 학생의 표준편차를 4열(E)에서 찾은 후 찾은 표준편차들(7.32, 12.33, 8.31, 8.97) 중 가장 큰 12.33을 반환합니다.
- ② DMIN(B2:E9, 4, C11:C12) : [B2:E9] 영역에서 [C11:C12] 영역의 조건, 즉 구분이 "사탐"인 학생의 표준편차를 4열(E)에서 찾은 후 찾은 표준편차들(7.32, 12.33, 8.31, 8.97) 중 가장 작은 7.32를 반환합니다. 12.33과 7.32를 ①과 ②에 대입하면 다음과 같습니다.
- =12.33 - 7.32 : 12.33에서 7.32를 뺀 5.01이 [D12] 셀에 입력됩니다.



전문가의 조언

'열 번호'에 4 대신 필드명인 표준편차를 직접 입력하거나 셀 주소인 E2를 입력해도 됩니다. 필드명 중 '구분'은 첫 번째 필드이고, '표준편차'는 네 번째 필드입니다.

025 가장 많이 팔린 과일의 금액 찾기 — DMAX

DMAX 함수는 데이터 목록에서 조건에 맞는 자료 중 가장 큰 값을 찾을 때 사용하는 함수입니다. 데이터 목록은 행(레코드)과 열(필드)로 이루어진 관련 데이터의 모임으로, 목록의 첫째 행에는 반드시 각 열의 제목(필드명)이 있어야 합니다. 그리고 DMAX 함수와 같은 데이터베이스 함수는 데이터 목록과는 별도의 셀에 조건을 지정한 후 함수의 인수로 사용하는 것이 일반적입니다.

형식 DMAX(범위, 열 번호, 조건) : 해당 '범위'에서 '조건'에 맞는 자료를 대상으로 지정된 '열 번호'에서 가장 큰 값을 찾습니다.

준비하세요! '길벗컴퓨터2급통합\부록\함수사전' 폴더의 'DMAX.xsm' 파일을 열어 '기본' 시트에서 실행하세요.

DMAX 함수를 이용하여 품목별로 최대 판매 금액을 찾아보겠습니다.

	A	B	C	D	E	F
1		과일 판매 현황				
2		날짜	품명	수량	단가	금액
3		1	사과	10	100	1,000
4		1	배	20	200	4,000
5		1	배	30	200	6,000
6		2	사과	20	100	2,000
7		2	배	20	200	4,000
8		2	감	30	300	9,000
9						
10		<품목별 최대 판매 금액>				
11			품명	품명	품명	
12			사과	배	감	
13			2,000	6,000	9,000	

- ① =DMAX(B2:F8, 5, C11:C12) : [B2:F8] 영역에서 [C11:C12] 영역의 조건, 즉 품명이 "사과"인 과일의 금액을 5열(F)에서 찾은 후 그 금액들(1,000, 2,000) 중 가장 큰 금액인 2,000이 [C13] 셀에 입력됩니다.
- ② =DMAX(B2:F8, 5, D11:D12) : [B2:F8] 영역에서 [D11:D12] 영역의 조건, 즉 품명이 "배"인 과일의 금액을 5열(F)에서 찾은 후 그 금액들(4,000, 6,000, 4,000) 중 가장 큰 금액인 6,000이 [D13] 셀에 입력됩니다.

조건을 지정할 때는 반드시 데이터 목록의 필드명과 같은 필드명을 사용해야 합니다.

기출문제 따라잡기

'기출' 시트에서 실행하세요.

구분이 "사람"인 과목 중에서 표준편차의 최고점수와 최저점수의 차이값을 구하세요.

▶ [C11:C12] 영역에 조건을 입력하여 함수 적용

- 정답**
- 먼저 [C11] 셀에 구분, [C12] 셀에 사람을 입력하여 조건을 만든 후 [D12] 셀에 수식을 입력합니다.
 - [D12] : =DMAX(B2:E9, 4, C11:C12) - DMIN(B2:E9, 4, C11:C12)

	A	B	C	D	E
1		선택과목 원점수 평균 및 표준편차			
2		구분	과목명	평균	표준편차
3		직탐	프로그래밍	21.19	10.56
4		사탐	한국지리	14.05	7.32
5		사탐	국사	23.34	12.33
6		사탐	윤리	23.22	8.31
7		과탐	생물	15.03	9.86
8		사탐	세계지리	15.31	8.97
9		직탐	공업입문	27.05	14.02
10					
11			구분	차이값	
12			사람	5.01	

수식의 이해

=DMAX(B2:E9, 4, C11:C12) - DMIN(B2:E9, 4, C11:C12)

- ① DMAX(B2:E9, 4, C11:C12) : [B2:E9] 영역에서 [C11:C12] 영역의 조건, 즉 구분이 "사람"인 학생의 표준편차를 4열(E)에서 찾은 후 찾은 표준편차들(7.32, 12.33, 8.31, 8.97) 중 가장 큰 12.33을 반환합니다.
- ② DMIN(B2:E9, 4, C11:C12) : [B2:E9] 영역에서 [C11:C12] 영역의 조건, 즉 구분이 "사람"인 학생의 표준편차를 4열(E)에서 찾은 후 찾은 표준편차들(7.32, 12.33, 8.31, 8.97) 중 가장 작은 7.32를 반환합니다. 12.33과 7.32를 ①과 ②에 대입하면 다음과 같습니다.
- =12.33 - 7.32 : 12.33에서 7.32를 뺀 5.01이 [D12] 셀에 입력됩니다.



전문가의 조언

'열 번호'에 4 대신 필드명인 **표준편차**를 직접 입력하거나 셀 주소인 **E2**를 입력해도 됩니다. 필드명 중 '구분'은 첫 번째 필드이고, '표준편차'는 네 번째 필드입니다.

026 과일의 판매 수량 합계 계산하기 — DSUM

DSUM 함수는 데이터 목록에서 조건에 맞는 자료들의 합계를 계산할 때 사용하는 함수입니다. 데이터 목록은 행(레코드)과 열(필드)로 이루어진 관련 데이터의 모임으로 목록의 첫째 행에는 반드시 각 열의 제목(필드명)이 있어야 합니다. 그리고 DSUM 함수와 같은 데이터베이스 함수는 데이터 목록과는 별도의 셀에 조건을 지정한 후 함수의 인수로 사용하는 것이 일반적입니다.

형식 DSUM(범위, 열 번호, 조건) : 해당 '범위'에서 '조건'에 맞는 자료를 대상으로 지정된 '열 번호'에서 합계를 계산합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'DSUM.xlsm' 파일을 열어 '기본' 시트에서 실습하세요.

DSUM 함수를 이용하여 품목별로 판매 수량의 합계를 계산해 보겠습니다.

	A	B	C	D	E	F
1		과일 판매 현황				
2		날짜	품명	수량	단가	금액
3		1	사과	10	100	1,000
4		1	배	20	200	4,000
5		1	배	30	200	6,000
6		2	사과	20	100	2,000
7		2	배	20	200	4,000
8		2	감	30	300	9,000
9						
10		<품목별 판매 수량의 합계>				
11			품명	품명	품명	
12			사과	배	감	
13		①	30	70	30	②

① =DSUM(B2:F8, 3, C11:C12) : [B2:F8] 영역에서 [C11:C12] 영역의 조건, 즉 품명이 "사과"인 과일의 수량을 3열(D)에서 찾아 그 수량들(10, 20)의 합계인 30이 [C13] 셀에 입력됩니다.

② =DSUM(B2:F8, 3, E11:E12) : [B2:F8] 영역에서 [E11:E12] 영역의 조건, 즉 품명이 "감"인 과일의 수량을 3열(D)에서 찾아 그 수량(30)의 합계인 30이 [E13] 셀에 입력됩니다.

조건을 지정할 때는 반드시 데이터 목록의 필드명과 같은 필드명을 사용해야 합니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

부서명이 "총무부"인 직원들의 출장비 합계를 총무부합계에 표시하세요.

정답 [E10] : =DSUM(B2:E8, 4, C2:C3)

	A	B	C	D	E
1		부서별 출장비 지급 내역			
2		일자	부서명	신청자	출장비
3		03월 05일	총무부	김승찬	123,000
4		03월 10일	홍보부	박준영	58,000
5		04월 04일	총무부	전운덕	64,000
6		04월 11일	인사부	정동명	115,000
7		04월 13일	홍보부	조순철	44,000
8		04월 15일	총무부	황규태	134,000
9					
10				총무부합계	321,000

수식의 이해

=DSUM(B2:E8, 4, C2:C3)

[B2:E8] 영역에서 [C2:C3] 영역의 조건, 즉 부서명이 "총무부"인 직원들([C3], [C5], [C8])의 출장비를 4열(E)에서 찾아 그 출장비들(123,000, 64,000, 134,000)의 합계인 321,000이 [E10] 셀에 입력됩니다.



전문가의 조언

'열 번호'에 4 대신 필드명인 **출장비**를 직접 입력하거나 셀 주소인 **E2**를 입력해도 됩니다. 필드명 중 '일자'는 첫 번째 필드이고, '출장비'는 네 번째 필드입니다.

027 과일의 판매 건수 계산하기 — DCOUNTA

DCOUNTA 함수는 데이터 목록에서 조건에 맞는 자료 중 데이터가 있는 레코드의 개수를 계산할 때 사용하는 함수입니다. 데이터 목록은 행(레코드)과 열(필드)로 이루어진 관련 데이터의 모임으로 목록의 첫째 행에는 반드시 각 열의 제목(필드명)이 있어야 합니다. 그리고 DCOUNTA 함수와 같은 데이터베이스 함수는 데이터 목록과는 별도의 셀에 조건을 지정한 후 함수의 인수로 사용하는 것이 일반적입니다.

형식 DCOUNTA(범위, 열 번호, 조건) : 해당 '범위'에서 '조건'에 맞는 자료를 대상으로 지정된 '열 번호'에서 비어 있지 않은 셀의 개수를 계산합니다.

준비하세요! : 길벗컴활2급통합\부록\함수사전 폴더의 'DCOUNTA.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

DCOUNTA 함수를 이용하여 품목별로 판매 건수를 계산해 보겠습니다.

	A	B	C	D	E	F
1		과일 판매 현황				
2		날짜	품명	수량	단가	금액
3		1	사과	10	100	1,000
4		1	배	20	200	4,000
5		1	배	30	200	6,000
6		2	사과	10	100	1,000
7		2	배	20	200	4,000
8		2	감	10	300	3,000
9						
10		<품목별 주문 건수>				
11			품명	품명	품명	
12			사과	배	감	
13		1	2	3	1	2

① =DCOUNTA(B2:F8, 2, C11:C12) : [B2:F8] 영역에서 [C11:C12] 영역의 조건, 즉 품명이 "사과"인 과일을 2열(C)에서 찾은 후 그 개수인 2가 [C13] 셀에 입력됩니다.

② =DCOUNTA(B2:F8, 2, E11:E12) : [B2:F8] 영역에서 [E11:E12] 영역의 조건, 즉 품명이 "감"인 과일을 2열(C)에서 찾은 후 그 개수인 1이 [E13] 셀에 입력됩니다.

조건을 지정할 때는 반드시 데이터 목록의 필드명과 같은 필드명을 사용해야 합니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

상태가 "쾌적"인 일수를 구하여 표시하세요.

▶ 일수 뒤에 "일"을 표시 [표시 예 : 5일]

정답 [C15] : =DCOUNTA(B2:D12, 3, B14:B15) & "일"

	A	B	C	D
1		불쾌지수 현황		
2		날짜	불쾌지수	상태
3		07월 22일	84	심한불쾌
4		07월 23일	94	심한불쾌
5		07월 24일	83	심한불쾌
6		07월 25일	75	불쾌
7		07월 26일	70	쾌적
8		07월 27일	67	쾌적
9		07월 28일	69	쾌적
10		07월 29일	73	불쾌
11		07월 30일	79	불쾌
12		07월 31일	82	심한불쾌
13				
14		상태	일수	
15		쾌적	3일	

수식의 이해

=DCOUNTA(B2:D12, 3, B14:B15) & "일"

①

• ① DCOUNTA(B2:D12, 3, B14:B15) : [B2:D12] 영역에서 [B14:B15] 영역의 조건, 즉 상태가 "쾌적"인 날(D7, D8, D9)을 3열(D)에서 찾아 그 상태들의 개수인 3을 반환합니다. 3을 ①에 대입하면 다음과 같습니다.

• =3 & "일" : 문자열 결합 연산자(&)에 의해 3과 "일"이 합쳐진 "3일"이 [C15] 셀에 입력됩니다.

※ 이 문제에서 열 번호는 빈 셀이 없는 열을 지정하면 되므로 3 대신 1, 2 중 하나를 입력해도 됩니다.

028 과일의 판매 건수 계산하기 — DCOUNT

19,참시, 15,참시, 10,3, 10,1

DCOUNT 함수는 데이터 목록에서 조건에 맞는 자료 중 숫자가 있는 레코드의 개수를 계산할 때 사용하는 함수입니다. 데이터 목록은 행(레코드)과 열(필드)로 이루어진 관련 데이터의 모임으로, 목록의 첫째 행에는 반드시 각 열의 제목(필드명)이 있어야 합니다. 그리고 DCOUNT 함수와 같은 데이터베이스 함수는 데이터 목록과는 별도의 셀에 조건을 지정한 후 함수의 인수로 사용하는 것이 일반적입니다. 문자, 오류값 등이 들어 있는 셀의 개수도 세어야 할 때는 DCOUNTA 함수를 사용하세요.

형식 DCOUNT(범위, 열 번호, 조건) : 해당 '범위'에서 '조건'에 맞는 자료를 대상으로 지정된 '열 번호'에서 숫자가 있는 셀의 개수를 계산합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'DCOUNT.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

DCOUNT 함수를 이용하여 품목별로 판매 건수를 계산해 보겠습니다.

	A	B	C	D	E	F
1		과일 판매 현황				
2		날짜	품명	수량	단가	금액
3		1	사과	10	100	1,000
4		1	배	20	200	4,000
5		1	배	취소	취소	취소
6		2	사과	10	100	1,000
7		2	배	20	200	4,000
8		2	감	10	300	3,000
9						
10			<품목별 판매 건수>			
11			품명	품명	품명	
12			사과	배	감	
13		1	2	2	1	

① =DCOUNT(B2:F8, 5, C11:C12) : [B2:F8] 영역에서 [C11:C12] 영역의 조건, 즉 품명이 "사과"인 과일의 금액을 5열(F)에서 찾은 후 찾은 금액들(1,000, 1,000)의 개수인 2가 [C13] 셀에 입력됩니다.

② =DCOUNT(B2:F8, 5, D11:D12) : [B2:F8] 영역에서 [D11:D12] 영역의 조건, 즉 품명이 '배'인 과일의 금액을 5열(F)에서 찾은 후 찾은 금액들(4,000, 취소, 4,000) 중 숫자가 있는 셀의 개수인 2가 [D13] 셀에 입력됩니다.

조건을 지정할 때는 반드시 데이터 목록의 필드명과 같은 필드명을 사용해야 합니다.

기출문제 따라잡기

'기출' 시트에서 실행하세요.

희망부서가 "기획부"인 사원의 인원수를 구하세요.

▶ 인원수 뒤에 "명"을 표시 [표시 예 : 3명]

정답 [C17] : =DCOUNT(B2:D14, 3, B16:B17) & "명"

	A	B	C	D
1		신입 사원 현황		
2		사원번호	희망부서	입사시험 성적
3		A1204	영업부	98
4		A1205	총무부	76
5		A1206	기획부	86
6		A1207	총무부	80
7		A1208	총무부	65
8		A1209	기획부	100
9		A1210	기획부	98
10		A1211	영업부	85
11		A1212	기획부	100
12		A1213	총무부	90
13		A1214	기획부	85
14		A1215	영업부	75
15				
16		희망부서	인원수	
17		기획부	5명	

수식의 이해

=DCOUNT(B2:D14, 3, B16:B17) & "명"

①

• ① DCOUNT(B2:D14, 3, B16:B17) : [B2:D14] 영역에서 [B16:B17] 영역의 조건, 즉 희망부서가 "기획부"인 사원들([C5], [C8], [C9], [C11], [C13])의 입사시험 성적을 3열(D)에서 찾아 그 입사시험 성적의 개수인 5를 반환합니다. 5를 ①에 대입하면 다음과 같습니다.

• =5 & "명" : 문자열 결합 연산자(&)에 의해 5와 "명"이 합쳐진 "5명"이 [C17] 셀에 입력됩니다.

※ DCOUNT 함수는 조건을 만족하는 숫자의 개수를 구하는 함수이므로 개수를 구할 열 번호는 숫자가 입력된 3열을 지정해야 합니다.

029 1에 가까운 방향으로 자리 올림하기 — ROUNDUP

ROUNDUP 함수는 숫자를 지정한 자리수로 자리 올림하여 표시하는 함수입니다. 예를 들어 'ROUNDUP(35.6713, 2)'라면 35.6713을 소수 셋째 자리에서 자리 올림하여 소수 이하 둘째 자리까지 표시하므로 35.68이 됩니다.

형식 ROUNDUP(인수, 올림 자리수) : '인수'에 대하여 지정한 '자리수'로 올림합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'ROUNDUP.xlsm' 파일을 열어 '기본' 시트에서 실험하세요.

ROUNDUP 함수로 평균을 소수 첫째 자리에서 자리 올림하겠습니다.

	A	B	C	D	E	F	G
1		심사 결과					
2		이름	근무	능력	실적	평균	평가점수
3		박주형	72	78	80	76.6667	77
4		남영진	100	90	94	94.6667	95
5		강구라	100	90	93	94.3333	95
6		안미리	76	72	70	72.6667	73
7		조석구	75	72	70	72.3333	73

① =ROUNDUP(F3, 0) : [F3] 셀의 값 76.6667을 소수 이하 1자리에서 자리 올림하여 소수 이하 0자리까지 표시하므로 77이 [G3] 셀에 입력됩니다.

② =ROUNDUP(F5, 0) : [F5] 셀의 값 94.333을 소수 이하 1자리에서 자리 올림하여 소수 이하 0자리까지 표시하므로 95가 [G5] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실험하세요.

판매수량과 할인율표를 이용하여 할인액을 계산하세요.

- ▶ 할인액 = 판매금액 × 할인율
- ▶ 할인액은 십의 자리에서 올림하여 백의 자리까지 표시
[표시 예 : 34,530 → 34,600]

정답 [F3] : =ROUNDUP(E3 * HLOOKUP(D3, \$C\$10:\$E\$11, 2), -2)

	A	B	C	D	E	F
1		상공 문구 판매 현황				
2		제품명	판매가격	판매수량	판매금액	할인액
3		다이어리	2,550	55	140,250	4,300
4		수첩	12,350	65	802,750	24,100
5		명함꽃이	3,450	60	207,000	6,300
6		딱풀	765	100	76,500	4,600
7		붓	7,650	77	589,050	35,400
8						
9		<할인율표>				
10		판매수량	1	40	70	
11		할인율	0%	3%	6%	

수식의 이해

=ROUNDUP(E3 * HLOOKUP(D3, C10:E11, 2), -2)

- ① HLOOKUP(D3, C10:E11, 2) : [D3] 셀의 값이 55이므로 계산 순서는 다음과 같습니다.
- [C10:E11] 영역의 첫 번째 행에서 55를 넘지 않는 근사값을 찾습니다. D열에 있는 40을 찾습니다.
- 40이 있는 열에서 2행에 있는 값 3%를 찾아서 반환합니다.
- 3%를 ①에 대입하면 다음과 같습니다.
- =ROUNDUP(E3 * 3%, -2) : [E3] 셀의 값 140,250에 3%를 곱한 값 4,207.5를 십의 자리에서 자리 올림하여 백의 자리까지 표시한 4,300이 [F3] 셀에 입력됩니다.

ROUNDUP 함수의 올림 자리수

올림 자리수가 0보다 크면 숫자는 지정한 소수 이하 자리수로, 0이면 가장 가까운 정수로, 0보다 작으면 소수점 왼쪽에서 올림됩니다.

3	8	6	4	.	5	5	8	8
-3자리	-2자리	-1자리	0자리		1자리	2자리	3자리	4자리



전문가의 조언

HLOOKUP 함수는 범위로 정한 영역의 첫 번째 행에서 특정 기준값으로 자료를 찾고, 그 자료가 속한 열 중에서 필요한 값이 있는 행의 위치를 지정하여 값을 반환하는 함수입니다. 자세한 내용은 44쪽을 참고하세요.

030 조건에 맞는 품목의 합계 구하기 — SUMIF

SUMIF 함수는 많은 자료 중에서 조건에 맞는 데이터만 찾아서 합계를 구하는 함수입니다. 조건이 적용될 범위에서 조건에 맞는 데이터를 찾아 합계를 구할 범위 중 같은 행에 있는 값들의 합계를 계산합니다.

형식 SUMIF(조건이 적용될 범위, 조건, 합계를 구할 범위) : '조건이 적용될 범위'에서 '조건'에 맞는 셀을 찾아 '합계를 구할 범위' 중 같은 행에 있는 값들의 합계를 계산합니다.

준비하세요! '길벗컴활2급통합\부록\함수사전' 폴더의 'SUMIF.xlsm' 파일을 열어 '기본' 시트에서 실험하세요.

SUMIF 함수를 이용하여 품목별 판매 금액의 합계를 계산해 보겠습니다.

	A	B	C	D	E	F	G
1	판매현황				품목별 합계		
2							
3	품목	수량	단가	금액	품목	금액	
4	냉장고	6	250	1,500	컴퓨터	7,500	①
5	컴퓨터	8	300	2,400	캠코더	6,000	
6	냉장고	5	250	1,250	냉장고	2,750	②
7	캠코더	7	500	3,500			
8	컴퓨터	10	300	3,000			
9	캠코더	5	500	2,500			
10	컴퓨터	7	300	2,100			

① =SUMIF(A4:A10, "컴퓨터", D4:D10) : [A4:A10] 영역에서 "컴퓨터"가 입력된 셀을 찾아, [D4:D10] 영역의 같은 행에 있는 금액들(2,400, 3,000, 2,100)의 합계인 7,500이 [G4] 셀에 입력됩니다.

② =SUMIF(A4:A10, "냉장고", D4:D10) : [A4:A10] 영역에서 "냉장고"가 입력된 셀을 찾아, [D4:D10] 영역의 같은 행에 있는 금액들(1,500, 1,250)의 합계인 2,750이 [G6] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실험하세요.

직위가 "사원"인 실적의 합계와 직위가 "대리"인 실적의 합계 차이를 구하여 절대값으로 표시하세요.

정답 [D9] : =ABS(SUMIF(C3:C7, "사원", E3:E7) - SUMIF(C3:C7, "대리", E3:E7))

	A	B	C	D	E
1	사원별 실적 현황				
2	사원명	직위	목표액	실적	
3	서용석	대리	4,500	4,250	
4	신영균	사원	3,600	3,590	
5	장영식	대리	3,200	3,210	
6	김인수	사원	4,230	4,200	
7	전지현	대리	3,780	3,890	
8					
9		차이값	3,560		

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=ABS(SUMIF(C3:C7, "사원", E3:E7) - SUMIF(C3:C7, "대리", E3:E7))

①

②

- ① SUMIF(C3:C7, "사원", E3:E7) : [C3:C7] 영역에서 "사원"이 입력된 셀([C4], [C6])을 찾은 후 [E3:E7] 영역의 같은 행([E4], [E6])에 있는 실적들의 값(3,590, 4,200)을 더한 7,790을 반환합니다.
- ② SUMIF(C3:C7, "대리", E3:E7) : [C3:C7] 영역에서 "대리"가 입력된 셀([C3], [C5], [C7])을 찾은 후 [E3:E7] 영역의 같은 행([E3], [E5], [E7])에 있는 실적들의 값(4,250, 3,210, 3,890)을 더한 11,350을 반환합니다. 7,790과 11,350을 각각 ①, ②에 대입하면 다음과 같습니다.
- =ABS(7790 - 11350) : 7,790 - 11,350의 결과인 -3,560의 절대값 3,560이 [D9] 셀에 입력됩니다.

031 절대값 계산하기 — ABS

24.상시, 23.상시, 22.상시, 21.상시, 19.상시, 09.1, 06.1, 05.2, 05.1, 03.3

ABS 함수는 숫자의 부호 없이 숫자의 크기만을 나타내는 절대값을 계산할 때 사용합니다. 예를 들어, -5의 절대값은 5이고 +5의 절대값도 5입니다. 절대값은 특정 값과의 차이를 알아내서 비교할 때 많이 사용됩니다.

형식 ABS(인수) : '인수'로 주어진 숫자의 절대값을 계산합니다.

준비하세요 : '길벗컴활2급통합\부록\함수사전' 폴더의 'ABS.xsm' 파일을 열어 '기본' 시트에서 실행하세요.

ABS 함수를 이용하여 키에 대한 평균과의 차이를 절대값으로 표시해 보겠습니다.

	A	B	C	D	E	F
1						
2						
3		성명	성별	키	평균과의 차이	절대값
4		고신애	여	168	-2	2
5		김애자	여	165	-5	5
6		김형민	남	180	10	10
7		박수동	남	175	5	5
8		서수일	남	179	9	9
9		소성환	남	180	10	10
10		신길자	여	155	-15	15
11		이승혁	남	168	-2	2
12		조보람	여	162	-8	8
13		평균		170		

① =ABS(E4) : [E4] 셀의 숫자에 절대값을 취한 값인 2가 [F4] 셀에 입력됩니다.

② =ABS(E7) : [E7] 셀의 숫자에 절대값을 취한 값인 5가 [F7] 셀에 입력됩니다.

③ =ABS(E10) : [E10] 셀의 숫자에 절대값을 취한 값인 15가 [F10] 셀에 입력됩니다.

④ =ABS(E12) : [E12] 셀의 숫자에 절대값을 취한 값인 8이 [F12] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실행하세요.

직위가 "사원"인 실적의 합계와 직위가 "대리"인 실적의 합계 차이를 구하여 절대값으로 표시하세요.

정답 [D9] : =ABS(SUMIF(C3:C7, "사원", E3:E7) - SUMIF(C3:C7, "대리", E3:E7))

	A	B	C	D	E
1					
2		사원명	직위	목표액	실적
3		서용석	대리	4,500	4,250
4		신영균	사원	3,600	3,590
5		장영식	대리	3,200	3,210
6		김인수	사원	4,230	4,200
7		전지현	대리	3,780	3,890
8					
9			차이값	3,560	

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=ABS(SUMIF(C3:C7, "사원", E3:E7) - SUMIF(C3:C7, "대리", E3:E7))

- ① SUMIF(C3:C7, "사원", E3:E7) : [C3:C7] 영역에서 "사원"이 입력된 셀([C4], [C6])을 찾은 후 [E3:E7] 영역의 같은 행([E4], [E6])에 있는 실적들의 값(3,590, 4,200)을 더한 7,790을 반환합니다.
- ② SUMIF(C3:C7, "대리", E3:E7) : [C3:C7] 영역에서 "대리"가 입력된 셀([C3], [C5], [C7])을 찾은 후 [E3:E7] 영역의 같은 행([E3], [E5], [E7])에 있는 실적들의 값(4,250, 3,210, 3,890)을 더한 11,350을 반환합니다. 7,790과 11,350을 각각 ①, ②에 대입하면 다음과 같습니다.
- =ABS(7790 - 11350) : 7,790 - 11,350의 결과인 -3,560의 절대값 3,560이 [D9] 셀에 입력됩니다.

032 나머지 계산하기 — MOD

23.상시, 22.상시, 21.상시, 19.상시, 18.2, 13.1, 10.2, 09.3, 09.2, 06.4, 05.3, 03.2

MOD 함수는 숫자를 나누기한 후 나머지를 구하는 함수입니다. 예를 들면, $5/2$ 는 몫이 2이고 나머지가 1인데 MOD는 나머지 1을 구하는 함수입니다. 결과는 나누는 수의 부호를 갖습니다. MOD는 홀짝을 판별하거나 어떤 수의 배수 여부를 판별할 때 많이 사용됩니다.

형식 MOD(인수1, 인수2) : '인수1'을 '인수2'로 나눈 나머지를 구합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'MOD.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

	A	B	C
1	나머지 계산하기		
2			
3	숫자	나누는 수	나머지
4	67	5	2
5	4	2	0
6	29	6	5
7	-7	3	2

- ① =MOD(A4, B4) : 67을 5로 나누기한 후 나머지 값인 2가 [C4] 셀에 입력됩니다.
- ② =MOD(A6, B6) : 29를 6으로 나누기한 후 나머지 값인 5가 [C6] 셀에 입력됩니다.
- ③ =MOD(A7, B7) : -7을 3으로 나누기한 후 나머지 값인 2가 [C7] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실행하세요.

주민등록번호의 8번째 숫자가 홀수면 "남", 짝수면 "여"로 표시하세요.

정답 [E3] : =IF(MOD(MID(D3, 8, 1), 2)=1, "남", "여")

	A	B	C	D	E
1	스포츠센터 회원 현황				
2	회원명	가입년도	주민등록번호	성별	
3	윤다현	2018년	800621-1*****	남	
4	오지윤	2016년	920101-2*****	여	
5	송주명	2019년	000317-4*****	여	
6	양명준	2018년	960725-1*****	남	
7	안성수	2017년	950226-1*****	남	

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=IF(MOD(MID(D3, 8, 1), 2)=1, "남", "여")

① ② ③

조건(①)이 참(TRUE)이면 ②를 표시하고, 거짓(FALSE)이면 ③을 표시합니다.

• ① MOD(MID(D3, 8, 1), 2)=1

④

— ④ MID(D3, 8, 1) : "800621-1*****"의 8번째 위치부터 1 글자만 추출한 "1"을 반환합니다. 1을 ④에 대입하면 다음과 같습니다.

— MOD(1, 2)=1 : 1을 2로 나누기한 후 나머지 값인 1을 반환합니다. 1은 1과 같으므로 참을 반환합니다. 참을 ①에 대입하면 다음과 같습니다.

• =IF(참, "남", "여") : 조건이 참이므로 [E3] 셀에 "남"이 입력됩니다.

033 반올림하기 — ROUND

ROUND 함수는 숫자를 지정한 자리수로 반올림하여 표시하는 함수입니다. 가령 'ROUND(35.6768, 2)'라면 35.6768을 소수 이하 셋째 자리에서 반올림하여 소수 이하 둘째 자리까지 표시하므로 35.68이 됩니다.

형식 ROUND(인수, 반올림 자리수) : '인수'에 대하여 지정한 '자리수'로 반올림합니다.

준비하세요! : '갈벗컴활2급통합\부록\함수사전' 폴더의 'ROUND.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

	A	B	C
1	반올림		
2			
3	숫자	자리수	결과
4	78325.67429	3	78325.674
5	78325.67429	2	78325.67
6	78325.67429	1	78325.7
7	78325.67429	0	78326
8	78325.67429	-1	78330
9	78325.67429	-2	78300
10	78325.67429	-3	78000

① =ROUND(A4, B4) : [A4] 셀의 값 78325.67429를 소수 이하 넷째 자리에서 반올림하여 소수 이하 셋째 자리까지 표시하므로 78325.674가 [C4] 셀에 입력됩니다.

② =ROUND(A9, B9) : [A9] 셀의 값 78325.67429를 십의 자리에서 반올림하여 백의 자리까지 표시하므로 78300이 [C9] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실행하세요.

"출신교"가 "우주교"인 학생들의 종합 점수 평균을 구하여 표시하세요.

▶ 우주교 종합 평균은 소수점 이하 둘째 자리에서 반올림하여 첫째 자리까지 표시 [표시 예 : 64.66 → 64.7]

정답 [C10] : =ROUND(DAVERAGE(B2:F7, 5, C2:C3), 1)

	A	B	C	D	E	F
1		경시대회 성적				
2		성명	출신교	필기	실기	종합
3		고영인	우주교	75	95	85
4		성수영	대한교	77	89	83
5		구정철	우주교	88	93	90.5
6		박대철	우주교	91	67	79
7		전소영	상공교	85	56	70.5
8						
9			우주교 종합 평균			
10			84.8			

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=ROUND(DAVERAGE(B2:F7, 5, C2:C3), 1)

①

• ① DAVERAGE(B2:F7, 5, C2:C3) : [B2:F7] 영역에서 [C2:C3] 영역의 조건, 즉 출신교가 "우주교"인 학생들([C3], [C5], [C6])의 종합 점수를 5열(F)에서 찾은 후 찾은 종합 점수들(85, 90.5, 79)의 평균인 84.833...을 반환합니다. 84.833...을 ①에 대입하면 다음과 같습니다.

• =ROUND(84.833..., 1) : 84.833...을 소수점 이하 둘째 자리에서 반올림하여 소수점 이하 첫째 자리까지 표시한 84.8이 [D10] 셀에 입력됩니다.

ROUND 함수의 반올림 자리수

반올림 자리수가 0보다 크면 숫자는 지정한 소수 이하 자리수로, 0이면 가장 가까운 정수로, 0보다 작으면 소수점 왼쪽에서 반올림됩니다.

3	8	6	4	.	5	5	8	8
-3자리	-2자리	-1자리	0자리		1자리	2자리	3자리	4자리



전문가의 조언

DAVERAGE 함수는 데이터 목록에서 조건에 맞는 자료만의 평균을 계산할 때 사용하는 함수입니다. 자세한 내용은 26쪽을 참고하세요.

034 실수를 정수로 변경하기 — INT

25.상시, 24.상시, 19.상시, 13.1, 09.3, 06.1

INT 함수는 실수의 소수점 이하를 제거하여 정수로 변환시킬 때 사용하는 함수입니다. INT는 인수로 주어진 실수보다 크지 않은 정수로 변환시킵니다. 예를 들어, 'INT(5.1)'은 5를 반환하고, 'INT(-5.1)'은 -6을 반환합니다.

형식 INT(인수) : '인수'로 주어진 실수를 정수로 변환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'INT.xsm' 파일을 열어 '기본' 시트에서 실험하세요.

	A	B
1	정수로 변환하기	
2		
3	실수	정수
4	4.5	4
5	4.99	4
6	125.12	125
7	-6.1	-7
8	-85.9	-86

- ① =INT(A4) : [A4] 셀의 값 '4.5' 보다 크지 않은 정수 4가 [B4] 셀에 입력됩니다.
- ② =INT(A6) : [A6] 셀의 값 '125.12' 보다 크지 않은 정수 125가 [B6] 셀에 입력됩니다.
- ③ =INT(A8) : [A8] 셀의 값 '-85.9' 보다 크지 않은 정수 -86이 [B8] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실험하세요.

영어, 경제, 경영의 평균이 10이면 "D", 2면 "C", 30이면 "B", 4면 "A"를 표시하세요.

정답 [F3] : =CHOOSE(INT(AVERAGE(C3:E3)), "D", "C", "B", "A")

	A	B	C	D	E	F
1	1학기 성적표					
2	이름	영어	경제	경영	학점	
3	김두영	2.3	2.7	2.4	C	
4	이항기	3.6	3.5	3.7	B	
5	한민관	4.2	4.1	4.2	A	
6	최시연	1.6	1.4	1.3	D	
7	유효진	3.2	3.3	3.5	B	

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=CHOOSE(INT(AVERAGE(C3:E3)), "D", "C", "B", "A")

①

②

- ① AVERAGE(C3:E3) : [C3:E3] 영역의 평균 2.467...을 반환합니다. 2.467...을 ①에 대입하면 다음과 같습니다.
- ② INT(2.467...) : 2.467...보다 크지 않은 정수 2를 반환합니다. 2를 ②에 대입하면 다음과 같습니다.
- =CHOOSE(2, "D", "C", "B", "A") : 두 번째에 있는 값 "C"가 [F3] 셀에 입력됩니다.

035 합계 구하기 — SUM

25.상시, 24.상시, 23.상시, 22.상시, 21.상시, 19.상시, 13.1, 10.3

SUM 함수는 인수로 주어진 숫자들의 합계를 계산하는 함수로, 인수는 255개까지 지정할 수 있습니다. 인수는 숫자이거나 숫자가 포함된 이름, 배열 또는 셀 주소이어야 합니다.

형식 SUM(인수1, 인수2, ...): '인수'로 주어진 숫자들의 합계를 계산합니다.

준비하세요! : '길벗컴퓨터2급통합\부록\함수사전' 폴더의 'SUM.xlsm' 파일을 열어 '기본' 시트에서 실험하세요.

	A	B	C	D
1	합계계산			
2				
3	숫자1	숫자2	숫자3	합계
4	6	7	8	21
5	20	30	40	90
6	-	30	26	56
7	8		10	18

① =SUM(A4:C4) : [A4:C4] 영역의 합계인 21이 [D4] 셀에 입력됩니다.

② =SUM(A6:C6) : [A6:C6] 영역의 합계인 56이 [D6] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실험하세요.

불량률과 공정개수를 이용하여 평균불량률을 구하세요.

- ▶ 평균불량률 = 불량률합계 / 0이 아닌 공정개수
- ▶ 평균불량률은 반올림 없이 정수로 표시

정답 [C11] : =TRUNC(SUM(B3:B9) / COUNTIF(C3:C9, "<>0"))

	A	B	C
1	지역별 불량률 현황		
2	지역	불량률	공정개수
3	서울	10	0
4	제주	8	1
5	청주	11	3
6	광주	9	2
7	부산	9	0
8	대구	10	2
9	인천	12	0
10			
11	평균불량률		17
12			

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=TRUNC(SUM(B3:B9) / COUNTIF(C3:C9, "<>0"))

- ① SUM(B3:B9) : [B3:B9] 영역의 합계인 69를 반환합니다.
- ② COUNTIF(C3:C9, "<>0") : [C3:C9] 영역에서 0이 아닌 셀([C4], [C5], [C6], [C8])의 개수 4를 반환합니다.
- TRUNC(69/4) : 69를 4로 나누기한 값 17.25를 반올림 없이 소수점 이하를 제거한 17이 [C11] 셀에 입력됩니다.



전문가의 조언

- TRUNC 함수는 인수에 대하여 지정된 자릿수 미만의 수치를 버리는 함수입니다. 자세한 내용은 39쪽을 참고하세요.
- COUNTIF 함수는 지정된 범위에서 조건에 맞는 셀의 개수를 반환하는 함수입니다. 자세한 내용은 61쪽을 참고하세요.

036 소수 이하 잘라내기 — TRUNC

TRUNC 함수는 숫자에서 지정한 자릿수 미만의 수치를 버릴 때 사용하는 함수입니다. 예를 들어, 'TRUNC(56.789, 1)'는 56.7을 반환합니다. 자릿수를 지정하지 않으면 자릿수가 0이 되어 INT와 유사한 기능을 합니다. TRUNC는 숫자의 소수 부분을 버리고, INT는 해당 숫자보다 크기 않은 정수로 변환합니다. INT와 TRUNC는 음수를 사용할 때만 다른 결과를 얻게 됩니다.

형식 TRUNC(인수, 자릿수) : '인수'에 대하여 지정한 '자릿수' 미만의 수치를 버립니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'TRUNC.xlsm' 파일을 열어 '기본' 시트에서 실험하세요.

	A	B	C
1	필요없는 수치 제거하기		
2			
3	숫자	자릿수	결과
4	78325.67429	3	78325.674
5	78325.67429	2	78325.67
6	78325.67429	1	78325.6
7	78325.67429	0	78325
8	78325.67429	-1	78320
9	78325.67429	-2	78300
10	78325.67429	-3	78000

- ① =TRUNC(A4, B4) : [A4] 셀의 값 78325.67429에서 소수점 이하 셋째 자리 미만의 숫자를 버린 값 78325.674가 [C4] 셀에 입력됩니다.
- ② =TRUNC(A8, B8) : [A8] 셀의 값 78325.67429에서 십의 자리 미만의 숫자를 버린 값 78320이 [C8] 셀에 입력됩니다.

TRUNC 함수의 자릿수

자릿수가 0보다 크면 숫자는 지정한 소수 이하 자릿수로, 0이면 소수 이하를 모두 제거하고, 0보다 작으면 소수점 왼쪽에서 제거합니다.

3	8	6	4	.	5	5	8	8
-3자리	-2자리	-1자리	0자리	1자리	2자리	3자리	4자리	

기출문제 따라잡기

'기출' 시트에서 실험하세요.

각 학생들의 중간, 수행, 기말 점수에 대한 평균을 구하여 표시하세요.

▶ 반올림 없이 소수 이하 첫째 자리까지 표시

[표시 예 : 94.37 → 94.3]

정답 [F3] : =TRUNC(AVERAGE(C3:E3), 1)

	A	B	C	D	E	F
1	1학기 국어 성적					
2	성명	중간	수행	기말	평균	
3	김정훈	78.45	45.78	87.23	70.4	
4	오석현	88.79	87.34	90.45	88.8	
5	이영선	92.45	80.23	78.23	83.6	
6	임현재	88.45	77.54	98.56	88.1	
7	남정왕	88.66	89.12	89.54	89.1	

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안에서서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=TRUNC(AVERAGE(C3:E3), 1)

①

- ① AVERAGE(C3:E3) : [C3:E3] 영역의 평균 70.486...을 반환합니다. 70.486...을 ①에 대입하면 다음과 같습니다.
- =TRUNC(70.486..., 1) : 70.486...을 반올림 없이 소수점 이하 첫째 자리까지 표시한 70.4가 [F3] 셀에 표시됩니다.



전문가의 조언

AVERAGE 함수는 인수로 주어진 숫자들의 평균을 계산하는 함수입니다. 자세한 설명은 64쪽을 참고하세요.

037 로또 번호 예상하기 1 — RAND

RAND 함수는 0보다 크거나 같고 1보다 작은 난수를 구하는 함수입니다. 즉, 이 함수를 사용하면 0.78640392와 같은 숫자가 무작위로 추출됩니다. RAND 함수를 사용해서 난수를 구하면 워크시트가 계산될 때마다 새로운 난수가 구해지므로 자동으로 시트의 데이터가 변경됩니다.

형식 RAND() : RAND 함수는 인수 없이 사용됩니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'RAND.xlsm' 파일을 열어 '기본' 시트에서 실험하세요.

	A	B	C
1	로또 번호 만들기		
2			
3	시작번호	끝번호	로또번호
4	1	43	37
5	1	43	30
6	1	43	17
7	1	43	43
8	1	43	4
9	1	43	31

① =INT(RAND() * (B4-A4+1)+A4) : [A4] 셀과 [B4] 셀의 값 사이의 난수 값 중 하나(정수)가 [C4] 셀에 입력됩니다.

- 두 수 A(작은 수), B(큰 수) 사이의 난수를 계산하려면 다음의 수식을 이용해야 합니다.
=RAND * (큰 수 - 작은 수 + 1) + 작은 수
- 결과는 소수점이 있는 실수로 나오며, 정수로 만들기 위해서 만든 수식을 INT() 함수의 인수로 지정해야 합니다.
- RAND 함수는 무작위 수를 구하는 함수로서 계산할 때마다 다른 난수를 표시합니다. 그러므로 위의 그림에 있는 로또 번호 결과가 같은 수는 없겠지요? 무작위 수를 계산한 후 재 계산 기능이 있는 [F9]를 눌러보면, 누를 때마다 결과가 변하는 것을 알 수 있습니다.

038 로또 번호 예상하기 2 — RANDBETWEEN

RANDBETWEEN 함수는 인수로 지정한 두 수 사이의 정수를 출력하는 함수입니다. 예를 들어, RANDBETWEEN(5, 10)을 입력하면 5~10 사이의 임의의 수가 반환됩니다. RANDBETWEEN 함수를 사용해서 임의의 수를 구하면 워크시트가 계산될 때마다 새로운 난수가 구해져 자동으로 시트의 데이터가 변경됩니다.

형식 RANDBETWEEN(인수1, 인수2) : '인수1'과 '인수2'의 사이에 있는 임의의 정수를 출력합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'RANDBETWEEN.xlsm' 파일을 열어 '기본' 시트에서 실험하세요.

	A	B	C
1	로또 번호 만들기		
2			
3	시작번호	끝번호	로또번호
4	1	43	32
5	1	43	5
6	1	43	16
7	1	43	1
8	1	43	40
9	1	43	26

① =RANDBETWEEN(A4, B4) : [A4] 셀의 값 1과 [B4] 셀의 값 43 사이의 임의의 정수가 [C4] 셀에 입력됩니다.

- RANDBETWEEN 함수를 사용하여 난수를 만들고 셀이 계산될 때마다 만들어진 난수가 변경되지 않도록 하려면 셀에 =RANDBETWEEN(1, 43)을 입력한 후 [Enter]를 누르기 전에 [F9]를 눌러 수식을 결과 값으로 변경하면 됩니다.
- RANDBETWEEN 함수로 구한 결과는 계산할 때마다 다른 수로 표시되므로 위 그림의 로또 번호와 다를 수 있습니다.

039 자리 내림하기 — ROUNDDOWN

ROUNDDOWN 함수는 숫자를 지정한 자릿수로 자리 내림하여 표시하는 함수입니다. 예를 들어, ROUNDDOWN(35.6768, 2)라면 35.6768을 소수 셋째 자리에서 자리 내림하여 소수 이하 둘째 자리까지 표시하므로 35.67이 됩니다.

형식 ROUNDDOWN(인수, 내림 자릿수) : '인수'에 대하여 지정한 '자릿수'로 내림합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'ROUNDDOWN.xlsm' 파일을 열어 '기본' 시트에서 실험하세요.

	A	B	C
1	자리내림		
2			
3	숫자	자릿수	결과
4	78325.67429	3	78325.674
5	78325.67429	1	78325.6
6	78325.67429	0	78325
7	78325.67429	-1	78320
8	78325.67429	-3	78000

- ① =ROUNDDOWN(A4, B4) : [A4] 셀의 값 78325.67429를 소수 이하 넷째 자리에서 자리 내림하여 소수 이하 셋째 자리까지 표시하므로 [C4] 셀에 78325.674가 입력됩니다.
- ② =ROUNDDOWN(A7, B7) : [A7] 셀의 값 78325.67429를 일의 자리에서 자리 내림하여 십의 자리까지 표시하므로 [C7] 셀에 78320이 입력됩니다.

ROUNDDOWN 함수의 내림 자릿수

내림 자릿수가 0보다 크면 숫자는 지정한 소수 이하 자릿수로, 0이면 가장 가까운 정수로, 0보다 작으면 소수점 왼쪽에서 내림됩니다.

3	8	6	4	.	5	5	8	8
-3자리	-2자리	-1자리	0자리		1자리	2자리	3자리	4자리

기출문제 따라잡기

'기출' 시트에서 실험하세요.

대리점이 "금성물산"인 자료들의 판매금액 합계를 구하여 표시하세요.

- ▶ 판매금액 합계는 십의 자리에서 내림하여 백의 자리까지 표시
[표시 예 : 1,550 → 1,500]

정답 [D12] : =ROUNDDOWN(SUMIF(A3:A10, "금성물산", D3:D10), -2)

	A	B	C	D
1	대리점 판매 현황			
2	대리점	단가	주문수량	판매금액
3	금성물산	1,673	40	66,920
4	우주상사	1,506	58	87,348
5	신안공업	1,126	91	102,466
6	삼성상사	2,953	99	292,347
7	금성물산	1,423	54	76,842
8	신안공업	1,338	40	53,520
9	우주상사	1,996	46	91,816
10	삼성상사	2,167	91	197,197
11				
12	금성물산 판매금액 합계			143,700

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=ROUNDDOWN(SUMIF(A3:A10, "금성물산", D3:D10), -2)

①

- ① SUMIF(A2:A10, "금성물산", D3:D10) : [A3:A10] 영역에서 "금성물산"이 입력된 셀([A3], [A7])을 찾은 후 [D3:D10] 영역의 같은 행([D3], [D7])에 있는 판매금액들의 값(66,920, 76,842)을 더한 143,762를 반환합니다. 143,762를 ①에 대입하면 다음과 같습니다.
- =ROUNDDOWN(143762, -2) : 143,762를 십의 자리에서 내림하여 백의 자리까지 표시한 143,700이 [D12] 셀에 입력됩니다.



전문가의 조언

SUMIF 함수는 조건이 적용될 범위에서 조건에 맞는 셀을 찾아 합계를 구할 범위 중 같은 행에 있는 값들의 합계를 구하는 함수입니다. 자세한 내용은 33쪽을 참고하세요.

040 부서별 직급별 기본급의 합계 계산하기 — SUMIFS

SUMIFS 함수는 여러 개의 조건에 맞는 자료의 합계를 구하는 함수입니다. 예를 들면 부서가 기획부이고, 급수가 1급이고, 남자인 사원들의 기본급 합계를 구할 수 있습니다. 조건은 최대 127개까지 지정할 수 있습니다.

형식 SUMIFS(합계를 구할 범위, 첫 번째 조건이 적용될 범위, 첫 번째 조건, 두 번째 조건이 적용될 범위, 두 번째 조건, ...) : 여러 개의 '조건이 적용될 범위'에서 여러 개의 '조건'에 맞는 셀을 찾아 '합계를 구할 범위' 중 같은 행에 있는 값들의 합계를 계산합니다.

준비하세요! : '갈벗컴활2급통합\부록\함수사전' 폴더의 'SUMIFS.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

SUMIFS 함수를 이용하여 부서별 직급별 기본급의 합계를 계산해 보겠습니다.

	A	B	C	D	E	F	G	H
1	기본급 지급 현황				부서별 직급별 기본급의 합계			
2								
3	성명	부서	직급	기본급	부서	1급	2급	
4	이승연	판매부	1급	1,450,000	판매부	2,900,000	2,650,000	
5	김경수	기획부	2급	1,350,000	기획부	2,650,000	2,550,000	
6	① 학봉	판매부	2급	1,350,000				
7	지순녀	기획부	2급	1,200,000				②
8	김자연	판매부	1급	1,450,000				
9	박원래	기획부	1급	1,450,000				
10	최지은	기획부	1급	1,200,000				
11	강유라	판매부	2급	1,300,000				

① =SUMIFS(D4:D11, B4:B11, "판매부", C4:C11, "1급") : [B4:B11] 영역에서 "판매부"가 입력된 셀들을 찾고, [C4:C11] 영역에서 같은 행들에 있는 "1급"이 입력된 셀들을 찾아 [D4:D11] 영역의 같은 행들에 있는 기본급(1,450,000, 1,450,000)의 합계인 2,900,000이 [G4] 셀에 입력됩니다.

② =SUMIFS(D4:D11, B4:B11, "기획부", C4:C11, "2급") : [B4:B11] 영역에서 "기획부"가 입력된 셀들을 찾고, [C4:C11] 영역에서 같은 행들에 있는 "2급"이 입력된 셀들을 찾아 [D4:D11] 영역의 같은 행들에 있는 기본급(1,350,000, 1,200,000)의 합계인 2,550,000이 [H5] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

장르가 "코미디"이면서 관람등급이 "전체"인 영화의 예매총액 합계를 계산하세요.

▶ 합계 뒤에 "만원"을 표시 [표시 예 : 123 → 123만원]

정답 [E10] : =SUMIFS(E3:E9, B3:B9, "코미디", C3:C9, "전체") & "만원"

	A	B	C	D	E
1	영화예매 현황				
2	영화명	장르	관람등급	예매량	예매총액
3	대구행	액션	12세 이상	5,336	4,802
4	구름남아서	드라마	15세 이상	4,423	3,981
5	꿈의선물	코미디	전체	3,686	3,317
6	변호사	드라마	15세 이상	6,987	6,288
7	범죄의마왕	코미디	전체	5,454	4,909
8	해적	액션	15세 이상	6,589	5,930
9	수상한아이	코미디	12세 이상	4,359	3,923
10	코미디-전체 예매총액 합계				8226만원

수식의 이해

=SUMIFS(E3:E9, B3:B9, "코미디", C3:C9, "전체") & "만원"

①

- ① SUMIFS(E3:E9, B3:B9, "코미디", C3:C9, "전체") : [B3:B9] 영역에서 "코미디"가 입력된 셀들([B5], [B7], [B9])을 찾고, [C3:C9] 영역에서 "코미디"가 입력된 셀과 같은 행들에 있는 셀 중 "전체"가 입력된 셀들([C5], [C7])을 찾고, [E3:E9] 영역에서 "코미디"와 "전체"가 입력된 셀과 같은 행인 [E5], [E7]에 있는 예매총액(3,317, 4,909)의 합계인 8,226을 반환합니다. 8,226을 ①에 대입하면 다음과 같습니다.
- =8226 & "만원" : 문자열 결합 연산자(&)에 의해 8,226과 "만원"이 합쳐진 "8,226만원"이 [E10] 셀에 입력됩니다.

041 거듭 제곱 계산하기 — POWER

25.상시, 23.상시, 22.상시, 21.상시

POWER 함수는 숫자의 거듭 제곱을 계산하는 함수입니다. 즉 POWER(5, 3)은 $5 \times 5 \times 5$ 를 의미합니다. 숫자는 실수를 지정해도 됩니다. POWER 함수는 연산자 ^와 같은 기능을 하므로 POWER(5, 3)을 5^3 과 같이 입력해도 결과는 같습니다.

형식 POWER(인수, 제곱값) : '인수'를 '제곱값'만큼 거듭 곱한 값을 계산합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'POWER.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C
1	거듭 제곱 구하기		
2			
3	숫자	지수	결과
4	2	1	2
5	3	2	9
6	5	3	125
7	7	4	2401

① =POWER(A4, B4) : [B4] 셀의 지수(1)만큼 [A4] 셀의 숫자(2)를 한 번 곱한 값 (2×1)인 2가 [C4] 셀에 입력됩니다.

② =POWER(A6, B6) : [B6] 셀의 지수(3)만큼 [A6] 셀의 숫자(5)를 세 번 곱한 값 ($5 \times 5 \times 5$)인 125가 [C6] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

체질량지수가 20 미만이면 "저체중", 20 이상 25 미만이면 "정상", 25 이상이면 "비만"으로 체질량지수에 표시하세요.

▶ 체질량지수(BMI) = 체중 ÷ (신장)²

정답 [D4] : =IF(C4/POWER(B4, 2)<20, "저체중", IF(C4/POWER(B4, 2) <25, "정상", "비만"))

	A	B	C	D
1	신체검사결과			
2				
3	성명	신장(m)	체중(kg)	체질량지수
4	한혜진	1.61	59	정상
5	황진주	1.57	62	비만
6	강재준	1.82	91	비만
7	김소희	1.65	54	저체중
8	안정환	1.77	69	정상
9	허지석	1.68	60	정상

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=IF(C4 / POWER(B4, 2)<20, "저체중", IF(C4/POWER(B4, 2)<25, "정상", "비만"))

① ② ③

조건 ①이 참(TRUE)이면 ②를 표시하고, 거짓(FALSE)이면 ③을 표시합니다.

- ① C4/POWER(B4, 2) < 20
 - POWER(B4, 2) : [B4] 셀의 값 1.61을 두 번 곱한 값 2.59를 반환합니다.
 - [C4] 셀의 값 59를 2.59로 나눈 값 22.76은 20보다 크므로 거짓을 반환합니다. 거짓을 ①에 대입하면 다음과 같습니다.
- =IF(거짓, "저체중", IF(C4/POWER(B4, 2)<25, "정상", "비만")) : 조건 ①이 거짓이므로 ③을 실행합니다.

① ② ③

- ③ IF(C4/POWER(B4, 2)<25, "정상", "비만")

④ ⑤ ⑥

조건 ④이 참이면 ⑤를 표시하고, 거짓이면 ⑥을 표시합니다.

- ④ C4/POWER(B4, 2) < 25
 - POWER(B4, 2) : [B4] 셀의 값 1.61을 두 번 곱한 값 2.59를 반환합니다.
 - [C4] 셀의 값 59를 2.59로 나눈 값 22.76은 25보다 작으므로 참을 반환합니다. 참을 ④에 대입하면 다음과 같습니다.
- IF(참, "정상", "비만") : 조건이 참이므로 [C4] 셀에 "정상"이 입력됩니다.

042 직위별 상여금 계산하기 — HLOOKUP

HLOOKUP 함수는 범위로 정한 영역의 첫 번째 행에서 특정 기준값으로 자료를 찾고, 그 자료가 속한 열 중에서 필요한 값이 있는 행의 위치를 지정하여 값을 반환하는 함수입니다. 만약 범위로 정한 영역의 맨 왼쪽 열에서 기준값을 찾고자 할 경우에는 VLOOKUP 함수를 이용하세요.

형식 HLOOKUP(찾을값, 범위, 행 번호, 옵션) : '범위'의 첫 번째 행에서 '찾을값'과 같은 데이터를 찾은 후 '찾을값'이 있는 열에서 지정된 '행 번호' 위치에 있는 데이터를 반환합니다.

준비하세요! : '갈벗컴월2급통합\북록\함수사전' 폴더의 'HLOOKUP.xlsm' 파일을 열어 '기본' 시트에서 실습하세요.

HLOOKUP 함수를 이용하여 상여금을 계산해 보겠습니다.

	A	B	C	D	E
1		상여금 지급			
2		이름	직위	상여금	
3		김신락	사원	500,000	①
4		홍길동	부장	700,000	
5		김천만	과장	650,000	
6		신선해	대리	600,000	
7		박막해	사원	500,000	
8					
9		<직위별 지급 기준>			
10		직위	사원	대리	과장
11		상여금	500,000	600,000	700,000
12		조과수당	3,500	4,500	5,500

① =HLOOKUP(C3, \$B\$10:\$E\$12, 2, FALSE) : [B10:E12] 영역의 첫 번째 행에서 [C3] 셀의 값, 즉 "사원"과 정확히 일치하는 값을 찾습니다. 찾은 값 "사원"이 있는 1열에서 행 번호로 지정된 두 번째 행의 값(상여금) 500,000이 [D3] 셀에 입력됩니다. "사원"과 정확히 일치하는 값을 찾는 이유는 옵션에 논리값 'FALSE'가 지정되었기 때문입니다.

옵션에 지정할 논리값

- TRUE 또는 생략 : 근사값을 찾습니다. 즉 정확하게 일치하는 값이 없으면 찾을값보다 작은 값 중에서 가장 근접한 값을 찾습니다.
- FALSE : 정확하게 일치하는 값을 찾으며, 정확히 일치하는 값이 없으면 '#N/A' 오류값이 반환됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

판매수량과 할인율표를 이용하여 할인액을 계산하세요.

- ▶ 할인액 = 판매금액 × 할인율
- ▶ 할인액은 십의 자리에서 올림하여 백의 자리까지 표시
[표시 예 : 34,530 → 34,600]

정답 [F3] : =ROUNDUP(E3 * HLOOKUP(D3, \$C\$10:\$E\$11, 2), -2)

	A	B	C	D	E	F
1		상공 문구 판매 현황				
2		제품명	판매가격	판매수량	판매금액	할인액
3		다이어리	2,550	55	140,250	4,300
4		수첩	12,350	65	802,750	24,100
5		명함꽃이	3,450	60	207,000	6,300
6		딱풀	765	100	76,500	4,600
7		붓	7,650	77	589,050	35,400
8						
9		<할인율표>				
10		판매수량	1	40	70	
11		할인율	0%	3%	6%	

수식의 이해

=ROUNDUP(E3 * HLOOKUP(D3, C10:E11, 2), -2)

①

- ① HLOOKUP(D3, C10:E11, 2) : [D3] 셀의 값이 55이므로 계산 순서는 다음과 같습니다.
 - ➡ [C10:E11] 영역의 첫 번째 행에서 55를 넘지 않는 근사값을 찾습니다. D열에 있는 40을 찾습니다.
 - ➡ 40이 있는 열에서 2행에 있는 값 3%를 찾아서 반환합니다.
 - ➡ 3%를 ①에 대입하면 다음과 같습니다.
- =ROUNDUP(E3 * 3%, -2) : [E3] 셀의 값 140,250에 3%를 곱한 값 4,207.5를 십의 자리에서 자리 올림하여 백의 자리까지 표시한 4,300이 [F3] 셀에 입력됩니다.

	A	B	C	D	E
9		<할인율표>			
10		판매수량	1	40	70
11		할인율	0%	3%	6%



전문가의 조언

ROUNDUP 함수는 숫자를 지정한 자릿수로 자리 올림하여 표시하는 함수입니다. 자세한 내용은 32쪽을 참고하세요.

043 직위별 상여급 계산하기 — VLOOKUP

VLOOKUP 함수는 범위로 정한 영역의 맨 왼쪽 열에서 특정 기준값으로 자료를 찾고, 그 자료가 속한 행 중에서 필요한 값이 있는 열의 위치를 지정하여 값을 반환하는 함수입니다. 만약 범위로 정한 영역의 첫 번째 행에서 기준값을 찾고자 할 경우에는 HLOOKUP 함수를 이용하세요.

형식 VLOOKUP(찾을값, 범위, 열 번호, 옵션) : '범위'의 맨 왼쪽 열에서 '찾을 값'과 같은 값을 찾은 후 '찾을 값'이 있는 행에서 지정된 '열 번호' 위치에 있는 데이터를 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'VLOOKUP.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

VLOOKUP 함수를 이용하여 상여급을 계산해 보겠습니다.

	A	B	C	D
1		상여급 지급		
2		이름	직위	상여급
3		김신락	과장	650,000
4		홍길동	부장	700,000
5		김천만	과장	650,000
6		신선해	대리	600,000
7		막막해	사원	500,000
8				
9		<직위별 지급 기준>		
10		직위	상여급	조과수당
11		사원	500,000	3,500
12		대리	600,000	4,500
13		과장	650,000	5,500
14		부장	700,000	6,000

① =VLOOKUP(C3, \$B\$11:\$D\$14, 2, FALSE) : [B11:D14] 영역의 맨 왼쪽 열에서 [C3] 셀의 값, 즉 "과장"과 정확히 일치하는 값을 찾습니다. 찾을 값 "과장"이 있는 3행에서 열 번호로 지정된 두 번째 열의 값(상여급) 650,000이 [D3] 셀에 입력됩니다. "과장"과 정확히 일치하는 값을 찾는 이유는 옵션에 논리값 'FALSE'가 지정되었기 때문입니다.

옵션에 지정할 논리값

- TRUE 또는 생략 : 근사값을 찾습니다. 즉 정확하게 일치하는 값이 없으면 찾을값 보다 작은 값 중에서 가장 근접한 값을 찾습니다.
- FALSE : 정확하게 일치하는 값을 찾으며, 정확히 일치하는 값이 없으면 '#N/A' 오류값이 반환됩니다.

기출문제 따라잡기

'기출' 시트에서 실행하세요.

취득총점의 순위를 구한 후 수상목록에서 순위에 해당하는 수상명을 찾아 표시하세요.

▶ 순위는 취득총점이 가장 높은 것이 1위

정답 [D3] : =VLOOKUP(RANK.EQ(C3, \$C\$3:\$C\$11), \$F\$4:\$G\$9, 2)

	A	B	C	D	E	F	G
1		모형 항공기 제작 경진 대회					
2		팀명	취득총점	수상명		<수상목록>	
3		비상	158	등상		순위	수상명
4		하늘	195	금상		1	대상
5		창공	157	등상		2	금상
6		송골매	154	장려상		3	은상
7		보라매	120			4	등상
8		이상	198	대상		6	장려상
9		불새	145	장려상		9	
10		마하	135	장려상			
11		최원영	160	은상			

수식의 이해

=VLOOKUP(RANK.EQ(C3, C3:C11), F4:G9, 2)

①

• ① RANK.EQ(C3, C3:C11) : [C3:C11] 영역에서 [C3] 셀의 값 158의 내림차순을 기준으로 계산한 순위 4를 반환합니다. 4를 ①에 대입하면 다음과 같습니다.

• =VLOOKUP(4, F4:G9, 2)

- ① [F4:G9] 영역의 맨 왼쪽 열에서 4를 넘지 않는 가장 근접한 값 7행에 있는 4를 찾습니다.
- ② 4가 있는 행에서 2열에 있는 값 "등상"을 찾아서 [D3] 셀에 입력합니다.

※ 4를 넘지 않는 가장 근접한 값을 찾는 것은 '옵션'에 지정된 논리값이 생략되었기 때문입니다. 생략하거나 'TRUE'인 경우에는 찾을 값보다 크기 않으면서 가장 근접한 값을 찾습니다.

	F	G
1		
2	<수상목록>	
3	순위	수상명
4	1	대상
5	2	금상
6	3	은상
7	4	등상
8	6	장려상
9	9	



전문가의 조언

RANK.EQ 함수는 지정된 범위 안에서 인수의 순위를 구하는 함수입니다. 자세한 내용은 62쪽을 참고하세요.

044 요일 번호로 요일 표시하기 — CHOOSE

CHOOSE 함수는 인덱스 번호를 이용하여 특정 번째에 있는 값을 반환하는 함수입니다. CHOOSE 함수를 사용하여 29개까지의 값 중에서 한 개의 값을 선택할 수 있습니다.

형식 CHOOSE(인수, 첫 번째, 두 번째, ... n번째) : '인수'가 1이면 '첫 번째'를, '인수'가 2이면 '두 번째'를, ... '인수'가 n이면 'n번째'를 반환합니다.

준비하세요! : '길벗컴퓨터2급통합\부록\함수사전' 폴더의 'CHOOSE.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

CHOOSE 함수로 날짜에 해당하는 요일번호를 이용해서 요일을 표시해 보겠습니다.

	A	B	C	D
1		요일 계산하기		
2		날짜	요일번호	요일
3		2025-01-19	1	일
4		2025-01-20	2	월
5		2025-01-21	3	화
6		2025-01-22	4	수
7		2025-01-23	5	목
8		2025-01-24	6	금
9		2025-01-25	7	토

① =CHOOSE(C3, "일", "월", "화", "수", "목", "금", "토") : [C3] 셀의 값이 1이므로 첫 번째에 있는 "일"이 [D3] 셀에 입력됩니다.

② =CHOOSE(C9, "일", "월", "화", "수", "목", "금", "토") : [C9] 셀의 값이 7이므로 일곱 번째에 있는 "토"가 [D9] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실행하세요.

접수번호의 첫 자리가 1이면 "사회", 20이면 "과학", 30이면 "직업"으로 선택과목에 표시하세요.

정답 [E3] : =CHOOSE(LEFT(B3, 1), "사회", "과학", "직업")

	A	B	C	D	E
1		학력평가 접수현황			
2		접수번호	학교명	성명	선택과목
3		10238	상공고	전현우	사회
4		20472	대한고	이승완	과학
5		10635	서울고	조인호	사회
6		30987	고려고	정명우	직업
7		20345	조선고	김중협	과학

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=CHOOSE(LEFT(B3, 1), "사회", "과학", "직업")

①

• ① LEFT(B3, 1) : [B3] 셀에 입력된 값의 왼쪽에서 1글자를 추출한 "1"이 반환됩니다. "1"을 ①에 대입하면 다음과 같습니다.

• =CHOOSE("1", "사회", "과학", "직업") : 첫 번째에 있는 값 "사회"가 [E3] 셀에 입력됩니다.

※ 'LEFT(B3, 1)'의 결과는 문자 "1"이 반환되지만 CHOOSE 함수의 '인수'는 숫자가 오는 자리이므로 문자화된 숫자가 숫자로 인식되어 계산됩니다.



전문가의 조언

LEFT 함수는 텍스트 문자열의 첫 문자부터 원하는 문자수만큼의 문자를 추출하는 함수입니다. 자세한 설명은 52쪽을 참고하세요.

045 직급별, 호봉별 급여 기준액 계산하기 — INDEX

INDEX 함수는 셀 범위나 데이터 배열에서 행 번호와 열 번호가 교차하는 곳에 있는 값을 반환하는 함수입니다. 행 번호와 열 번호가 셀 범위를 벗어나면 '#REF!' 오류값이 반환됩니다.

형식 INDEX(범위, 행 번호, 열 번호) : 지정된 '범위'에서 '행 번호'와 '열 번호'에 위치한 데이터를 입력합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'INDEX.xlsm' 파일을 열어 '기본' 시트에서 실습하세요.

INDEX 함수로 급여 기준표에서 직급별, 호봉별 급여를 찾아 급여 지급표를 완성해 보겠습니다.

	A	B	C	D	E
1		급여 지급			
2		성명	직급	호봉	급여
3		김신락	2	2	620,000
4		홍길동	3	3	750,000
5		김천만	4	3	850,000
6		신선해	1	3	550,000
7		막막해	5	2	920,000
8					
9		급여 기준표			
10		구분	1호봉	2호봉	3호봉
11		1급	500,000	520,000	550,000
12		2급	600,000	620,000	650,000
13		3급	700,000	720,000	750,000
14		4급	800,000	820,000	850,000
15		5급	900,000	920,000	950,000

① =INDEX(\$C\$11:\$E\$15, C3, D3) : [C3] 셀의 값 2를 행 번호로 지정하고, [D3] 셀의 값 2를 열 번호로 지정하였으므로, [C11:E15] 영역에서 2행 2열의 값 620,000을 찾아 [E3] 셀에 입력합니다.

② =INDEX(\$C\$11:\$E\$15, C7, D7) : [C7] 셀의 값 5를 행 번호로 지정하고, [D7] 셀의 값 2를 열 번호로 지정하였으므로, [C11:E15] 영역에서 5행 2열의 값 920,000을 찾아 [E7] 셀에 입력합니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

[B3:E6] 영역을 참조하여 출발지(서울)에서 도착지(수원)까지의 택배요금을 표시하세요.

▶ 출발지와 도착지의 구분은 ()안의 두 자리 숫자를 이용

정답 [E8] : =INDEX(B3:E6, MID(C8, 5, 1), MID(D8, 5, 1))

	A	B	C	D	E
1		수도권 택배 요금표			
2		서울(01)	인천(02)	수원(03)	안양(04)
3	서울(01)	5,000	10,000	15,000	14,000
4	인천(02)	10,000	5,000	18,000	17,000
5	수원(03)	15,000	18,000	5,000	8,000
6	안양(04)	14,000	17,000	8,000	5,000
7			출발코드	도착코드	요금
8			서울(01)	수원(03)	15,000

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=INDEX(B3:E6, MID(C8, 5, 1), MID(D8, 5, 1))

①

②

- ① MID(C8, 5, 1) : '서울(01)(C8)'의 5번째 위치부터 1글자만 추출한 "1"을 반환합니다.
- ② MID(D8, 5, 1) : '수원(03)(D8)'의 5번째 위치부터 1글자만 추출한 "3"을 반환합니다.
- "1", "3"을 ①, ②에 대입하면 다음과 같습니다.
- =INDEX(B3:E6, "1", "3") : [B3:E6] 영역의 1행 3열의 값 15,000을 찾아 [E8] 셀에 입력합니다.



전문가의 조언

MID 함수는 지정된 문자수에 따라 텍스트 문자열의 특정 위치에서 원하는 수만큼의 문자를 추출하는 함수입니다. 자세한 설명은 51쪽을 참고하세요.

046 자료가 기록된 위치 찾기 — MATCH

25.상시, 24.상시, 23.상시, 22.상시, 21.상시, 18.1

MATCH 함수는 지정된 범위에서 기준값과 같은 데이터를 찾아 범위 내에서의 상대적인 위치를 반환하는 함수입니다. 항목의 위치 대신 항목의 값이 필요한 경우는 LOOKUP 함수를 사용합니다.

형식 MATCH(찾을값, 범위, 옵션) : '범위'에서 '찾을값'과 같은 데이터를 찾아 '옵션'을 적용하여 그 위치를 일련번호로 표시합니다.

준비하세요! : '길벗컴퓨터2급통합\부록\함수사전' 폴더의 'MATCH.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

MATCH 함수를 이용하여 실험 결과 영역에서 점수의 위치를 찾아보겠습니다.

A	B	C
1	실험 결과의 위치 찾기	
2	점수	상대위치
3	59	1
4	80	4
5		
6	실험 결과	
7	56	
8	60	
9	70	
10	80	
11	90	

- ① =MATCH(B3, \$B\$7:\$B\$11, 1) : [B7:B11] 영역에서 [B3] 셀의 값 59를 찾는데, 옵션이 1이므로 59보다 작거나 같은 값 중 가장 근접한 값을 찾습니다. 59보다 작은 값 중 가장 근접한 값 56을 찾아 [B7:B11]에서의 위치인 1이 [C3] 셀에 입력됩니다.

MATCH 함수의 옵션

- -1 : 찾을값보다 크거나 같은 값 중 가장 작은 값을 찾습니다. 범위는 반드시 내림차순으로 정렬되어 있어야 합니다.
- 0 : 찾을값과 첫 번째로 정확하게 일치하는 값을 찾습니다. 범위는 정렬되어 있지 않아도 됩니다.
- 1 : 찾을값보다 작거나 같은 값 중에서 가장 큰 값을 찾습니다. 범위는 반드시 오름차순으로 정렬되어 있어야 합니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

판매량과 가격표를 이용하여 판매총액을 계산하세요.

- ▶ 판매총액 = 판매량 × 할인가
- ▶ 할인가는 의류코드와 <가격표>를 이용하여 산출

정답 [D3] : =C3 * INDEX(\$B\$12:\$D\$13, 2, MATCH(A3, \$B\$11:\$D\$11, 0))

	A	B	C	D
1	의류 판매 현황			
2	의류코드	사이즈	판매량	판매총액
3	mk-101	M	31	790,500
4	mk-101	L	29	739,500
5	mk-102	M	35	966,875
6	mk-102	L	24	663,000
7	mk-103	M	28	826,560
8	mk-103	L	33	974,160
9				
10	<가격표>			
11	의류코드	mk-101	mk-102	mk-103
12	판매가	30,000	32,500	36,000
13	할인가	25,500	27,625	29,520

수식의 이해

=C3 * INDEX(\$B\$12:\$D\$13, 2, MATCH(A3, \$B\$11:\$D\$11, 0))

- ① MATCH(A3, \$B\$11:\$D\$11, 0) : 옵션이 0이므로 [B11:D11] 영역에서 [A3] 셀의 문자열 "mk-101"과 정확히 일치하는 문자열을 찾습니다. "mk-101"을 찾은 후 [B11:D11] 영역에서 "mk-101"의 상대적 위치 1을 반환합니다. 1을 ①에 대입하면 다음과 같습니다.
- ② INDEX(\$B\$12:\$D\$13, 2, 1) : [B12:D13] 영역의 2행 1열의 값 25,500을 찾습니다. 25,500을 ②에 대입하면 다음과 같습니다.
- C3 * 25,500 : [C3] 셀의 값 31에 25,500을 곱한 값 790,500이 [D3] 셀에 입력됩니다.



전문가의 조언

- MATCH 함수는 범위에서 일치하는 값이 아니라 일치하는 값의 위치를 반환합니다.
- MATCH는 텍스트의 대·소문자를 구분하지 않습니다.

047 현재 셀의 열 번호 알아내기 — COLUMN

COLUMN 함수는 셀 주소에 대한 열 번호를 알아내는 함수로, 행의 위치에 관계없이 1열에서부터 계산한 해당 열에 대한 번호를 반환합니다. 예를 들어 [F2] 셀은 F가 처음부터 여섯 번째 열에 해당하므로 6을 반환합니다. 인수를 생략하면 현재 셀의 열 번호를 반환합니다.

형식 COLUMN(범위) : 지정된 '범위'의 열 번호를 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'COLUMN.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

	A	B	C
1		열 번호 알아내기	
2		주소	열 번호
3		현재 셀의 열 번호	3
4		D2 셀의 열 번호	4

①

① =COLUMN() : 현재 셀의 열 번호인 3이 [C3] 셀에 입력됩니다.

②

② =COLUMN(D2) : [D2] 셀의 열 번호인 4가 [C4] 셀에 입력됩니다.

048 셀 범위에 포함된 열의 개수 알아내기 — COLUMNS

COLUMNS 함수는 지정된 범위 안에 들어 있는 열의 개수를 알아내는 함수로, 행의 개수에 관계 없이 열의 개수만을 반환합니다. 예를 들어 'COLUMNS(B1:D100)'은 B, C, D 세 열을 포함하므로 3을 반환합니다.

형식 COLUMNS(범위) : 지정된 '범위' 안에 들어 있는 열의 개수를 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'COLUMNS.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

	A	B	C
1		열의 수 계산	
2		셀 범위	열의 개수
3		A1:F2	6
4		B10:H1000	7

①

① =COLUMNS(A1:F2) : [A1:F2] 영역의 열의 개수인 6이 [C3] 셀에 입력됩니다.

②

② =COLUMNS(B10:H1000) : [B10:H1000] 영역의 열의 개수인 7이 [C4] 셀에 입력됩니다.

049 현재 셀의 행 번호 알아내기 — ROW

ROW 함수는 셀 주소에 대한 행 번호를 알아내는 함수로, 열의 위치에 관계없이 1행에서부터 계산한 해당 행에 대한 번호를 반환합니다. 예를 들어 [F2] 셀은 2가 처음부터 두 번째 행에 해당하므로 2를 반환합니다. 인수를 생략하면 현재 셀의 행 번호를 반환합니다.

형식 ROW(범위) : 지정된 '범위'의 행 번호를 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'ROW.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

A	B	C
1	행 번호 알아내기	
2	주소	행 번호
3	현재 셀의 행 번호	3
4	D2 셀의 행 번호	2

① =ROW() : 현재 셀의 행 번호인 3이 [C3] 셀에 입력됩니다.

② =ROW(D2) : [D2] 셀의 행 번호인 2가 [C4] 셀에 입력됩니다.

050 셀 범위에 포함된 행의 개수 알아내기 — ROWS

ROWS 함수는 지정된 범위 안에 들어 있는 행의 개수를 알아내는 함수로, 열의 개수에 관계 없이 행의 개수만을 반환합니다. 예를 들어 'ROWS(B2:P3)'은 2와 3, 두 행을 포함하므로 2를 반환합니다.

형식 ROWS(범위) : 지정된 '범위' 안에 들어 있는 행의 개수를 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'ROWS.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

A	B	C
1	행의 수 계산	
2	셀 범위	행의 개수
3	A1:F2	2
4	B10:H20	11

① =ROWS(A1:F2) : [A1:F2] 영역의 행의 개수인 2가 [C3] 셀에 입력됩니다.

② =ROWS(B10:H20) : [B10:H20] 영역의 행의 개수인 11이 [C4] 셀에 입력됩니다.

051 시작 위치부터 지정한 수만큼 추출하기 — MID

MID 함수는 텍스트 문자열의 특정 위치에서 원하는 수만큼의 문자를 추출합니다. 예를 들어 MID("길벗출판사", 2, 3)은 "벗출판"을 추출합니다.

형식 MID(텍스트, 시작위치, 개수) : '텍스트'의 '시작위치'부터 지정한 '개수'만큼 추출합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'MID.xlsm' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C	D
1	중간에서 문자열 추출하기			
2				
3	문자열	시작위치	문자수	결과
4	KOREA	3	2	RE
5	KOREA	2	4	OREA
6	KOREA	-1	3	#VALUE!
7	KOREA	2	-1	#VALUE!
8	대한민국	2	2	한민
9	12345678	4	3	456

- 1 =MID(A4, B4, C4) : [B4] 셀의 값을 시작위치로 지정하고, [C4] 셀의 값을 추출할 문자수로 지정하였으므로, [A4] 셀의 값 "KOREA"의 3번째 위치부터 2글자를 추출한 "RE"가 [D4] 셀에 입력됩니다.
- 2 =MID(A6, B6, C6) : 시작위치가 0보다 작으므로 오류값(#VALUE!)이 [D6] 셀에 입력됩니다.
- 3 =MID(A9, B9, C9) : [B9] 셀의 값을 시작위치로 지정하고, [C9] 셀의 값을 추출할 문자수로 지정하였으므로, [A9] 셀의 값 "12345678"의 4번째 위치부터 3글자를 추출한 "456"이 [D9] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

주민등록번호를 이용하여 생년월일을 표시하세요.

▶ 표시 예 : 1988-10-12 → 1988년 10월 12일

정답 [D3] : =DATE(MID(C3, 1, 2), MID(C3, 3, 2), MID(C3, 5, 2))

	A	B	C	D
1	동호회 회원 명단			
2	성명	주민등록번호	생년월일	
3	이정우	881012-1*****	1988-10-12	
4	남국현	861123-1*****	1986-11-23	
5	연제식	970203-1*****	1997-02-03	
6	송달호	960420-1*****	1996-04-20	
7	원정균	890804-2*****	1989-08-04	

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=DATE(MID(C3, 1, 2), MID(C3, 3, 2), MID(C3, 5, 2))

① ② ③

- ① MID(C3, 1, 2) : "881012-1*****"의 1 번째 위치부터 2 글자만 추출한 "88"을 반환합니다.
- ② MID(C3, 3, 2) : "881012-1*****"의 3 번째 위치부터 2 글자만 추출한 "10"을 반환합니다.
- ③ MID(C3, 5, 2) : "881012-1*****"의 5 번째 위치부터 2 글자만 추출한 "12"를 반환합니다.
- "88", "10", "12"를 ①, ②, ③에 대입하면 다음과 같습니다.
- =DATE("88", "10", "12") : "88"을 연도로, "10"을 월로, "12"를 일로 적용한 날짜 "1988-10-12"가 [D3] 셀에 입력됩니다.



전문가의 조언

DATE 함수는 1900년 1월 1일을 기준으로 하여 특정 날짜에 대한 일련번호를 반환하는 함수로 결과는 날짜 서식으로 표시됩니다. 자세한 설명은 6쪽을 참고하세요.

052 왼쪽에서 지정한 수만큼 추출하기 — LEFT

LEFT 함수는 텍스트 문자열의 첫 문자부터 원하는 문자수만큼의 문자를 추출합니다. 예를 들어 LEFT("길벗출판사", 2)는 "길벗"을 추출합니다.

형식 LEFT(텍스트, 개수) : '텍스트'의 왼쪽부터 지정한 '개수'만큼 추출합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'LEFT.xlsm' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C
1	왼쪽에 있는 문자 추출하기		
2			
3	문자	문자수	결과
4	KOREA	2	KO
5	KOREA	-1	#VALUE!
6	KOREA	0	
7	KOREA	6	KOREA
8	대한민국	2	대한
9	1234	2	12

- ① =LEFT(A4, B4) : [B4] 셀의 값을 추출할 문자수로 지정하였으므로, [A4] 셀의 값 "KOREA"의 왼쪽 첫 글자부터 2만큼 추출한 "KO"가 [C4] 셀에 입력됩니다.
- ② =LEFT(A5, B5) : 추출할 문자수가 0보다 작으므로 오류값(#VALUE!)이 [C5] 셀에 입력됩니다.
- ③ =LEFT(A8, B8) : [B8] 셀의 값을 추출할 문자수로 지정하였으므로, [A8] 셀의 값 "대한민국"의 왼쪽 첫 글자부터 2만큼 추출한 "대한"이 [C8] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

접수번호의 첫 자리가 10이면 "사회", 20이면 "과학", 30이면 "직업"으로 선택과목에 표시하세요.

정답 [E3] : =CHOOSE(LEFT(B3, 1), "사회", "과학", "직업")

	A	B	C	D	E
1	학력평가 접수현황				
2	접수번호	학교명	성명	선택과목	
3	10238	상공고	전현우	사회	
4	20472	대한고	이승완	과학	
5	10635	서울고	조인호	사회	
6	30987	고려고	정명우	직업	
7	20345	조선고	김중협	과학	

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=CHOOSE(LEFT(B3, 1), "사회", "과학", "직업")

①

• ① LEFT(B3, 1) : [B3] 셀에 입력된 값의 왼쪽에서 1글자를 추출한 "1"이 반환됩니다. "1"을 ①에 대입하면 다음과 같습니다.

• =CHOOSE("1", "사회", "과학", "직업") : 첫 번째에 있는 값 "사회"가 [E3] 셀에 입력됩니다.

※ 'LEFT(B3, 1)'의 결과는 문자 "1"이 반환되지만 CHOOSE() 함수의 첫 번째 인수는 숫자가 오는 자리이므로 문자화된 숫자가 숫자로 인식되어 계산됩니다.



전문가의 조언

CHOOSE 함수는 인덱스 번호를 이용하여 특정 번째에 있는 값을 반환하는 함수입니다. 자세한 내용은 46쪽을 참고하세요.

053 오른쪽에서 지정한 수만큼 추출하기 — RIGHT

RIGHT 함수는 텍스트 문자열의 끝(오른쪽) 문자부터 원하는 문자 수만큼의 문자를 추출합니다. 예를 들어 RIGHT("길벗출판사", 3)은 "출판사"를 추출합니다.

형식 RIGHT(텍스트, 개수) : '텍스트'의 오른쪽부터 지정한 '개수'만큼 추출합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'RIGHT.xlsm' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C
1	오른쪽에 있는 문자 추출하기		
2			
3	문자	문자수	결과
4	KOREA	2	EA
5	KOREA	-1	#VALUE!
6	KOREA	0	
7	KOREA	6	KOREA
8	대한민국	2	민국
9	1234	2	34

- ① =RIGHT(A4, B4) : [B4] 셀의 값을 추출할 문자수로 지정하였으므로, [A4] 셀의 값 "KOREA"의 오른쪽 끝 글자부터 2만큼 추출한 "EA"가 [C4] 셀에 입력됩니다.
- ② =RIGHT(A5, B5) : 추출할 문자수가 0보다 작으므로 오류값(#VALUE!)이 [C5] 셀에 입력됩니다.
- ③ =RIGHT(A8, B8) : [B8] 셀의 값을 추출할 문자수로 지정하였으므로, [A8] 셀의 값 "대한민국"의 오른쪽 끝 글자부터 2만큼 추출한 "민국"이 [C8] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

사원코드의 오른쪽 끝 문자가 10이면 "영업부", 20이면 "인사부", 30이면 "총무부", 40이면 "기획부"로 소속부서를 표시하세요.

정답 [E3] : =CHOOSE(RIGHT(B3, 1), "영업부", "인사부", "총무부", "기획부")

	A	B	C	D	E
1		상공전자 사원 현황			
2		사원코드	성명	근무년수	소속부서
3		H203-1	이지원	12	영업부
4		K102-2	나오미	13	인사부
5		B333-3	권경애	8	총무부
6		D104-2	강수영	15	인사부
7		F405-3	나우선	19	총무부

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=CHOOSE(RIGHT(B3, 1), "영업부", "인사부", "총무부", "기획부")

①

• ① RIGHT(B3, 1) : [B3] 셀에 입력된 값의 오른쪽 끝의 1글자를 추출한 "1"이 반환됩니다. "1"을 ①에 대입하면 다음과 같습니다.

• =CHOOSE("1", "영업부", "인사부", "총무부", "기획부") : 첫 번째에 있는 값 "영업부"가 [E3] 셀에 입력됩니다.

※ 'RIGHT(B3, 1)'의 결과는 문자 "1"이 반환되지만 CHOOSE 함수의 '인수'는 숫자가 오는 자리이므로 문자화된 숫자가 숫자로 인식되어 계산됩니다.



전문가의 조언

CHOOSE 함수는 인덱스 번호를 이용하여 특정 번째에 있는 값을 반환하는 함수입니다. 자세한 내용은 46쪽을 참고하세요.

054 대·소문자 알맞게 변환하기 — PROPER

21.공개, 20.상시, 19.상시, 18.2, 17.상시, 05.3, 05.2

PROPER 함수는 문자열에 포함된 대·소문자를 적절하게 변환하는 함수로, 영어 단어의 첫 번째 문자와 영문자가 아닌 문자 다음에 오는 영문자를 대문자로 변환하고, 나머지 문자들은 소문자로 변환합니다. 예를 들어 PROPER("24TH SEOUL OLYMPIC")은 "24Th Seoul Olympic"이 됩니다.

형식 PROPER(텍스트) : '텍스트'의 첫 번째 문자만 대문자로 변환합니다.

준비하세요! : '길벗컴퓨터2급통합\부록\함수사전' 폴더의 'PROPER.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B
1	대소문자를 적절하게 변환하기	
2		
3	문자열	결과
4	24TH SEOUL OLYMPIC	24Th Seoul Olympic
5	2-cent's worth	2-Cent'S Worth
6	this is a title	This Is A Title
7	길벗 publisher	길벗 Publisher
8	king kong	King Kong
9	a-123b	A-123B

① =PROPER(A4) : 첫 번째 영어 단어인 "SEOUL"의 "S"와 영문자가 아닌 문자 다음에 오는 영문자인 "24TH"의 "T", 그리고 공백 다음에 오는 문자인 "OLYMPIC"의 "O"는 대문자로 변환되고 나머지 영문자는 소문자로 변환된 "24Th Seoul Olympic"이 [B4] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

팀명은 전체 문자를 대문자로 변환하고, 국가는 첫 문자를 대문자로 변환하여 팀명(국가)에 표시하세요.

▶ 표시 예 : SUSUNG(Korea)

정답 [E3] : =UPPER(C3) & "(" & PROPER(D3) & ")"

	A	B	C	D	E
1	세계 클럽컵 축구대회				
2	순위	팀명	국가	팀명(국가)	
3	1	susung	korea	SUSUNG(Korea)	
4	2	baroserona	spain	BAROSERONA(Spain)	
5	3	chelsy	england	CHELSEY(England)	
6	4	roma	italy	ROMA(Italy)	
7	5	hoven	netherlands	HOVEN(Netherlands)	

수식의 이해

=UPPER(C3) & "(" & PROPER(D3) & ")"

①

②

- ① UPPER(C3) : [C3] 셀의 문자열을 대문자로 변환한 "SUSUNG"을 반환합니다.
- ② PROPER(D3) : [D3] 셀의 첫 번째 문자를 대문자로 변환한 "Korea"를 반환합니다. "SUSUNG"과 "Korea"를 각각 ①, ②에 대입하면 다음과 같습니다.
- ="SUSUNG" & "(" & "Korea" & ")" : & 연산자에 의해 모든 문자열이 합쳐진 "SUSUNG(Korea)"가 [E3] 셀에 입력됩니다.



전문가의 조언

UPPER 함수는 문자열에 포함된 소문자를 모두 대문자로 변환하는 함수입니다. 자세한 설명은 55쪽을 참고하세요.

055 소문자를 대문자로 변환하기 — UPPER

UPPER 함수는 문자열에 포함된 소문자를 모두 대문자로 변환합니다. 예를 들어, UPPER("Miss Kim")은 "MISS KIM"이 됩니다. 숫자, 특수 문자, 한글과 같이 대·소문자 구별이 없는 문자는 변환되지 않습니다.

형식 UPPER(텍스트) : 문자열에 포함된 소문자를 모두 대문자로 변환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'UPPER.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

	A	B
1	대문자로 변환하기	
2		
3	문자열	변환
4	korea	KOREA
5	대한민국	대한민국
6	Miss 경기도	MISS 경기도
7	!error#	!ERROR#
8	1234567	1234567
9	King, Sejong.	KING, SEJONG.

- ① =UPPER(A4) : [A4] 셀의 "korea"를 대문자로 변환한 "KOREA"가 [B4] 셀에 입력됩니다.
- ② =UPPER(A6) : [A6] 셀의 "Miss 경기도"를 대문자로 변환한 "MISS 경기도"가 [B6] 셀에 입력됩니다.
- ③ =UPPER(A8) : [A8] 셀의 "1234567"을 대문자로 변환한 "1234567"이 [B8] 셀에 입력됩니다. 숫자가 문자로 변환됩니다.

기출문제 따라잡기

'기출' 시트에서 실행하세요.

팀명은 대해 전체 문자를 대문자로 변환하고, 국가는 첫 문자를 대문자로 변환하여 팀명(국가)에 표시하세요.

▶ 표시 예 : SUSUNG(Korea)

정답 [E3] : =UPPER(C3) & "(" & PROPER(D3) & ")"

	A	B	C	D	E
1			세계 클럽컵 축구대회		
2		순위	팀명	국가	팀명(국가)
3		1	susung	korea	SUSUNG(Korea)
4		2	baroseron	spain	BAROSERON(Spain)
5		3	chelsy	england	CHELSEY(England)
6		4	roma	italy	ROMA(Italy)
7		5	hoven	netherlands	HOVEN(Netherlands)

수식의 이해

=UPPER(C3) & "(" & PROPER(D3) & ")"

- ① UPPER(C3) : [C3] 셀의 문자열을 대문자로 변환한 "SUSUNG"을 반환합니다.
 - ② PROPER(D3) : [D3] 셀의 첫 번째 문자를 대문자로 변환한 "Korea"를 반환합니다. "SUSUNG"과 "Korea"를 각각 ①, ②에 대입하면 다음과 같습니다.
- ="SUSUNG" & "(" & "Korea" & ")": & 연산자에 의해 모든 문자열이 합쳐진 "SUSUNG(Korea)"가 [E3] 셀에 입력됩니다.



전문가의 조언

PROPER 함수는 문자열에 포함된 대·소문자를 적절하게 변환하는 함수입니다. 자세한 설명은 52쪽을 참고하세요.

056 문자열의 양쪽 끝 공백 제거하기 — TRIM

TRIM 함수는 단어 사이에 있는 한 칸의 공백을 제외하고 텍스트에 포함된 모든 공백을 제거합니다. 다른 응용 프로그램에서 전달받은 텍스트에 불필요한 공백이 있을 때 TRIM 함수를 사용합니다. 예를 들어, TRIM(" 거 울 연 가 ")는 단어 양쪽의 공백을 모두 제거하고 글자 사이에 있는 공백은 한 칸만 남겨 "거 울 연 가"가 됩니다.

형식 TRIM(텍스트) : 문자열의 공백을 제거합니다.

준비하세요! : '갈벗컴활2급통합\부록\함수사전' 폴더의 'TRIM.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B
1	문자열의 공백제거	
2		
3	문자열	결과
4	KOREA	KOREA
5	대한민국	대한민국
6	Miss 경기도	Miss 경기도
7	!ERROR!	!ERROR!
8	1 2 3 4 5 6 7	1 2 3 4 5 6 7
9	Kim mi na	Kim mi na

① =TRIM(A4) : [A4] 셀에 입력된 " KOREA "에서 양쪽의 공백을 제거한 "KOREA"가 [B4] 셀에 입력됩니다.

② =TRIM(A6) : [A6] 셀에 입력된 " Miss 경기도 "에서 양쪽의 공백을 제거하고, "Miss"와 "경기도" 사이의 공백 4칸을 제거한 "Miss 경기도"가 [B6] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

도서코드 앞뒤에 있는 공백을 제거한 후 전체 문자를 대문자로 변환하고, 변환된 문자열 뒤에 "-KR"를 추가하여 변환도서코드에 표시하세요.

▶ 표시 예 : mng-002 → MNG-002-KR

정답 [D3] : =UPPER(TRIM(A3)) & "-KR"

	A	B	C	D
1	도서관리			
2	도서코드	출판사	출판년도	변환도서코드
3	mng-002	한국산업	2015	MNG-002-KR
4	psy-523	민음사	2013	PSY-523-KR
5	mng-091	두란노	2015	MNG-091-KR
6	psy-725	예코의 서재	2015	PSY-725-KR
7	nov-264	마티	2016	NOV-264-KR
8	lan-183	상공사	2010	LAN-183-KR
9	lan-184	민음사	2016	LAN-184-KR

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=UPPER(TRIM(A3)) & "-KR"

①

- ① TRIM(A3) : [A3] 셀에 입력된 " mng-002 "에서 양쪽의 공백이 제거된 "mng-002"를 반환합니다. "mng-002"를 ①에 대입하면 다음과 같습니다.
- UPPER("mng-002") : [A3] 셀의 문자열을 대문자로 변환한 "MNG-002"를 반환합니다.
- "MNG-002" & "-KR" : & 연산자에 의해 문자열이 합쳐져 "MNG-002-KR"이 [D3] 셀에 입력됩니다.



전문가의 조언

UPPER 함수는 문자열에 포함된 소문자를 모두 대문자로 변환하는 함수입니다. 자세한 설명은 55쪽을 참고하세요.

057 대문자를 소문자로 변환하기 — LOWER

23.상시, 22.상시, 21.상시

LOWER 함수는 문자열의 대문자를 모두 소문자로 변환합니다. 예를 들어, LOWER("Miss Kim")은 "miss kim"이 됩니다. 숫자, 특수 문자, 한글과 같이 소문자가 없는 문자는 변환되지 않습니다.

형식 LOWER(텍스트) : 문자열에 포함된 대문자를 모두 소문자로 변환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'LOWER.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B
1	소문자로 변환하기	
2		
3	문자열	변환
4	KOREA	korea
5	대한민국	대한민국
6	Miss 경기도	miss 경기도
7	!ERROR#	!error#
8	1234567	1234567
9	King, Sejong.	king, sejong.

- ① =LOWER(A4) : [A4] 셀의 "KOREA"를 소문자로 변환한 "korea"가 [B4] 셀에 입력됩니다.
- ② =LOWER(A6) : [A6] 셀의 "Miss 경기도"를 소문자로 변환한 "miss 경기도"가 [B6] 셀에 입력됩니다.
- ③ =LOWER(A8) : [A8] 셀의 "1234567"을 소문자로 변환한 "1234567"이 [B8] 셀에 입력됩니다(숫자가 문자로 변환됨).

기출문제 따라잡기

'기출' 시트에서 실습하세요.

국가는 모두 대문자로 변환하고, 수도는 모두 소문자로 변환하여 지역에 표시하세요.

▶ 표시 예 : KOREA(seoul)

정답 [C4] : =UPPER(A4) & "(" & LOWER(B4) & ")"

	A	B	C
1	국가별 수도		
2			
3	국가	수도	국가(수도)
4	Germany	Berlin	GERMANY(berlin)
5	Canada	Ottawa	CANADA(ottawa)
6	Japan	Tokyo	JAPAN(tokyo)
7	Portugal	Lisbon	PORTUGAL(lisbon)
8	China	Beijing	CHINA(beijing)
9	Angola	Luanda	ANGOLA(luanda)
10			

수식의 이해

=UPPER(A4) & "(" & LOWER(B4) & ")"

- ① UPPER(A4) : [A4] 셀의 문자열을 대문자로 변환한 "GERMANY"를 반환합니다.
 - ② LOWER(B4) : [B4] 셀의 문자열을 소문자로 변환한 "berlin"을 반환합니다.
- ="GERMANY" & "(" & "berlin" & ")" : 문자열 결합 연산자(&)에 의해 모든 문자열이 합쳐진 "GERMANY(berlin)"이 [C4] 셀에 입력됩니다.

058 문자열에서 특정 문자의 위치 찾기(글자 단위) — FIND

FIND 함수는 문자열에 포함된 특정 문자를 찾아 그 위치를 반환하는 함수인데, 글자 단위로 구분하여 위치를 카운트하며, 찾는 문자가 없을 경우 오류값을 표시합니다. 예를 들어 FIND("R", "★오必勝 KOREA★", 1)은 8을 반환합니다. 이때 대·소문자를 구분하며, 와일드 카드(*, ?)는 사용할 수 없습니다.

형식 FIND(찾을 텍스트, 문자열, 시작 위치) : '문자열'의 '시작 위치'에서부터 '찾을 텍스트'를 찾아 그 위치를 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'FIND.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

	A	B	C	D
1	찾을 텍스트 위치 값 반환하기			
2				
3	문자	찾을 텍스트	시작 위치	결과
4	KOREA	r	1	#VALUE!
5	No.1 sinagong	o	4	11
6	♥I Love YOU♥	O	2	10
7	★꿈! 夢! Dream★	e	3	10
8	1234	3	4	#VALUE!

- ① =FIND(B4, A4, C4) : [A4] 셀의 "KOREA"에 영문 소문자 "r"이 없으므로 오류값(#VALUE!)이 [D4] 셀에 입력됩니다.
- ② =FIND(B6, A6, C6) : [A6] 셀에 입력된 "♥I Love YOU♥"의 두 번째 글자에서부터 영문 대문자 "O"를 찾아 그 위치인 10을 [D6] 셀에 입력합니다.
- ③ =FIND(B7, A7, C7) : [A7] 셀에 입력된 "★꿈! 夢! Dream★"의 세 번째 글자에서부터 영문 소문자 "e"를 찾아 그 위치인 10을 [D7] 셀에 입력합니다.

059 문자열의 길이 알아내기 — LEN

LEN 함수는 문자열의 문자 수를 구할 때 사용하는 함수입니다. 예를 들어, LEN("유비쿼터스")는 5를 반환합니다. 이 때 한글과 영문, 공백은 똑같이 한 글자로 취급합니다.

형식 LEN(텍스트) : '텍스트'의 길이(개수)를 구합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'LEN.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

	A	B
1	문자열의 길이 세기	
2		
3	문자열	문자수
4	KOREA	5
5	대한민국	4
6	Miss 경기도	8
7	!ERROR!	7
8	1234567	7
9	King Sejong	11

- ① =LEN(A4) : [A4] 셀의 값 "KOREA"의 문자 수인 5가 [B4] 셀에 입력됩니다.
- ② =LEN(A6) : [A6] 셀의 값 "Miss 경기도"의 문자 수인 8이 [B6] 셀에 입력됩니다.
- ③ =LEN(A9) : [A9] 셀의 값 "King Sejong"의 문자 수인 11이 [B9] 셀에 입력됩니다.



전문가의 조언

문자열의 길이

- 공백도 문자 수에 포함됩니다.
- 문자열의 길이를 바이트 수로 구하려면 LENB를 이용하세요.

060 문자열에서 특정 문자의 위치 찾기(글자 단위) — SEARCH

SEARCH 함수는 문자열에 포함된 특정 문자를 찾아 그 위치를 반환하는 함수인데, 글자 단위로 구분하여 위치를 카운트하며, 찾는 문자가 없을 경우 오류값을 표시합니다. 예를 들어 SEARCH("R", "★오必勝 KOREA★", 1)은 8을 반환합니다. FIND 함수와 다른 점은 대·소문자 구분이 없고, 와일드 카드(*, ?)를 사용할 수 있다는 것입니다.

형식 SEARCH(찾을 텍스트, 문자열, 시작 위치) : '문자열'의 '시작 위치'에서부터 '찾을 텍스트'를 찾아 그 위치를 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'SEARCH.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C	D
1	찾을 텍스트 위치 값 반환하기			
2				
3	문자	찾을 텍스트	시작 위치	결과
4	KOREA	r	1	3
5	No.1 sinagong	o	4	11
6	♥I Love YOU♥	O	2	5
7	★꿈! 夢! Dream★	e	3	10
8	1234	3	4	#VALUE!

- ① =SEARCH(B4, A4, C4) : [A4] 셀에 입력된 "KOREA"의 첫 번째 글자에서부터 "r"을 찾아 그 위치인 3을 [D4] 셀에 입력합니다.
- ② =SEARCH(B6, A6, C6) : [A6] 셀에 입력된 "♥I Love YOU♥"의 두 번째 글자에서부터 "O"를 찾아 그 위치인 5를 [D6] 셀에 입력합니다.
- ③ =SEARCH(B8, A8, C8) : [A8] 셀에 입력된 "1234"의 네 번째 글자에서부터 3을 찾는데 3이 없으므로 오류값 (#VALUE!)이 [D8] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

E-메일에서 "@" 앞의 문자열만 추출하여 닉네임에 표시하세요.

▶ 표시 예 : abc@naver.com → abc

정답 [C4] : =MID(D4, 1, SEARCH("@", D4, 1) - 1)

	A	B	C	D
1	카페 신입회원 정보			
2				
3	성명	지역	닉네임	E-메일
4	최정예	서울	love99	love99@naver.com
5	심일훈	경기	muakiea	muakiea@nate.com
6	이아랑	인천	starcnk	starcnk@nate.com
7	김정필	부산	99023	99023@gmail.com
8	홍현서	대전	yses	yses@daum.net

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=MID(D4, 1, SEARCH("@", D4, 1) - 1)

①

- ① SEARCH("@", D4, 1) : [D4] 셀에 입력된 "love99@naver.com"의 첫 번째 글자에서부터 "@"를 찾아 그 위치인 7을 반환합니다.
- MID(D4, 1, 7 - 1) : "love99@naver.com"의 1 번째 위치부터 6(7-1) 글자만 추출한 'love99'가 [C4] 셀에 입력됩니다.



전문가의 조언

MID 함수는 지정된 문자 수에 따라 텍스트 문자열의 특정 위치에서 원하는 수 만큼의 문자를 추출하는 함수입니다. 자세한 설명은 5쪽을 참고하세요.

061 부서별 직급별 인원수 파악하기 — COUNTIFS

COUNTIFS 함수는 여러 개의 조건에 맞는 자료의 개수를 구하는 함수입니다. 예를 들면, 부서가 판매부이고, 급수가 1급이고, 남자인 사원들의 수를 셀 수 있습니다. 조건은 최대 127개까지 지정할 수 있습니다.

형식 COUNTIFS(첫 번째 조건이 적용될 범위, 첫 번째 조건, 두 번째 조건이 적용될 범위, 두 번째 조건, ...): 여러 개의 '조건이 적용될 범위'에서 여러 개의 '조건'에 맞는 셀을 찾아 개수를 계산합니다.

준비하세요! '길벗컴활2급통합\부록\함수사전' 폴더의 'COUNTIFS.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

COUNTIFS 함수를 이용하여 부서별 직급별 사원수를 계산해 보겠습니다.

	A	B	C	D	E	F	G	H	I
1		기본급 지급 현황				부서별 직급별 인원수			
2									
3		성명	부서	직급	기본급		직급	1급	2급
4		이승연	판매부	1급	1,450,000		판매부	2	2
5		김경수	기획부	2급	1,350,000		기획부	2	2
6		이학봉	판매부	2급	2,350,000				2
7		지순녀	기획부	2급	1,200,000				
8		김지연	판매부	1급	1,450,000				
9		박원래	기획부	1급	1,450,000				
10		최지은	기획부	1급	1,200,000				
11		강유라	판매부	2급	1,300,000				

① =COUNTIFS(C4:C11, "판매부", D4:D11, "1급")
: [C4:C11] 영역에서 "판매부"가 입력된 셀들을 찾아 [D4:D11] 영역에서 같은 행들에 "1급"이 입력된 셀들의 개수인 2가 [H4] 셀에 입력됩니다.

② =COUNTIFS(C4:C11, "기획부", D4:D11, "2급")
: [C4:C11] 영역에서 "기획부"가 입력된 셀들을 찾아 [D4:D11] 영역에서 같은 행들에 "2급"이 입력된 셀들의 개수인 2가 [I5] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

적립포인트가 600 이상이면서 등급이 "VIP"인 고객수를 구하세요.

▶ 고객수 뒤에 "명"을 표시 [표시 예: 1명]

정답 [E12] : =COUNTIFS(D3:D10, ">=600", E3:E10, "VIP") & "명"

	A	B	C	D	E
1	고객 현황				
2	고객코드	성별	나이	적립포인트	등급
3	K1001	남	66	580	VIP
4	K1125	남	48	700	VIP
5	K3948	여	32	650	일반
6	K2840	여	29	500	일반
7	K1753	여	46	685	VIP
8	K2385	남	33	420	일반
9	K9375	남	52	600	VIP
10	K8923	여	45	360	일반
11					
12	적립포인트가 600 이상인 VIP 고객				3명

수식의 이해

=COUNTIFS(D3:D10, ">=600", E3:E10, "VIP") & "명"

①

- ① COUNTIFS(D3:D10, ">=600", E3:E10, "VIP") : [D3:D10] 영역에서 적립포인트가 600 이상인 셀들([D4], [D6], [D7], [D9])을 찾는 다음 [E3:E10] 영역의 같은 행들 중에서 "VIP"가 입력된 셀들([D4], [D7], [D9])의 개수인 3을 반환합니다. 3을 ①에 대입하면 다음과 같습니다.
- =3 & "명" : 3과 "명"을 결합한 "3명"이 [E12] 셀에 입력됩니다.

062 품목의 판매 건수 구하기 — COUNTIF

COUNTIF 함수는 많은 자료 중에서 조건에 맞는 데이터의 개수만을 구하는 함수입니다. 찾을 조건이 적용된 범위에서 조건에 맞는 데이터를 찾아 개수를 계산합니다.

형식 COUNTIF(범위, 조건) : 지정된 '범위'에서 '조건'에 맞는 셀의 개수를 계산합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'COUNTIF.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

COUNTIF 함수를 이용하여 품목별 판매 건수를 계산해 보겠습니다.

	A	B	C	D	E	F	G
1		판매현황				품목별 판매건수	
2							
3		품목	수량	단가	금액	품목	건수
4		냉장고	6	250	1,500	컴퓨터	3
5		컴퓨터	8	300	2,400	캠코더	2
6		냉장고	5	250	1,250	냉장고	2
7		캠코더	7	500	3,500		
8		컴퓨터	10	300	3,000		
9		캠코더	5	500	2,500		
10		컴퓨터	7	300	2,100		
11							

① =COUNTIF(A4:A10, "컴퓨터") : [A4:A10] 영역에서 "컴퓨터"가 입력된 셀의 개수 3이 [G4] 셀에 입력됩니다.

② =COUNTIF(A4:A10, "냉장고") : [A4:A10] 영역에서 "냉장고"가 입력된 셀의 개수 2가 [G6] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실행하세요.

도서코드가 "B002", "B004"인 도서의 수를 구하여 표시하세요.

▶ 도서수 뒤에 "권"을 표시 [표시 예 : 3권]

정답 [C13] : =COUNTIF(\$B\$3:\$B\$10, C12) & "권"

	A	B	C	D
1		도서 관리		
2		도서코드	도서명	대여료
3		B002	성공심리학	500
4		B003	컴퓨터구조	800
5		B001	한국사	700
6		B005	엑셀2000	1,200
7		B004	기업윤리	300
8		B002	성공심리학	500
9		B004	기업윤리	300
10		B002	성공심리학	700
11				
12		도서코드	B002	B004
13		도서수	3권	2권
14				

수식의 이해

=COUNTIF(B3:B10, C12) & "권"

①

- ① COUNTIF(B3:B10, C12) : [B3:B10] 영역에서 [C12] 셀의 값과 같은 셀([B3], [B8], [B10]), 즉 도서코드가 "B002"인 셀의 개수 3을 반환합니다. 3을 ①에 대입하면 다음과 같습니다.
- =3 & "권" : 3과 "권"을 결합한 "3권"이 [C13] 셀에 입력됩니다.

063 순위 계산하기 1 — RANK.EQ

RANK.EQ 함수는 지정된 범위 안에서 인수의 순위를 구하고, 동일한 값들은 동일하지 않을 경우 나올 수 있는 순위들 중 가장 높은 순위를 동일하게 표시하는 함수입니다. 예를 들어 2위인 점수 90이 3개가 있다면 동일하지 않을 경우 나올 수 있는 순위 2, 3, 4 중 가장 높은 순위인 2가 동일하게 표시되고, 다음 순위는 3위, 4위없이 5위가 표시됩니다.

형식 RANK.EQ(인수, 범위, 옵션) : 지정된 '범위' 안에서 '인수'의 순위를 구하고, 동일한 값들은 동일하지 않을 경우 나올 수 있는 순위들 중 가장 높은 순위를 동일하게 표시합니다.

준비하세요! : '길벗컴퓨터2급통합\부록\함수사전' 폴더의 'RANK.EQ.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

RANK.EQ 함수를 이용하여 총점을 기준으로 한 순위를 계산하여 표시해 보겠습니다.

	A	B	C	D	E	F
1	성적표					
2	성명	국어	영어	수학	총점	순위
3	고아라	72	90	78	240	2
4	나영희	88	80	72	240	2
5	박철수	75	98	75	248	1
6	안도해		100	100	200	4
7	최순이	85		85	170	5

① =RANK.EQ(E3, \$E\$3:\$E\$7) : [E3:E7] 영역에서 [E3] 셀의 값 240의 순위를 가장 높은 점수에 1위를 부여하는 방식(내림차순)을 적용하여 계산하되, 240이 두 개이므로 동일하지 않을 경우 나올 수 있는 순위 2, 3 중 가장 높은 순위 2가 [F3] 셀에 입력됩니다. 논리값이 생략되었으므로 내림차순으로 계산합니다.

RANK.EQ 함수의 옵션

- 0 또는 생략 : 내림차순을 기준으로 한 순위를 부여(가장 큰 값에 1위를 부여)
- 0 이외의 값 : 오름차순을 기준으로 한 순위를 부여(가장 작은 값에 1위를 부여)

기출문제 따라잡기

'기출' 시트에서 실행하세요.

취득총점의 순위를 구한 후 수상목록에서 순위에 해당하는 수상명을 찾아 표시하세요.

▶ 순위는 취득총점이 가장 높은 것이 1위

정답 [D3] : =VLOOKUP(RANK.EQ(C3, \$C\$3:\$C\$11), \$F\$4:\$G\$9, 2)

	A	B	C	D	E	F	G
1	모형 항공기 제작 경진 대회						
2	팀명	취득총점	수상명	<수상목록>			
3	비상	158	동상	순위	수상명		
4	하늘	195	금상	1	대상		
5	창공	157	동상	2	금상		
6	송골매	154	장려상	3	은상		
7	보라매	120		4	동상		
8	이상	198	대상	6	장려상		
9	볼새	145	장려상	9			
10	마하	135	장려상				
11	최원영	160	은상				

수식의 이해

=VLOOKUP(RANK.EQ(C3, C3:C11), F4:G9, 2)

①

- ① RANK.EQ(C3, C3:C11) : [C3:C11] 영역에서 [C3] 셀의 값 158의 내림차순을 기준으로 계산한 순위 4를 반환합니다. 4를 ①에 대입하면 다음과 같습니다.
- =VLOOKUP(4, F4:G9, 2)
 - ➡ [F4:G9] 영역의 맨 왼쪽 열에서 4를 넘지 않는 가장 근접한 값, 4행에 있는 4를 찾습니다.
 - ➡ 4가 있는 행에서 2열에 있는 값 "대상"을 찾아서 [D3] 셀에 입력합니다.
 - ※ 4를 넘지 않는 가장 근접한 값을 찾는 것은 '옵션'에 지정된 논리값이 생략되었기 때문입니다. 생략하거나 'TRUE'인 경우에는 찾을 값보다 크지 않으면서 가장 근접한 값을 찾습니다.

	F	G
1		
2	<수상목록>	
3	순위	수상명
4	1	대상
5	2	금상
6	3	은상
7	4	동상
8	6	장려상
9	9	



전문가의 조언

VLOOKUP 함수는 범위로 정한 영역의 맨 왼쪽 열에서 특정 기준으로 자료를 찾고, 그 자료가 속한 행 중에서 필요한 값이 있는 열의 위치를 지정하여 값을 반환하는 함수입니다. 자세한 내용은 45쪽을 참고하세요.

064 N번째로 큰 수 찾기 — LARGE

LARGE 함수는 자료 범위에서 N번째로 큰 값을 반환하는 함수입니다. 이 함수를 사용하여 상대적인 순위에 해당하는 값을 선택할 수 있습니다. 예를 들면, LARGE 함수를 사용하여 1등, 2등, 3등의 점수를 구할 수 있습니다.

형식 LARGE(범위, N번째) : '범위' 중 'N번째'로 큰 값을 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'LARGE.xlsx' 파일을 열어 '기본' 시트에서 실험하세요.

	A	B	C	D	E	F
1	지정된 큰 수 찾기					
2						
3	숫자1	숫자2	숫자3			
4	12	21	18	제일 큰 수	24	①
5	5	2	19	두 번째 큰 수	21	
6	7	16	18	세 번째 큰 수	19	②
7	4	15	24			
8						

① =LARGE(A4:C7, 1) : [A4:C7] 영역에서 첫 번째로 큰 수, 즉 가장 큰 수인 24가 [F4] 셀에 입력됩니다.

② =LARGE(A4:C7, 3) : [A4:C7] 영역에서 세 번째로 큰 수인 19가 [F6] 셀에 입력됩니다.

- 자료 범위가 비어 있으면 '#NUM!' 오류값이 반환됩니다.
- 인수 N이 0보다 작거나 데이터 요소 개수보다 크면 '#NUM!' 오류값이 반환됩니다.
- 범위에 있는 데이터 요소가 n개이면, LARGE(범위, 1)은 가장 큰 값을 LARGE(범위, n)은 가장 작은 값을 구합니다.

기출문제 따라잡기

'기출' 시트에서 실험하세요.

가장 큰 총점과 두 번째로 큰 총점의 차이를 구하여 표시하세요.

정답 [E9] : =MAX(E3:E7) - LARGE(E3:E7, 2)

	A	B	C	D	E
1		교원 공채 시험 결과			
2		성명	이론	논술	총점
3		장진영	98	98	196
4		황갑영	97	100	197
5		심순영	95	90	185
6		이철순	85	87	172
7		장영혜	100	99	199
8					
9		1등과 2등의 점수 차이			2
10					

수식의 이해

=MAX(E3:E7) - LARGE(E3:E7, 2)

①

②

- ① MAX(E3:E7) : [E3:E7] 영역에서 가장 큰 값 199를 반환합니다.
- ② LARGE(E3:E7, 2) : [E3:E7] 영역에서 2번째로 큰 값 197을 반환합니다. 199와 197을 각각 ①, ②에 대입하면 다음과 같습니다.
- =199 - 197 : 199에서 197을 뺀 값 2가 [E9] 셀에 입력됩니다.



전문가의 조언

MAX 함수는 인수로 주어진 숫자들 중에서 최대값을 구하는 함수입니다. 자세한 내용은 67쪽을 참고하세요.

065 평균 계산하기 — AVERAGE

AVERAGE 함수는 인수로 주어진 숫자들의 평균을 계산하는 함수로, 인수는 1개에서 255개까지 지정할 수 있습니다. 인수는 숫자이거나 숫자가 포함된 이름, 배열 또는 셀 주소이어야 합니다.

형식 AVERAGE(인수1, 인수2, ...): '인수'로 주어진 숫자들의 평균을 계산합니다.

준비하세요! : '길벗컴퓨터2급통합\부록\함수사전' 폴더의 'AVERAGE.xlsm' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C	D
1	평균 계산			
2				
3	숫자1	숫자2	숫자3	평균
4	5.5	6.5	7.5	6.5
5	20	30	40	30
6	0	30	26	18.6667
7	8		10	9

- ① =AVERAGE(A4:C4) : [A4:C4] 영역의 평균 6.5가 [D4] 셀에 입력됩니다.
- ② =AVERAGE(A6:C6) : [A6:C6] 영역의 평균 18.6667이 [D6] 셀에 입력됩니다.

[파일] → [옵션]을 선택한 후 'Excel 옵션' 대화상자의 '고급' 탭에서 '이 워크시트의 표시 옵션'의 '0값이 있는 셀에 0 표시' 확인란이 선택되어 있지 않으면 0이 들어 있는 셀에 데이터가 없는 것처럼 빈 셀로 표시되어 혼동할 수 있습니다. 왜냐하면 셀의 평균을 구할 때 0이 들어 있는 셀은 평균에 포함되어 계산되지만 빈 셀은 계산되지 않기 때문입니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

각 학생들의 중간, 수행, 기말 점수에 대한 평균을 구하여 표시하세요.

▶ 반올림 없이 소수 이하 첫째 자리까지 표시

[표시 예] : 94.37 → 94.3]

정답 [F3] : =TRUNC(AVERAGE(C3:E3), 1)

	A	B	C	D	E	F
1	1학기 국어 성적					
2		성명	중간	수행	기말	평균
3		김정훈	78.45	45.78	87.23	70.4
4		오석현	88.79	87.34	90.45	88.8
5		이영선	92.45	80.23	78.23	83.6
6		임현재	88.45	77.54	98.56	88.1
7		남정왕	88.66	89.12	89.54	89.1

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=TRUNC(AVERAGE(C3:E3), 1)

①

- ① AVERAGE(C3:E3) : [C3:E3] 영역의 평균 70.486...을 반환합니다. 70.486...을 ①에 대입하면 다음과 같습니다.
- =TRUNC(70.486..., 1) : 70.486...을 반올림 없이 소수 이하 첫째 자리까지 표시한 70.4가 [F3] 셀에 표시됩니다.



전문가의 조언

TRUNC 함수는 숫자에서 지정한 자릿수 이하의 숫자를 버릴 때 사용하는 함수입니다. 자세한 설명은 39쪽을 참고하세요.

066 문자도 포함하여 평균 계산하기 — AVERAGEA

AVERAGEA 함수는 빈 셀을 제외한 모든 인수를 포함하여 평균을 계산하는 함수입니다. 문자가 포함되어 있으면 문자를 0(영)으로 취급하여 평균을 계산합니다. 인수는 숫자, 이름, 배열, 참조 영역 등으로 1개에서 255개까지 지정할 수 있습니다. 문자를 포함하지 않는 평균을 구할 때에는 AVERAGE 함수를 사용합니다.

형식 AVERAGEA(인수1, 인수2, ...): '인수'로 주어진 값들의 평균을 계산합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'AVERAGEA.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

	A	B	C	D
1	평균 계산			
2				
3	숫자1	숫자2	숫자3	평균
4	FALSE	2	4	2
5	20	무효	40	20
6	0		5	2.5
7	10	TRUE	4	5

- ① =AVERAGEA(A4:C4) : FALSE는 0으로 간주됩니다. 즉 '(0+2+4)/3'을 계산한 2가 [D4] 셀에 입력됩니다.
- ② =AVERAGEA(A5:C5) : "무효"는 문자이므로 0으로 간주됩니다. 즉 '(20+0+40)/3'을 계산한 20이 [D5] 셀에 입력됩니다.
- ③ =AVERAGEA(A7:C7) : TRUE는 1로 간주됩니다. 즉 '(10+1+4)/3'을 계산한 5가 [D7] 셀에 입력됩니다.

[파일] → [옵션]을 선택한 후 'Excel 옵션' 대화상자의 '고급' 탭에서 '이 워크시트의 표시 옵션'의 '0값이 있는 셀에 0 표시' 확인란이 선택되어 있지 않으면 0이 들어 있는 셀에 데이터가 없는 것처럼 빈 셀로 표시되어 혼동할 수 있습니다. 왜냐하면 셀의 평균을 구할 때 0이 들어 있는 셀은 평균에 포함되어 계산되지만 빈 셀은 계산되지 않기 때문입니다.

067 분산 계산하기 — VAR.S

VAR.S 함수는 인수로 주어진 숫자들에 대한 분산을 계산할 때 사용하는 함수입니다. VAR.S 함수는 인수를 모집단의 표본으로 간주하며, 논리값이나 텍스트는 무시됩니다.

형식 VAR.S(인수1, 인수2, ...): '인수'로 주어진 숫자들의 분산을 구합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'VAR.S.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

	A	B	C	D
1	분산 계산			
2				
3	숫자1	숫자2	숫자3	분산
4	1	2	3	1.00
5	2	2	2	0.00
6	0	3	5	6.33
7	4		2	2.00

- ① =VAR.S(A4:C4) : [A4:C4] 영역에 대한 표본 분산 1.00이 [D4] 셀에 입력됩니다.
- ② =VAR.S(A7:C7) : [A7:C7] 영역에 대한 표본 분산 2.00이 [D7] 셀에 입력됩니다.

068 자료가 없는 셀의 개수 세기 — COUNTBLANK

19, 상시, 05.4, 05.3, 04.2

COUNTBLANK 함수는 주어진 셀 범위에서 자료가 없는 셀의 개수를 구하는 함수입니다.

형식 COUNTBLANK(범위) : '범위' 중에서 자료가 없는 셀의 개수를 구합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'COUNTBLANK.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

	A	B	C	D
1	빈셀의 수			
2				
3	자료1	자료2	자료3	빈셀의 수
4	2021-09-14	52		1
5	TRUE	0	85	0
6	엑셀		67	1
7	4	92	#DIV/0!	0
8				

① =COUNTBLANK(A4:C4) : [A4:C4] 영역 중 빈 셀의 개수 1이 [D4] 셀에 입력됩니다.

② =COUNTBLANK(A7:C7) : [A7:C7] 영역 중 빈 셀의 개수 0이 [D7] 셀에 입력됩니다.

빈 텍스트(Null 값)를 반환하는 수식이 들어 있는 셀은 개수 계산에 포함되지만 0값을 갖는 셀은 계산에 포함되지 않습니다.

기출문제 따라잡기

'기출' 시트에서 실행하세요.

수업료, 급식비, 교과서 대금의 미납자수를 구하여 미납자수에 표시하세요.

정답 [D11] : =COUNTBLANK(D3:D10)

	A	B	C	D	E	F
1	각종 대금 납부 현황					
2		학번	성명	수업료	급식비	교과서
3		4101	김미연		납부	
4		4102	원동철	납부		납부
5		4103	이지함	납부	납부	
6		4104	한두리		납부	납부
7		4105	신민영	납부	납부	납부
8		4106	구찬우	납부	납부	
9		4107	장철우		납부	
10		4108	김인애	납부		납부
11			미납자수	3	2	4
12						

수식의 이해

=COUNTBLANK(D3:D10)

[D3:D10] 영역에서 비어 있는 셀[D3], [D6], [D9]의 개수 3이 [D11] 셀에 입력됩니다.

069 가장 큰 수 찾기 — MAX

25.상시, 23.상시, 22.상시, 21.상시, 20.상시, 06.1, 04.4

MAX 함수는 인수로 주어진 숫자들 중에서 최대값을 구하는 함수로, 1개부터 255개까지의 인수를 사용할 수 있습니다. 인수에 포함된 텍스트, 논리값 또는 빈 셀은 무시됩니다. 인수에 숫자가 하나도 없으면 0이 반환됩니다.

형식 MAX(인수1, 인수2, ...): '인수'로 주어진 숫자 중에서 가장 큰 수를 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'MAX.xlsm' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C	D	E
1	가장 큰 수 찾기				
2					
3	숫자1	숫자2	숫자3	숫자4	가장 큰 수
4	12	21	18	5	21
5	5	2	19	21	21
6	7	16	18	14	18
7	4	15	24	2	24

① =MAX(A4:D4) : [A4:D4] 영역에서 가장 큰 값 21이 [E4] 셀에 입력됩니다.

② =MAX(A6:D6) : [A6:D6] 영역에서 가장 큰 값 18이 [E6] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

가장 큰 총점과 두 번째로 큰 총점의 차이를 구하여 표시하세요.

정답 [E9] : =MAX(E3:E7) - LARGE(E3:E7, 2)

	A	B	C	D	E
1	교원 공채 시험 결과				
2		성명	이론	논술	총점
3		장진영	98	98	196
4		황갑영	97	100	197
5		심순영	95	90	185
6		이철순	85	87	172
7		장영혜	100	99	199
8					
9		1등과 2등의 점수 차이			2

수식의 이해

=MAX(E3:E7) - LARGE(E3:E7, 2)

①

②

- ① MAX(E3:E7) : [E3:E7] 영역에서 가장 큰 값 199를 반환합니다.
- ② LARGE(E3:E7, 2) : [E3:E7] 영역에서 2번째로 큰 값 197을 반환합니다. 199와 197을 각각 ①, ②에 대입하면 다음과 같습니다.
- =199 - 197 : 199에서 197을 뺀 값 2가 [E9] 셀에 입력됩니다.



전문가의 조언

LARGE 함수는 자료 범위에서 n번째로 큰 값을 반환하는 함수입니다. 자세한 내용은 63쪽을 참고하세요.

070 가장 작은 수 찾기 — MIN

25.상시, 20.상시, 10.3, 06.1

MIN 함수는 인수로 주어진 숫자들 중에서 최소값을 구하는 함수로, 1개부터 255개까지의 인수를 사용할 수 있습니다. 인수에 텍스트, 논리값 또는 빈 셀이 포함되는 경우 그 값은 무시됩니다. 인수에 숫자가 하나도 없으면 0이 반환됩니다.

형식 MIN(인수1, 인수2, ...): '인수'로 주어진 숫자 중에서 가장 작은 수를 반환합니다.

준비하세요! : '길벗컴퓨터2급통합\부록\함수사전' 폴더의 'MIN.xlsx' 파일을 열어 '기본' 시트에서 실험하세요.

	A	B	C	D	E
1	가장 작은 수 찾기				
2					
3	숫자1	숫자2	숫자3	숫자4	가장 작은 수
4	12	21	18	5	5
5	5	2	19	21	2
6	7	16	18	14	7
7	4	15	24	2	2

- ① =MIN(A4:D4) : [A4:D4] 영역에서 가장 작은 값 5가 [E4] 셀에 입력됩니다.
- ② =MIN(A6:D6) : [A6:D6] 영역에서 가장 작은 값 7이 [E6] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실험하세요.

가격을 기준으로 가격이 가장 높으면 "최고가", 가장 낮으면 "최저가", 나머지 셀은 공백으로 표시하세요.

정답 [D3] : =IF(MAX(\$C\$3:\$C\$9)=C3, "최고가", IF(MIN(\$C\$3:\$C\$9)=C3, "최저가", " "))

	A	B	C	D
1	핸드폰 가격 비교			
2		쇼핑몰	가격	비교
3		F-MARKET	464,310	
4		K-MARKET	465,000	최고가
5		B-MARKET	449,820	
6		A-MARKET	440,000	최저가
7		H-MARKET	459,620	
8		D-MARKET	454,100	
9		C-MARKET	453,000	

수식의 이해

=IF(MAX(C3:C9)=C3, "최고가", IF(MIN(C3:C9)=C3, "최저가", " "))

조건①이 참(TRUE)이면 ②를 실행하고 거짓(FALSE)이면 ③을 실행합니다.

- ① MAX(C3:C9)=C3 : [C3:C9] 영역에서 가장 큰 값 465,000과 [C3] 셀의 값 464,310이 다르므로 거짓을 반환합니다. 거짓을 ①에 대입하면 다음과 같습니다.

- =IF(거짓, "최고가", IF(MIN(C3:C9)=C3, "최저가", " ")) : 조건①이 거짓이므로 ③을 실행합니다.

- ③ IF(MIN(C3:C9)=C3, "최저가", " ")

조건④이 참이면 ⑤를 실행하고 거짓이면 ⑥을 실행합니다.

- ④ MIN(C3:C9)=C3 : [C3:C9] 영역에서 가장 작은 값 440,000과 [C3] 셀의 값 464,310이 다르므로 거짓을 반환합니다. 거짓을 ④에 대입하면 다음과 같습니다.

- =IF(거짓, "최저가", " ") : 조건이 거짓이므로 [D3] 셀에 " "(공백)이 입력됩니다.



전문가의 조언

IF 함수는 참과 거짓에 관한 논리식을 판별하여 참일 때와 거짓일 때 서로 다른 값을 반환하기 위해 사용하는 함수입니다. 자세한 설명은 20쪽을 참고하세요.

071 숫자가 들어 있는 셀의 개수 세기 — COUNT

COUNT 함수는 인수로 주어진 값에서 숫자가 있는 셀의 개수를 구하는 함수입니다. 인수는 1개에서 255개까지 사용할 수 있으나 개수 계산에는 숫자만 포함됩니다. 논리값, 텍스트 또는 오류값도 포함하여 개수를 계산해야 하는 경우에는 COUNTA 함수를 사용합니다.

형식 COUNT(인수1, 인수2, ...): '인수'로 주어진 값 중 숫자가 있는 셀의 개수를 구합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'COUNT.xlsm' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C	D
1	자료수 세기			
2				
3	자료1	자료2	자료3	자료의 수
4	2021-09-14	52		2
5	TRUE	1:01 AM	85	2
6	엑셀		67	1
7	4	92	#DIV/0!	1

- ① =COUNT(A4:C4) : [A4:C4] 영역에서 숫자가 들어 있는 셀의 개수 2가 [D4] 셀에 입력됩니다.
- ② =COUNT(A6:C6) : [A6:C6] 영역에서 숫자가 들어 있는 셀의 개수 1이 [D6] 셀에 입력됩니다.
- ③ =COUNT(A7:C7) : [A7:C7] 영역에서 숫자가 들어 있는 셀의 개수 1이 [D7] 셀에 입력됩니다. [A7] 셀의 "4"는 문자 "4"이므로 개수에 포함되지 않습니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

근무점수가 70 이상 80 미만인 사람의 수를 구하세요.

정답 [B10] : =COUNT(D3:D7) - COUNTIF(D3:D7, "<70") - COUNTIF(D3:D7, ">=80")

	A	B	C	D
1		직원 근무 평가		
2		성명	입사일	근무점수
3		박정호	2014-06-06	73
4		신정희	2016-04-01	68
5		김용태	2012-05-06	98
6		김진영	2015-11-01	73
7		유현숙	2013-01-01	69
8				
9		70점대		
10		2		

수식의 이해

=COUNT(D3:D7) - COUNTIF(D3:D7, "<70") - COUNTIF(D3:D7, ">=80")

① ② ③

- ① COUNT(D3:D7) : 전체 인원수를 구하는 것으로, [D3:D7] 영역에서 숫자가 있는 셀의 개수 5를 반환합니다.
- ② COUNTIF(D3:D7, "<70") : 근무점수가 70 미만인 사람의 인원수를 구하는 것으로, [D3:D7] 영역에서 70점 미만인 셀의 개수 2를 반환합니다.
- ③ COUNTIF(D3:D7, ">=80") : 근무점수가 80 이상인 사람의 인원수를 구하는 것으로, [D3:D7] 영역에서 80점 이상인 셀의 개수 1을 반환합니다. 5, 2, 1을 각각 ①, ②, ③에 대입하면 다음과 같습니다.
- =5-2-1 : '5-2-1'의 결과 2가 [B10] 셀에 입력됩니다.



전문가의 조언

- 근무점수가 70 이상이고 80 미만이라는 것은 전체 인원수에서 근무점수가 70 미만인 사람의 수와 80 이상인 사람의 수를 뺀 것과 같은 의미입니다.
- COUNTIF 함수는 많은 자료 중에서 조건에 맞는 데이터의 개수만을 구하는 함수입니다. 자세한 내용은 6쪽을 참고하세요.

072 자료가 입력되어 있는 모든 셀의 개수 세기 — COUNTA

COUNTA 함수는 인수로 주어진 값 중에서 자료가 입력되어 있는 모든 셀의 개수를 세는 함수입니다. 인수는 1개에서 255개까지 사용할 수 있으나 빈 셀은 개수 계산에서 제외됩니다. 숫자가 들어 있는 셀만 세고자 할 경우에는 COUNT 함수를 사용합니다.

형식 COUNTA(인수1, 인수2, ...): '인수'로 주어진 값 중 자료가 입력되어 있는 셀의 개수를 구합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'COUNTA.xlsm' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C	D
1	자료수 세기			
2				
3	자료1	자료2	자료3	자료의 수
4	2021-09-14	52		2
5	TRUE	1:01 AM	85	3
6	엑셀		67	2
7	4	92	#DIV/0!	3

- ① =COUNTA(A4:C4) : [A4:C4] 영역 중 빈 셀을 제외한 셀의 개수 2가 [D4] 셀에 입력됩니다.
- ② =COUNTA(A6:C6) : [A6:C6] 영역 중 빈 셀을 제외한 셀의 개수 2가 [D6] 셀에 입력됩니다.

기출문제 따라잡기 '기출' 시트에서 실습하세요.

컴퓨터일반과 엑셀이 모두 60점 이상이면 합격입니다. 합격률을 표시하세요.

▶ 합격률 = '컴퓨터일반'과 '엑셀'이 모두 60점 이상인 수험생 수 / 총 수험생 수

정답 [C11] : =COUNTIFS(B3:B10, ">=60", C3:C10, ">=60") / COUNTA(A3:A10)

	A	B	C
1	컴활2급 시험 결과		
2	수험번호	컴퓨터일반	엑셀
3	1270121	92	76
4	1270122	38	55
5	1270123	86	92
6	1270124	62	48
7	1270125	91	93
8	1270126	49	56
9	1270127	89	93
10	1270128	67	68
11	합격률		63%

수식의 이해

=COUNTIFS(B3:B10, ">=60", C3:C10, ">=60") / COUNTA(A3:A10)

- ① COUNTIFS(B3:B10, ">=60", C3:C10, ">=60") : [B3:B10] 영역에서 60 이상인 셀들([B3], [B5], [B6], [B7], [B9], [B10])을 찾은 다음 [C3:C10] 영역의 같은 행들 중에서 60 이상인 셀들([B3], [B5], [B7], [B9], [B10])의 개수인 5를 반환합니다.
 - ② COUNTA(A3:A10) : [A3:A10] 영역에서 자료가 입력되어 있는 모든 셀의 개수 8을 반환합니다. 5와 8을 ①과 ②에 대입하면, 5 / 8, 즉 0.625가 [C11] 셀에 입력됩니다.
- ※ 셀 서식이 '백분율'로 지정되어 있어 '63%'로 표시됩니다.

073 중간에 위치한 값 찾아내기 — MEDIAN

25.상시, 23.상시, 19.상시, 16.3, 07.3

MEDIAN 함수는 인수로 주어진 숫자들을 크기 순으로 나열했을 때 중간 위치에 해당하는 값을 반환하는 함수입니다. 즉, 수의 반은 중간값보다 큰 값을 가지고 나머지 반은 중간값보다 작은 값을 가집니다. 인수에 텍스트, 논리값 또는 빈 셀이 포함되는 경우 그 값은 무시되지만 값이 0인 셀은 계산에 포함됩니다.

형식 MEDIAN(인수1, 인수2, ...): '인수'로 주어진 숫자들 중에서 중간에 해당하는 값을 반환합니다.

준비하세요! '길벗컴활2급통합\부록\함수사전' 폴더의 'MEDIAN.xlsx' 파일을 열어 '기분' 시트에서 실습하세요.

	A	B	C	D	E	F
1	중간 값 찾기					
2						
3	숫자1	숫자2	숫자3	숫자4	숫자5	중간값
4	1	2	3	4	5	3
5	0	1	2	3	4	2
6	7	14	16	18		15
7	25	15	7	3	1	7

- ① =MEDIAN(A4:E4) : [A4:E4] 영역에서 중간에 해당하는 값 3이 [F4] 셀에 입력됩니다.
- ② =MEDIAN(A6:E6) : [A6:E6] 영역에서 중간에 해당하는 값 15가 [F6] 셀에 입력됩니다. 인수의 개수가 짝수인 경우 중간에 두 값의 평균을 반환합니다. 즉 중간에 두 값 [B6] 셀의 14와 [C6] 셀의 16을 더한 후 2로 나눈 값 15가 반환됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

각 상품의 분기합계가 전체 분기합계의 중앙값 이상이면 "우수상품", 그 이외에는 공백으로 비교에 표시하세요.

정답 [F3] : =IF(E3>=MEDIAN(\$E\$3:\$E\$10), "우수상품", "")

	A	B	C	D	E	F
1	2/4분기 판매현황					
2	상품명	4월	5월	6월	분기합계	비고
3	DMB	1,250	1,542	1,871	4,663	
4	핸드폰	2,451	2,874	3,120	8,445	우수상품
5	MP3	3,245	3,382	3,579	10,206	우수상품
6	DVD	1,479	1,657	1,897	5,033	우수상품
7	LCD-TV	358	457	587	1,402	
8	PDP-TV	748	854	698	2,300	
9	드럼세탁기	1,458	1,987	2,147	5,592	우수상품
10	에어컨	587	789	1,548	2,924	

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=IF(E3>=MEDIAN(\$E\$3:\$E\$10), "우수상품", "")

① ② ③

- 조건(①)이 참(TRUE)이면 ②를, 거짓(FALSE)이면 ③을 표시합니다.
- ① E3>=MEDIAN(\$E\$3:\$E\$10)
 - MEDIAN(\$E\$3:\$E\$10) : [E3:E10] 영역에서 중간에 해당하는 값 4,848을 반환합니다.
 - [E3] 셀의 값 4,663이 [E3:E10] 영역의 중간값 4,848보다 작으므로 거짓을 반환합니다.
 - 조건(①)이 거짓이므로 ③을 수행하여 [F3]셀에 " "(공백)이 표시됩니다.



전문가의 조언

IF 함수는 참과 거짓에 관한 논리식을 판별하여 참일 때와 거짓일 때 서로 다른 값을 반환하기 위해 사용하는 함수입니다. 자세한 설명을 20쪽을 참고하세요.

074 가장 많이 나오는 수(최빈수) 찾아내기 — MODE.SNGL

MODE.SNGL 함수는 인수로 주어진 숫자들 중 가장 많이 발생한 값(최빈수)을 반환하는 함수로, 인수는 1개에서 255개까지 지정할 수 있습니다. 인수는 숫자이거나 숫자가 포함된 이름, 배열 또는 셀 범위여야 합니다. 지정된 인수에 텍스트, 논리값 또는 빈 셀이 포함되는 경우 그 값은 무시되지만 값이 0인 셀은 계산에 포함됩니다.

형식 MODE.SNGL(인수1, 인수2, ...): '인수'로 주어진 숫자들 중에서 가장 빈도수가 많은 값을 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'MODE.SNGL.xlsm' 파일을 열어 '기본' 시트에서 실험하세요.

	A	B	C	D	E	F
1	최빈 값 찾기					
2						
3	숫자1	숫자2	숫자3	숫자4	숫자5	최빈값
4	1	1	2	2	4	1
5	1	2	3	4	5	#N/A
6	4	4	2	2	0	4
7		15	1	15		15

- ① =MODE.SNGL(A4:E4) : [A4:E4] 영역의 숫자 중 가장 많이 발생한 값(최빈수)인 1이 [F4] 셀에 입력됩니다.
- ② =MODE.SNGL(A5:E5) : [A5:E5] 영역의 숫자들은 빈도수가 모두 같으므로 오류값(#N/A)이 [F5] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실험하세요.

구입수량에서 가장 높은 빈도를 가진 고객들의 구입총액 합계를 계산하세요.

정답 [E9] : =SUMIF(D3:D8, MODE.SNGL(D3:D8), E3:E8)

	A	B	C	D	E
1	고객별 구입 현황				
2	고객명	성별	등급	구입수량	구입총액
3	허영욱	남	골드	4	1,208,000
4	최주원	여	일반	9	2,214,000
5	이수학	남	골드	5	1,425,000
6	김신성	남	VIP	4	1,168,000
7	선기섭	남	일반	4	1,020,000
8	정신영	여	VIP	9	1,967,000
9	구입빈도 높은 고객의 구입총액 합계				3,396,000

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=SUMIF(D3:D8, MODE.SNGL(D3:D8), E3:E8)

①

- ① MODE.SNGL(D3:D8) : [D3:D8] 영역의 숫자 중 최빈수인 4를 반환합니다. 4를 ①에 대입하면 다음과 같습니다.
- =SUMIF(D3:D8, 4, E3:E8) : [D3:D8] 영역에서 4가 입력된 셀([D3], [D6:D7])을 찾은 후 [E3:E8] 영역의 같은 행([E3], [E6:E7])에 있는 구입총액들의 값(1,208,000, 1,168,000, 1,020,000)을 더한 3,396,000이 [E9] 셀에 입력됩니다.



전문가의 조언

SUMIF 함수는 조건이 적용될 범위에서 조건에 맞는 셀을 찾아 합계를 구할 범위 중 같은 행에 있는 값들의 합계를 구하는 함수입니다. 자세한 내용은 33쪽을 참고하세요.

075 N번째로 작은 수 찾기 — SMALL

24.상시, 23.상시, 22.상시, 21.상시, 19.상시, 15.3, 05.2, 04.4

SMALL 함수는 자료 범위에서 N번째로 작은 값을 반환하는 함수입니다. 이 함수를 사용하여 상대적인 순위에 해당하는 값을 선택할 수 있습니다.

형식 SMALL(범위, N번째) : '범위' 중 'N번째'로 작은 값을 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'SMALL.xlsm' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C	D	E	F
1	지정된 작은 수 찾기					
2						
3	숫자1	숫자2	숫자3			
4	12	21	18	제일 작은 수	2	①
5	5	2	19	두번째 작은 수	4	
6	7	16	18	세번째 작은 수	5	②
7	4	15	24			

① =SMALL(A4:C7, 1) : [A4:C7] 영역에서 첫 번째로 작은 수, 즉 가장 작은 수 2가 [F4] 셀에 입력됩니다.

② =SMALL(A4:C7, 3) : [A4:C7] 영역에서 세 번째로 작은 수인 5가 [F6] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

이용금액이 가장 작은 회원의 이용ID를 찾아 표시하세요.

정답 [A11] : =VLOOKUP(SMALL(C3:C8, 1), C3:D8, 2, FALSE)

	A	B	C	D
1	개나리 클럽 이용 현황			
2	회원번호	회원명	이용금액	이용ID
3	10202	김신우	98,000	ksh0202
4	10203	나영미	285,000	nymi1004
5	10204	우승현	365,000	wsh3702
6	10205	변정수	56,000	bjs-0808
7	10206	윤석현	154,000	wsh3402
8	10207	도연지	567,000	djy
9				
10	이용금액이 가장 작은 회원의 이용ID			
11	bjs-0808			

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=VLOOKUP(SMALL(C3:C8, 1), C3:D8, 2, FALSE)

①

- ① SMALL(C3:C8, 1) : [C3:C8] 영역에서 가장 작은 이용금액인 56,000을 반환합니다. 56,000을 ①에 대입하면 다음과 같습니다.
- =VLOOKUP(56,000, C3:D8, 2, FALSE)
 - ② [C3:D8] 영역의 맨 왼쪽 열에서 56,000을 찾습니다.
 - ③ 56,000이 있는 행에서 2열에 있는 값 "bjs-0808"을 찾아서 [B11] 셀에 입력합니다.

	C	D
2	이용금액	이용ID
3	98,000	ksh0202
4	285,000	nymi1004
5	365,000	wsh3702
6	56,000	bjs-0808
7	154,000	wsh3402
8	567,000	djy



전문가의 조언

VLOOKUP 함수는 범위로 정한 영역의 맨 왼쪽 열에서 특정 기준값으로 자료를 찾고, 그 자료가 속한 행 중에서 필요한 것이 있는 열의 위치를 지정하여 값을 반환하는 함수입니다. 자세한 내용은 45쪽을 참고하세요.

076 표준편차 계산하기 — STDEV.S

25.상시, 23.상시, 22.상시, 21.상시, 19.상시, 14.2, 13.상시

STDEV.S 함수는 인수로 주어진 숫자들의 표준편차를 계산하는 함수로, 인수는 1개에서 255개까지 지정할 수 있습니다. 인수에 텍스트, 논리값 또는 빈 셀이 포함되는 경우 그 값은 무시되지만 값이 0인 셀은 포함됩니다.

형식 STDEV.S(인수1, 인수2, ...): '인수'로 주어진 숫자들의 표준편차를 구합니다.

준비하세요! : '길벗컴퓨터2급통합\부록\함수사전' 폴더의 'STDEV.S.xlsx' 파일을 열어 '기본' 시트에서 실습하세요.

	A	B	C	D
1	표준 편차 계산			
2				
3	숫자1	숫자2	숫자3	표준편차
4	1	2	3	1.00
5	2	2	2	0.00
6	0	3	5	2.52
7	4		2	1.41

① =STDEV.S(A4:C4) : [A4:C4] 영역에 대한 표본 표준편차를 계산한 1.00이 [D4] 셀에 입력됩니다.

② =STDEV.S(A6:C6) : [A6:C6] 영역에 대한 표본 표준편차를 계산한 2.52가 [D6] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

생산물(B)의 표본 표준편차를 구하세요.

- ▶ 표준편차는 소수점 이하 두 번째 자리에서 내림하여 첫 번째 자리까지 표시 [표시 예 : 123.45 → 123.4]

정답 [D15] : =ROUNDDOWN(STDEV.S(C3:C14), 1)

	A	B	C	D
1	월별생산현황			
2	월	생산품(A)	생산품(B)	생산품(C)
3	1월	5,535	6,021	4,831
4	2월	5,468	6,871	5,001
5	3월	5,724	6,278	4,835
6	4월	5,689	6,389	4,297
7	5월	5,179	6,172	5,017
8	6월	5,348	6,008	4,983
9	7월	5,493	6,217	4,998
10	8월	5,157	6,397	4,328
11	9월	5,537	6,284	4,682
12	10월	5,399	6,316	4,179
13	11월	5,176	6,784	4,385
14	12월	5,697	6,418	4,267
15	생산품(B) 표준편차			261.6

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=ROUNDDOWN(STDEV.S(C3:C14), 1)

①

- ① STDEV.S(C3:C14) : [C3:C14] 영역의 표본 표준편차를 계산한 261.682...를 반환합니다.
- ROUNDDOWN(261.682..., 1) : 261.682...를 소수점 이하 둘째 자리에서 내림하여 소수점 이하 첫째 자리까지 표시한 261.60이 [D15] 셀에 입력됩니다.



전문가의 조언

ROUNDDOWN 함수는 숫자를 지정한 자릿수로 내림하여 표시하는 함수입니다. 자세한 내용은 41쪽을 참고하세요.

077 과목별 최대 가산점 구하기 — MAXA

MAXA 함수는 주어진 인수 내에서 가장 큰 값을 반환하는 함수로, 1~255개의 인수를 사용할 수 있습니다. MAX와 다른 점은 숫자는 물론, 빈 셀, 논리값(TRUE/FALSE), 숫자로 표시된 텍스트 등도 인수로 사용할 수 있다는 것입니다.

형식 MAXA(인수1, 인수2, ...): '인수' 중에서 가장 큰 값을 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'MAXA.xlsm' 파일을 열어 '기본' 시트에서 실험하세요.

MAXA 함수를 이용하여 최대 가산점을 계산해 보겠습니다.

	A	B	C	D	E
1		성적표			
2		※ 보훈 자녀는 TRUE로 표시하며, 가산점은 1점입니다!			
3		성명	학과	영어	수학
4		전현수	건축과	0.72	0.88
5		김명훈	기계과	0.55	0.7
6		하현호	경영과	0.98	0.8
7		강진성	기계과	0.76	0.94
8		박희선	건축과	TRUE	0.67
9		엄정희	건축과	0.84	0.9
10		최대 가산점		1	0.94

① =MAXA(D4:D9) : [D4:D9] 영역에서 가장 큰 값인 1(TRUE)이 [D10] 셀에 입력됩니다.

② =MAXA(E4:E9) : [E4:E9] 영역에서 가장 큰 값인 0.94가 [E10] 셀에 입력됩니다.

078 과목별 최소 가산점 구하기 — MINA

MINA 함수는 주어진 인수 내에서 가장 작은 값을 반환하는 함수로, 1~255개의 인수를 사용할 수 있습니다. MIN과 다른 점은 숫자는 물론, 빈 셀, 논리값(TRUE/FALSE), 숫자로 표시된 텍스트 등도 인수로 사용할 수 있다는 것입니다.

형식 MINA(인수1, 인수2, ...): '인수' 중에서 가장 작은 값을 반환합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'MINA.xlsm' 파일을 열어 '기본' 시트에서 실험하세요.

MINA 함수를 이용하여 최소 가산점을 계산해 보겠습니다.

	A	B	C	D	E
1		성적표			
2		※ 보훈 자녀는 TRUE로 표시하며, 가산점은 1점입니다!			
3		성명	학과	영어	수학
4		전현수	건축과	0.72	0.88
5		김명훈	기계과	0.55	0.7
6		하현호	경영과	0.98	0.8
7		강진성	기계과	0.76	0.94
8		박희선	건축과	TRUE	0.67
9		엄정희	건축과	0.84	0.9
10		최소 가산점		0.55	0.67

① =MINA(D4:D9) : [D4:D9] 영역에서 가장 작은 값인 0.55가 [D10] 셀에 입력됩니다.

② =MINA(E4:E9) : [E4:E9] 영역에서 가장 작은 값인 0.67이 [E10] 셀에 입력됩니다.

079 부서별 기본급의 평균 계산하기 — AVERAGEIF

AVERAGEIF 함수는 많은 자료 중에서 지정한 조건에 맞는 데이터만 찾아서 평균을 구하는 함수입니다. 찾을 조건이 있는 범위에서 조건에 맞는 데이터를 찾아 평균을 계산할 범위 중 같은 행에 있는 값들의 평균을 계산합니다.

형식 AVERAGEIF(조건이 적용될 범위, 조건, 평균을 구할 범위) : '조건이 적용될 범위'에서 '조건'에 맞는 셀을 찾아 '평균을 구할 범위' 중 같은 행에 있는 값들의 평균을 계산합니다.

준비하세요! : '갈벗컴월2급통합\부록\함수사전' 폴더의 'AVERAGEIF.xsm' 파일을 열어 '기본' 시트에서 실습하세요.

AVERAGEIF 함수를 이용하여 부서별 기본급의 평균을 계산해 보겠습니다.

	A	B	C	D	E	F	G
1		기본급 지급 현황			부서별 기본급의 평균		
2							
3		성명	부서	기본급	부서	평균	
4		이승연	판매부	1,450,000	판매부	1,387,500	①
5		김경수	기획부	1,350,000	기획부	1,300,000	②
6		이학봉	판매부	1,350,000			
7		지순녀	기획부	1,200,000			
8		김지연	판매부	1,450,000			
9		박원래	기획부	1,450,000			
10		최지은	기획부	1,200,000			
11		강유리	판매부	1,300,000			

① =AVERAGEIF(C4:C11, "판매부", D4:D11) : [C4:C11] 영역에서 "판매부"가 입력된 셀들을 찾아, [D4:D11] 영역의 같은 행들에 있는 기본급(1,450,000, 1,350,000, 1,450,000, 1,300,000)의 평균인 1,387,500이 [G4] 셀에 입력됩니다.

② =AVERAGEIF(C4:C11, "기획부", D4:D11) : [C4:C11] 영역에서 "기획부"가 입력된 셀들을 찾아, [D4:D11] 영역의 같은 행들에 있는 기본급(1,350,000, 1,200,000, 1,450,000, 1,200,000)의 평균인 1,300,000이 [G5] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실습하세요.

남학생들의 영어 평균을 계산하여 표시하세요.

▶ 평균 점수는 반올림없이 정수로 표시

정답 [F9] : =TRUNC(AVERAGEIF(C3:C8, "남", E3:E8))

	A	B	C	D	E	F
1		중간고사 성적표				
2		성명	성별	국어	영어	수학
3		변지서	여	80	85	86
4		최서정	여	95	92	94
5		박승완	남	75	78	80
6		김석훈	남	89	80	86
7		박상아	여	90	93	92
8		이재준	남	87	83	80
9		남학생 영어 평균점수				
10						80

수식의 이해

중첩 함수가 사용된 수식을 만들 때는 최종적으로 값을 반환하는, 즉 가장 바깥쪽에 사용할 함수부터 찾아서 수식을 세우고 수식을 이해할 때는 우선순위에 따라 안쪽에서부터 바깥쪽 방향으로 하나씩 상수로 변환하면서 이해하면 쉽습니다.

=TRUNC(AVERAGEIF(C3:C8, "남", E3:E8))

①

- ① AVERAGEIF(C3:C8, "남", E3:E8) : [C3:C8] 영역에서 "남"이 입력된 셀들([C5:C6], [C8])을 찾고, [E3:E8] 영역에서 "남"이 입력된 셀과 같은 행([E5:E6], [E8])에 있는 영어(78, 80, 83)의 평균 80.333...을 반환합니다. 80.333...을 ①에 대입하면 다음과 같습니다.
- =TRUNC(80.333...) : 80.333...을 반올림 없이 정수로 표시한 80이 [F9] 셀에 표시됩니다.



전문가의 조언

TRUNC 함수는 인수에 대하여 지정한 자릿수 미만의 수치를 버리는 함수입니다. 자세한 내용은 39쪽을 참고하세요.

080 부서별 직급별 기본급의 평균 계산하기 — AVERAGEIFS

AVERAGEIFS 함수는 여러 개의 조건에 맞는 자료의 평균을 구하는 함수입니다. 예를 들면 부서가 판매부이고, 급수가 1급이고, 남자인 사람들의 기본급의 평균을 구할 수 있습니다. 조건은 최대 127개까지 지정할 수 있습니다.

형식 AVERAGEIFS(평균을 구할 범위, 첫 번째 조건이 적용될 범위, 첫 번째 조건, 두 번째 조건이 적용될 범위, 두 번째 조건, ...) : 여러 개의 '조건이 적용될 범위'에서 여러 개의 '조건'에 맞는 셀을 찾아 '평균을 구할 범위' 중 같은 행에 있는 값들의 평균을 계산합니다.

준비하세요! : '길벗컴활2급통합\부록\함수사전' 폴더의 'AVERAGEIFS.xlsm' 파일을 열어 '기본' 시트에서 실행하세요.

AVERAGEIFS 함수를 이용하여 부서별 직급별 기본급의 평균을 계산해 보겠습니다.

	A	B	C	D	E	F	G	H	I
1		기본급 지급 현황				부서별 직급별 기본급의 평균			
2									
3		성명	부서	직급	기본급	부서	직급	1급	2급
4		이승연	판매부	1급	1,450,000	판매부	1급	1,450,000	1,325,000
5		김경수	기획부	2급	1,350,000	기획부	2급	1,325,000	1,275,000
6		1 왕봉	판매부	2급	1,350,000				
7		지은녀	기획부	2급	1,200,000				
8		김재현	판매부	1급	1,450,000				
9		박원래	기획부	1급	1,450,000				
10		최지은	기획부	1급	1,200,000				
11		강유라	판매부	2급	1,300,000				

① =AVERAGEIFS(E4:E11, C4:C11, "판매부", D4:D11, "1급") : [C4:C11] 영역에서 "판매부"가 입력된 셀들을 찾고, [D4:D11] 영역에서 같은 행들에 있는 "1급"이 입력된 셀들을 찾아 [E4:E11] 영역의 같은 행들에 있는 기본급(1,450,000, 1,450,000)의 평균인 1,450,000이 [H4] 셀에 입력됩니다.

② =AVERAGEIFS(E4:E11, C4:C11, "기획부", D4:D11, "2급") : [C4:C11] 영역에서 "기획부"가 입력된 셀들을 찾고, [D4:D11] 영역에서 같은 행들에 있는 "2급"이 입력된 셀들을 찾아 [E4:E11] 영역의 같은 행들에 있는 기본급(1,350,000, 1,200,000)의 평균인 1,275,000이 [I5] 셀에 입력됩니다.

기출문제 따라잡기

'기출' 시트에서 실행하세요.

학과가 "생물"이 아니면서 전공이 30 이상인 학생들의 총점 평균을 계산하여 표시하세요.

▶ 평균 점수는 소수점 이하 둘째 자리에서 반올림하여 첫째 자리까지 표시

정답 [F13] : =ROUND(AVERAGEIFS(F3:F12, C3:C12, "<>생물", D3:D12, ">=30"), 1)

	A	B	C	D	E	F
1		중간고사 성적표				
2		성명	학과	전공	교양	총점
3		임소희	생물	48	46	94
4		태우식	조경	45	40	85
5		서기운	환경	44	39	83
6		한재연	조경	28	33	61
7		김부성	생물	40	35	75
8		최성완	환경	18	26	44
9		김승현	생물	27	29	56
10		이수민	조경	25	31	56
11		김성은	생물	33	35	68
12		장태일	환경	40	37	77
13		조경/환경학과의 전공 우수자 총점 평균				
14						81.7

수식의 이해

=ROUND(AVERAGEIFS(F3:F12, C3:C12, "<>생물", D3:D12, ">=30"), 1)

①

• ① AVERAGEIFS(F3:F12, C3:C12, "<>생물", D3:D12, ">=30") : [C3:C12] 영역에서 "생물"이 입력되지 않은 셀들([C4:C6], [C8], [C10], [C12])을 찾고, [D3:D12] 영역에서 "생물"이 입력되지 않은 셀과 같은 행들에 있는 셀 중 30 이상인 셀들([D4:D5], [D12])을 찾고, [F3:F12] 영역에서 "생물"이 입력되지 않고 전공이 30 이상인 셀과 같은 행([F4:F5], [F12])에 있는 총점(85, 83, 77)의 평균 81.666...을 반환합니다. 81.666...을 ①에 대입하면 다음과 같습니다.

• =ROUND(81.666..., 1) : 81.666...을 소수점 이하 두 번째 자리에서 반올림하여 첫 번째 자리까지 표시한 81.7이 [F13] 셀에 표시됩니다.



전문가의 조언

ROUND 함수는 숫자를 지정한 자리수로 반올림하여 표시하는 함수입니다. 자세한 내용은 36쪽을 참고하세요.



A series of horizontal dashed lines for writing, spanning the width of the page.

계산작업 문제 모음

컴퓨터활용능력 2급 실기

	A	B	C	D	E	F	G	H	I	J	K	L
1	[표1]	계열 학과별 취업률					[표2]	등록금 실납입금액				
2	계열	학과	경쟁률	재학생	취업률		학번	이름	학점	등록금	실납입금액	
3	공학계열	컴퓨터학과	2.75	275	4.75		24020352	박지영	2.87	3,580,000	3,580,000	
4	자연계열	생명과학과	1.83	239	3.81		24020967	조재성	4.12	3,600,000	720,000	
5	자연계열	식품조리과	2.14	302	2.94		23080108	김민서	3.76	4,150,000	1,660,000	
6	공학계열	로봇공학과	3.23	261	5.62		24020135	이중희	4.30	3,860,000	0	
7	공학계열	IT융합학과	1.98	225	3.27		23080579	신영숙	2.53	4,270,000	4,270,000	
8	자연계열	대기과학과	2.43	195	4.53		24020251	최경민	3.45	3,750,000	3,750,000	
9	공학계열	신소재학과	3.06	238	2.76		24020608	김경아	3.89	3,670,000	1,468,000	
10	계열	공학계열에서 취업률이 가장 높은 학과명				로봇공학과	23080774	한정수	4.26	4,590,000	918,000	
11	공학계열											
12							<학점순위별 장학금지급률>					
13							순위	1	2	4	6	
14							장학금지급률	100%	80%	60%	0%	
15												
16	[표3]	국가별 수출/수입 현황					[표4]	성적현황				
17	국가명	수출액	수입액	수출현황			이름	1과목	2과목	3과목	4과목	비고
18	프랑스	43,252,500	36,765,300				이하늘	3.47	3.18	4.05	3.82	중급수업대상
19	일본	63,824,100	54,551,000	주요수출국			박정훈	2.99	3.15	2.68	2.71	초급수업대상
20	캐나다	34,280,000	41,045,900				엄지영	3.92	4.52	4.17	4.27	
21	중국	75,360,200	65,657,000	주요수출국			김동준	3.84	3.51	3.36	3.62	중급수업대상
22	미국	84,822,400	97,853,300	주요수출국			유서하	4.68	4.76	4.83	4.89	
23	스페인	52,461,000	44,592,000				황성철	1.95	1.52	1.28	1.63	기초수업대상
24	러시아	69,021,000	83,487,400	주요수출국			윤미진	4.25	4.22	4.05	4.16	
25	브라질	49,240,000	58,104,900				김청옥	2.26	2.54	2.38	2.43	초급수업대상
26												
27	[표5]	도서 분류										
28	분류코드	도서명	입고일	출판사	저자	가격						
29	3333	불법	2024-03-05	나무서재	이수현	16,900						
30	1111	회장의생각	2024-03-07	경제북스	박서진	24,500						
31	2222	모든편의점	2024-03-10	한영사	문영철	18,300						
32	3333	작별하는것	2024-03-13	북산책	장시원	15,200						
33	1111	토탈브레인	2024-03-18	베타미디어	선우진	19,000						
34	1111	리더의질문	2024-03-24	경영하우스	유인명	21,600						
35	2222	인기팡팡	2024-03-27	미래아이	현진원	17,500						
36	가장 빈도가 높은 분류코드 도서의 개수					3권						
37												

1. [표1]에서 계열[A3:A9]이 “공학계열”인 학과[B3:B9] 중 취업률[E3:E9]이 가장 높은 학과를 찾아 [E10] 셀에 표시하시오. (8점)

- ▶ 조건은 [A10:A11] 영역에 입력하시오.
- ▶ INDEX, MATCH, DMAX 함수 사용

[]

2. [표2]에서 등록금[J3:J10]과 학점[I3:I10], 학점순위별 장학금지급률[H13:K14]을 이용하여 학생들의 실납입금액[K3:K10]을 계산하시오. (8점)

- ▶ 실납입금액 = 등록금 × (1 - 장학금지급률)
- ▶ 장학금지급률은 학점을 기준으로 한 순위가 1위면 100%, 2~3위면 80%, 4~5위면 60%, 그 외에는 0%로 계산(순위는 학점이 가장 높은 것이 1위)

▶ HLOOKUP, RANK, EQ 함수 사용

[

]

3. [표3]에서 국가명[A18:A25]별로 수출액[B18:B25]이 수출액 중간값 이상이면 “주요수출국”을, 그 외에는 공백을 수출현황[D18:D25]에 표시하십시오. (8점)

▶ IF, AVERAGE, MEDIAN 함수 중 알맞은 함수들을 사용

[

]

4. [표4]에서 이름별로 1~4과목[H18:K25]의 평점이 1이면 “기초수업대상”, 2면 “초급수업대상”, 3이면 “중급수업대상”, 4면 공백을 비교[L18:L25]에 표시하십시오. (8점)

▶ 평점은 1~4과목의 평균을 구하여 소수점 이하를 버린 정수로 계산

▶ CHOOSE, AVERAGE, INT 함수 사용

[

]

5. [표5]에서 분류코드[A29:A35]의 빈도가 가장 높은 도서의 개수를 [F36] 셀에 계산하십시오. (8점)

▶ 계산된 도서수 뒤에는 “권”을 포함하여 표시 [표시 예 : 5권]

▶ COUNTIF, MODE, SNGL 함수와 & 연산자 사용

[

]

수식의 이해

1. [E10] : =INDEX(A3:E9, MATCH(DMAX(A2:E9, 5, A10:A11), E3:E9, 0), 2)

2. [K3] : =J3 * (1 - HLOOKUP(RANK.EQ(I3, \$I\$3:\$I\$10), \$H\$13:\$K\$14, 2))

3. [D18] : =IF(B18>MEDIAN(\$B\$18:\$B\$25), “주요수출국”, “ ”)

4. [L18] : =CHOOSE(INT(AVERAGE(H18:K18)), “기초수업대상”, “초급수업대상”, “중급수업대상”, “ ”)

5. [F36] : =COUNTIF(A29:A35, MODE.SNGL(A29:A35)) & 권)

	A	B	C	D	E	F	G	H	I	J	
1	[표1]	성적현황					[표2]	국가대표 선발전			
2	이름	국어	영어	수학	평균		이름	나이	누적점수	결과	
3	한태영	82.6	85.5	84.7	84.3		황현진	24	399	상비군	
4	정경호	88.5	78.4	80.5	82.5		유인원	28	451	국가대표	
5	이세영	82.6	90.8	92.5	88.6		송재우	31	394		
6	최시아	84.5	85.5	80.5	83.5		박정원	25	433	국가대표	
7	고승기	82.6	78.4	82.4	81.1		고종완	29	418	상비군	
8	임보미	81.9	85.5	84.7	84.0		손준혁	24	380		
9	김효주	93.4	94.2	92.5	93.4		박연욱	22	425	상비군	
10	유일한	88.5	90.8	93.6	91.0		전명식	30	427	국가대표	
11	영어 빈도가 가장 높은 점수				85		노경은	26	386		
12											
13	[표3]	영화관람현황					[표4]	회원관리현황			
14	영화관람일	장르	결제방법	결제비용			회원명	가입년도	아이디	닉네임	회원코드
15	03월04일	액션	카드	13000			홍해인	2020	hihi3322	david	HIH-20-DA
16	03월08일	드라마	현금	11000			고수정	2024	ident97	naomi	IDE-24-NA
17	04월09일	SF	현금	13000			문규명	2019	ancon11	stella	ANC-19-ST
18	04월15일	액션	카드	13000			이조아	2022	sns486	victoria	SNS-22-VI
19	05월15일	액션	카드	11000			김수철	2021	gofind25	alice	GOF-21-AL
20	06월02일	SF	카드	12000			정서빈	2023	kes0823	rania	KES-23-RA
21	06월13일	드라마	현금	12000			나상환	2021	chang02	daniel	CHA-21-DA
22							임시연	2022	infj0917	gilbert	INF-22-GI
23	카드	57					노성민	2020	gksk105	conrad	GKS-20-CO
24	현금	42					허은아	2023	lsh1224	enoch	LSH-23-EN
25											
26	[표5]	전기사용량									
27	호	사용량	청구량			<할인율표>					
28	101호	268	246.5			사용량		할인율			
29	102호	135	128.2			0 이상		100 미만	2%		
30	103호	422	358.7			100 이상		200 미만	5%		
31	201호	92	90.1			200 이상		300 미만	8%		
32	202호	371	326.4			300 이상		400 미만	12%		
33	203호	293	269.5			400 이상			15%		
34	301호	409	347.6								
35	302호	323	284.2								
36	303호	256	235.5								
37											

1. [표1]의 영어[C3:C10] 점수 중 빈도가 가장 높은 점수를 정수로 [E11] 셀에 계산하시오. (8점)

▶ INT, MODE, SNGL 함수 사용

[

]

2. [표2]에서 누적점수[I3:I11]에 대한 순위를 구하여 1~3위는 “국가대표”, 4~6위는 “상비군”, 그 외에는 공백을 결과 [J3:J11]에 표시하십시오. (8점)

- ▶ 순위는 누적점수가 가장 높은 것이 1위
- ▶ IF, RANK, EQ 함수 사용

[]

3. [표3]에서 결제방법[C15:C21]과 결제비용[D15:D21]을 이용하여 결제방법별 결제비율을 [B23:B24] 영역에 계산하십시오. (8점)

- ▶ 결제방법별 결제비율 = 각 결제방법별 결제비용의 합계 / 전체 결제비용의 합계 × 100
- ▶ 계산된 결제비율은 소수점 이하 자리의 숫자를 버리고 정수로 표시
- ▶ SUMIF, SUM, TRUNC 함수 사용

[]

4. [표4]에서 아이디[H15:H24]의 앞 3글자, 가입년도[G15:G24]의 마지막 2글자, 닉네임[I15:I24]의 앞 2글자를 이용하여 회원코드[J15:J24]를 표시하되, 영문은 대문자로 표시하십시오. (8점)

- ▶ 표시 예 : 아이디가 “angel77”, 가입년도가 2024, 닉네임이 “zena”인 경우 “ANG-24-ZE”로 표시
- ▶ UPPER, LEFT, RIGHT 함수와 & 연산자 사용

[]

5. [표5]에서 사용량[B28:B36]과 할인율표[E29:G33]를 이용하여 청구량[C28:C36]을 계산하십시오. (8점)

- ▶ 청구량 = 사용량 × (1 - 할인율)
- ▶ 청구량은 소수점 이하 둘째 자리에서 내림하여 첫째 자리까지 표시 [표시 예 : 345.67 → 345.6]
- ▶ HLOOKUP, VLOOKUP, ROUND, ROUNDUP, ROUNDDOWN 함수 중 알맞은 함수들을 선택하여 사용

[]

수식의 이해

1. [E11] : =INT(MODE,SNGL(C3:C10))

2. [J3] : =IF(RANK.EQ(I3, \$I\$3:\$I\$11)<=3, “국가대표”, IF(RANK.EQ(I3, \$I\$3:\$I\$11)<=6, “상비군”, “ ”))

3. [B23] : =TRUNC(SUMIF(\$C\$15:\$C\$21, A23, \$D\$15:\$D\$21) / SUM(\$D\$15:\$D\$21) * 100)

4. [J15] : =UPPER(LEFT(H15, 3) & “-” & RIGHT(G15, 2) & “-” & LEFT(I15, 2))

5. [C28] : =ROUNDDOWN(B28 * (1 - VLOOKUP(B28, \$E\$29:\$G\$33, 3)), 1)

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	[표1]	필기시험결과					[표2]	심사결과					
2	수험번호	1과목	2과목	평균	결과		이름	심사점수1	심사점수2	심사점수3	심사점수4	심사점수5	평균
3	423-001	67	71	69	합격		한명희	84	86	88	90	79	86
4	423-002	88	86	87	합격		이혜영	91	90	95	94	92	92.33333
5	423-003	87	38	63	불합격		조경화	73	71	77	75	74	74
6	423-004	92	95	94	합격		우정승	89	86	88	88	85	87.33333
7	423-005	93	90	92	합격		박훈규	93	91	94	96	97	94.33333
8	423-006	46	50	48	불합격		김은소	68	71	72	69	75	70.66667
9	423-007	87	90	89	합격		지승대	88	90	91	87	88	88.66667
10	423-008	55	63	59	불합격		박해수	94	93	91	90	92	92
11	423-009	92	89	91	합격		김종모	87	92	90	91	93	91
12													
13	[표3]	사원관리현황			기준일 : 2025-04-07		[표4]	지역별 매출현황 (단위 : 만원)					
14	사원코드	사원명	부서명	직위	입사년차		지역	지점	직원수	매출액			
15	M1406L	최성완	경영지원부	과장	11년차		서울	노원점	12	149,000			
16	M2002N	서기운	경영지원부	대리	5년차		경기	용인점	15	96,100			
17	P0811S	김윤지	기획부	부장	17년차		서울	마포점	10	136,200			
18	P2302Y	이부성	기획부	사원	2년차		경기	성남점	13	152,000			
19	D0903T	박윤희	생산부	부장	16년차		인천	계양점	14	84,300			
20	D1904H	김지선	생산부	대리	6년차		인천	연수점	12	124,000			
21	D2208F	정원석	생산부	사원	3년차		경기	수원점	16	112,500			
22	S1505A	강민호	영업부	과장	10년차		경기	안산점	15	81,300			
23	S1910E	양정화	영업부	대리	6년차		인천	미추홀점	12	57,900			
24	S2412P	윤성철	영업부	사원	1년차		서울	서초점	14	168,200			
25													
26	[표5]	환자진료기록					지역	경기 최대-최소 매출액 차이					
27	진료일자	환자명	주민등록번호	진료과	성별		경기	70,700					
28	05월02일	김영택	861123-1*****	내과	남								
29	05월07일	강정수	970203-2*****	외과	여								
30	05월09일	김나영	120420-4*****	안과	여								
31	05월13일	서울한	890804-5*****	신경외과									
32	05월15일	주진명	090820-4*****	소아과	여								
33	05월20일	오인영	911201-2*****	외과	여								
34	05월22일	한서라	030317-3*****	내과	남								
35	05월26일	강사을	110425-3*****	소아과	남								
36	05월27일	박하은	881012-6*****	내과									
37	05월30일	이태윤	820927-1*****	신경외과	남								
38													

1. [표1]에서 1과목[B3:B11]과 2과목[C3:C11]이 모두 40점 이상이면서 평균[D3:D11]이 60점 이상이면 “합격”을, 그 외에는 “불합격”을 결과[E3:E11]에 표시하시오. (8점)

▶ IF, COUNTIF, AND 함수 사용

[]

2. [표2]에서 이름별로 심사점수1~5[H3:L11] 중 가장 큰 점수와 가장 작은 점수를 제외한 나머지 점수들의 평균[M3:M11]을 계산하시오. (8점)

▶ AVERAGEIFS, MAX, MIN 함수와 & 연산자 사용

[]

3. [표3]에서 기준일[E13]과 사원코드[A15:A24]를 이용하여 입사년차[E15:E24]를 계산하시오. (8점)

- ▶ 입사년차 = 기준일의 년도 - (2000 + 입사년도)
- ▶ 입사년도는 사원코드의 2, 3번째 문자임
- ▶ 계산된 입사년차 뒤에 “년차”를 포함하여 표시 [표시 예 : 10년차]
- ▶ YEAR, MID 함수와 & 연산자 사용

[

]

4. [표4]에서 지역[G15:G24]이 “경기”인 지점들의 매출액[J15:J24] 최대값과 최소값의 차이를 [H27] 셀에 계산하시오. (8점)

- ▶ 조건은 [G26:G27] 영역에 입력하시오.
- ▶ DMAX, DMIN 함수 사용

[

]

5. [표5]에서 주민등록번호[C28:C37]의 8번째 자리가 1이나 3이면 “남”을, 2나 4면 “여”를, 그 외에는 공백을 성별 [E28:E37]에 표시하시오. (8점)

- ▶ IFERROR, CHOOSE, MID 함수 사용

[

]

수식의 이해

1. [E3] : =IF(AND(COUNTIF(B3:C3, ">=40")=2, D3>=60), “합격”, “불합격”)
2. [M3] : =AVERAGEIFS(H3:L3, H3:L3, “<”&MAX(H3:L3), H3:L3, “<”&MIN(H3:L3))
3. [E15] : =YEAR(\$E\$13) - (2000 + MID(A15, 2, 2)) & “년차”
4. [H27] : =DMAX(G14:J24, 4, G26:G27) - DMIN(G14:J24, 4, G26:G27)
5. [E28] : =IFERROR(CHOOSE(MID(C28, 8, 1), “남”, “여”, “남”, “여”, “ ”)

	A	B	C	D	E	F	G	H	I	J	K
1	[표1]	쇼핑몰 판매현황					[표2]	성적현황			
2	상품코드	판매가	판매량	총판매액	결과		학생명	학과	전공	교양	출석
3	S3472	36,500	967	35,295,500			조동원	기계공학과	79	89	8
4	T3348	42,000	855	35,910,000			김시진	건축공학과	82	78	9
5	R4930	38,800	1,211	46,986,800	★★		신지민	도시공학과	94	95	10
6	S8645	32,600	759	24,743,400			고인숙	건축공학과	93	92	10
7	T6941	41,500	1,036	42,994,000	★		남윤후	기계공학과	85	88	9
8	R3057	40,000	875	35,000,000			박준성	도시공학과	67	62	7
9	R4996	39,000	1,365	53,235,000	★★★		이인하	건축공학과	76	74	8
10	S1935	34,500	684	23,598,000			임선호	기계공학과	91	93	10
11	T5377	37,600	799	30,042,400			한동우	도시공학과	98	97	10
12											
13	[표3]	상품판매현황					학과	건축공학과 학생들의 전공 평균			
14	지점	관리자	상품코드	판매량	총판매액		건축공학과	83.7			
15	강북	윤은서	N-124-B	18	612,000						
16	강남	손영심	S-924-A	16	432,000						
17	강남	주건후	S-505-B	21	588,000						
18	강북	강시윤	N-688-A	26	780,000						
19	강북	장준성	N-395-A	19	570,000						
20	강남	안승현	S-331-A	15	405,000						
21	강남	서보인	S-805-B	20	560,000		<상품단가표>				
22	강북	조민성	N-770-B	16	544,000		지점명	강남A	강남B	강북A	강북B
23	강남	양심선	S-369-A	22	594,000		출고가	25,000	24,000	28,000	32,000
24	강북	우연희	N-613-B	17	578,000		판매가	27,000	28,000	30,000	34,000
25											
26	[표4]	자격시험결과					[표5]	신체검사결과			
27	이름	1과목	2과목	3과목	총점		이름	성별	신장(m)	체중(kg)	비만도
28	곽혜은	84	86	88	258		황찬수	남	1.76	67	표준
29	홍정훈	96	92	94	282		신민서	여	1.63	60	표준
30	양찬호	85	81	84	250		조현우	남	1.69	73	비만
31	문가영	77	79	74	230		온우섭	남	1.71	60	표준
32	강하늘	62	61	55	178		이향지	여	1.57	48	저체중
33	안서연	59	55	61	175		한우리	여	1.65	52	저체중
34	허은지	84	86	83	253		김시환	남	1.72	61	표준
35							강현민	남	1.81	85	비만
36			총점의 평균(표준편차)				엄미윤	여	1.62	55	표준
37			232(41)				김예소	여	1.58	63	비만
38											

1. [표1]에서 총판매액[D3:D11]을 기준으로 순위를 구하여 1위는 “★★★”, 2위는 “★★”, 3위는 “★”, 그 외에는 공백을 결과[E3:E11]에 표시하시오. (8점)

- ▶ 순위는 총판매액이 가장 많은 것이 1위임
- ▶ IF, RANK, EQ, CHOOSE 함수 사용

[]

2. [표2]에서 학과[H3:H11]가 “건축공학과”인 학생들의 전공[I3:I11] 평균을 [H14] 셀에 계산하시오. (8점)

- ▶ 조건은 [G13:G14] 영역에 입력하시오.
- ▶ 전공 평균은 소수점 이하 둘째 자리에서 올림하여 첫째 자리까지 표시 [표시 예 : 87.43 → 87.5]
- ▶ DAVERAGE 함수와 ROUND, ROUNDUP, ROUNDDOWN 중 알맞은 함수를 선택하여 사용

[]

3. [표3]에서 지점[A15:A24]과 상품코드[C15:C24], 상품단가표[H22:K24]를 이용하여 총판매액[E15:E24]을 계산하시오. (8점)

- ▶ 총판매액 = 판매량 × 판매가
- ▶ 상품단가표의 지점명은 [표3]의 지점에 상품코드의 마지막 문자를 붙여 표시한 것임
- ▶ HLOOKUP, RIGHT 함수와 & 연산자 사용

[]

4. [표4]에서 총점[E28:E34]의 평균과 표준편차를 [C37] 셀에 계산하시오. (8점)

- ▶ 총점의 평균과 표준편차는 반올림 없이 정수로 표시
[표시 예 : 평균이 87.654이고, 표준편차가 12.345인 경우 → 87(12)]
- ▶ TRUNC, AVERAGE, STDEV.S 함수와 & 연산자 사용

[]

5. [표5]에서 신장[I28:I37]과 체중[J28:J37]을 이용하여 비만도[K28:K37]를 계산하시오. (8점)

- ▶ 비만도 = 체중 ÷ 신장²
- ▶ 비만도가 20 미만이면 “저체중”, 20 이상 25 미만이면 “표준”, 그 외에는 “비만”을 표시
- ▶ IF, POWER 함수 사용

[]

수식의 이해

1. [E3] : =IF(RANK.EQ(D3, \$D\$3:\$D\$11)<=3, CHOOSE(RANK.EQ(D3, \$D\$3:\$D\$11), “★★★★”, “★★★”, “★★”, “★”), “ ”)

2. [H14] : =ROUNDUP(DAVERAGE(G2:K11, 3, G13:G14), 1)

3. [E15] : =D15 * HLOOKUP(A15&RIGHT(C15, 1), \$H\$22:\$K\$24, 3, FALSE)

4. [C37] : =TRUNC(AVERAGE(E28:E34)) & “(” & TRUNC(STDEV.S(E28:E34)) & “)”

5. [K28] : =IF(J28/POWER(I28, 2)<20, “저체중”, IF(J28/POWER(I28, 2)<25, “표준”, “비만”))

	A	B	C	D	E	F	G	H	I	J	K	L
1	[표1]	사원별 판매 실적					[표2]	K리그 순위				
2	사원코드	부서	4월	5월	6월		구단명	승	무	패	승점	최종결과
3	DS-351	영업1팀	2,534	2,463	2,954		서울FC	15	11	12	56	
4	DS-932	영업2팀	5,381	5,071	4,866		강원FC	14	8	16	50	
5	DS-667	영업1팀	1,967	3,549	2,672		인천FC	9	10	19	37	
6	DS-804	영업2팀	2,648	2,786	3,078		전북FC	22	13	3	79	우승
7	DS-272	영업1팀	4,259	4,862	5,037		상주FC	16	7	15	55	
8	DS-683	영업2팀	3,809	3,793	3,945		수원FC	12	12	14	48	
9	DS-550	영업2팀	1,661	2,158	1,998		포항FC	16	8	14	56	
10	DS-964	영업1팀	3,940	3,704	3,513		울산FC	22	10	6	76	준우승
11	영업1팀 4월, 5월 평균 실적 차이				470		대구FC	13	16	9	55	
12												
13	[표3]	신입사원 관리			<부서코드표>		[표4]	제품출고 현황				
14	사원명	임시코드	부서명		코드	부서명	제품	구분코드	출고가	출고량	출고일자	
15	전윤주	p58482	PLANNING		p	planning	원피스	33	34,400	200	4월2일	
16	이아현	f40175	FINANCE		l	logistics	구두	11	68,200	120	4월2일	
17	박대인	l93647	LOGISTICS		s	sales	스커트	33	22,800	240	4월2일	
18	문정은	f38466	FINANCE		f	finance	모자	22	15,000	100	4월8일	
19	조수연	s75801	SALES				운동화	11	31,000	300	4월8일	
20	양이윤	p20079	PLANNING				자갑	22	29,400	140	4월8일	
21	신승윤	l10663	LOGISTICS				청바지	33	30,100	350	4월8일	
22	안범준	s67168	SALES				면바지	33	27,600	320	4월16일	
23	김은소	p82011	PLANNING				벨트	22	12,800	170	4월16일	
24							스니커즈	11	25,400	130	4월16일	
25	[표5]	일기예보 현황					빈도가 가장 높은 제품 수				4개	
26	날짜	최저	최고	강수량	나들이							
27	2024-04-05	12	17	0								
28	2024-04-06	8	14	5								
29	2024-04-07	4	12	15								
30	2024-04-12	9	17	10								
31	2024-04-13	10	18	0								
32	2024-04-14	9	16	0	적합							
33	2024-04-19	8	17	20								
34	2024-04-20	12	25	0								
35	2024-04-21	11	24	0	적합							

1. [표1]에서 부서[B3:B10]가 “영업1팀”인 사원들의 4월[C3:C10] 실적 평균과 5월[D3:D10] 실적 평균의 차이를 절대값으로 [E1] 셀에 계산하시오. (8점)

▶ ABS, AVERAGEIF 함수 사용

[

]

2. [표2]에서 승점[K3:K11]에 대한 순위를 구하여 1위는 “우승”, 2위는 “준우승”, 그 외에는 공백을 최종결과[L3:L11]에 표시하시오. (8점)

▶ 순위는 승점이 가장 높은 것이 1위

▶ IFERROR, CHOOSE, RANK, EQ 함수 사용

[

]

3. [표3]에서 임시코드[B15:B23]의 첫 번째 글자와 부서코드표[E15:F18]를 이용하여 각 사원들의 부서명[C15:C23]을 표시하시오. (8점)

- ▶ 부서명은 모두 대문자로 표시 [표시 예 : p12345 → PLANNING]
- ▶ VLOOKUP, LEFT, UPPER 함수 사용

[

]

4. [표4]에서 구분코드[I15:I24]의 빈도가 가장 높은 제품의 개수를 [L25] 셀에 계산하시오. (8점)

- ▶ 계산된 제품의 개수 뒤에는 “개”를 포함하여 표시 [표시 예 : 2개]
- ▶ COUNTIF, MODE, SNGL 함수와 & 연산자 사용

[

]

5. [표5]에서 날짜[A27:A35]의 요일이 “일요일”이면서 강수량[D27:D35]이 0이면 “적합”을, 그 외에는 공백을 나들이 [E27:E35]에 표시하시오. (8점)

- ▶ 단, 요일 계산 시 일요일이 1인 유형으로 지정
- ▶ IF, AND, WEEKDAY 함수 사용

[

]

수식의 이해

1. [E11] : =ABS(AVERAGEIF(B3:B10, “영업1팀”, C3:C10) – AVERAGEIF(B3:B10, “영업1팀”, D3:D10))

2. [L3] : =IFERROR(CHOOSE(RANK.EQ(K3, \$K\$3:\$K\$11), “우승”, “준우승”), “ ”)

3. [C15] : =UPPER(VLOOKUP(LEFT(B15, 1), \$E\$15:\$F\$18, 2, FALSE))

4. [L25] : =COUNTIF(I15:I24, MODE(SNGL(I15:I24)) & “개”

5. [E27] : =IF(AND(WEEKDAY(A27)=1, D27=0), “적합”, “ ”)

	A	B	C	D	E	F	G	H	I	J	K	L
1	[표1]	상공컨벤션 예약 현황				[표2]	사원평가표					
2	예약코드	구분	예약인원	할인율		이름	근태	실적	시험	총점	평가	
3	1-K-24	이벤트	120	15%		최소희	82	94	86	262	B	
4	3-R-48	세미나	100	30%		조채아	69	88	62	219	D	
5	1-R-72	이벤트	150	15%		고아지	91	95	97	283	A	
6	3-K-18	이벤트	180	15%		손호준	80	70	79	229	C	
7	1-R-39	세미나	130	30%		허은우	77	71	69	217	D	
8	1-K-88	세미나	120	15%		황지성	96	93	95	284	A	
9	3-K-62	기업회의	160	15%		송도원	84	88	91	263	B	
10	3-R-90	기업회의	150	15%		양주희	94	90	90	274	A	
11	1-K-55	세미나	180	15%		김성한	86	84	87	257	C	
12												
13	[표3]	일일 매출 현황				<평가기준표>						
14	분류	제품명	수량	매출액		순위	1	4	6	8		
15	생활용품	종이컵	23	41,400		평가	A	B	C	D		
16	사무용품	포스트잇	37	74,000								
17	사무용품	테이프	51	84,150								
18	사무용품	가위	19	23,750								
19	생활용품	물티슈	34	74,800								
20	생활용품	화장지	22	79,200								
21	사무용품	볼펜	65	71,500		<조건>						
22	생활용품	건전지	24	84,000		분류						
23	사무용품 매출액 평균			63,300		사무용품						
24												
25	[표4]	박스오피스				[표5]	자동차 정보			(단위 : 만원)		
26	영화명	등급	평점	관객수		모델명	연료	연비	가격			
27	극한도시	15세 이상	8.1	2,524,124		스타디아	휘발유	12.4	3,650			
28	슈가보이	전체	6.7	1,835,991		카니윌	경유	13.8	4,210			
29	산의정원	전체	8.8	3,365,725		크나	전기	14.2	5,100			
30	스텝트	15세 이상	9.4	6,189,472		아이오	전기	12.5	4,980			
31	지구탈출	12세 이상	7.1	1,365,543		다나타	휘발유	11.8	3,250			
32	더공조	전체	8.5	5,395,001		타싼	경유	12.5	2,880			
33	분노의신	15세 이상	8.1	2,654,872		아반스	휘발유	14.3	4,670			
34	골드라인	12세 이상	7.5	2,204,589		마티아	경유	15.2	2,260	연료가 휘발유인 자동차 중 연비가 가장 높은 모델명 아반스		
35	두남자	전체	92	1,854,204		투슬라	전기	11.4	4,840			
36	관객수가 3,000,000 이상인 비율			33%		유리오	휘발유	12.7	4,190			
37												

1. [표1]에서 예약코드[A3:A11]의 세 번째 글자가 “R”이면서 구분[B3:B11]이 “세미나”이면 “30%”, 그 외에는 “15%”를 할인율[D3:D11]에 표시하십시오. (8점)

▶ IF, MID, AND 함수 사용

[]

2. [표2]에서 총점[J3:J11]을 기준으로 한 순위와 평가기준표[G14:J15]를 이용하여 평가[K3:K11]를 표시하십시오. (8점)

▶ 순위는 총점이 가장 높은 것이 1위

▶ HLOOKUP, RANK, EQ 함수 사용

[]

3. [표3]에서 분류[A15:A22]가 “사무용품”인 제품의 매출액[D15:D22] 평균을 [D23] 셀에 계산하시오. (8점)
- ▶ 매출액 평균은 십의 자리에서 내림하여 백의 자리까지 표시 [표시 예 : 5,670 → 5,600]
 - ▶ 조건은 [F22:F23] 영역에 입력하시오.
 - ▶ DAVERAGE, DSUM, ROUND, ROUNDUP, ROUNDDOWN 함수 중 알맞은 함수들을 사용
- []
4. [표4]에서 관객수[D27:D35]가 3,000,000 이상인 영화의 비율을 [D36] 셀에 계산하시오. (8점)
- ▶ 비율 = 관객수가 3,000,000 이상인 영화 수 / 전체 영화 수
 - ▶ COUNT, COUNTIF 함수 사용
- []
5. [표5]에서 연료[G27:G36]가 “휘발유”인 자동차 중 연비[H27:H36]가 가장 높은 자동차의 모델명[F27:F36]을 찾아 [J36] 셀에 표시하시오. (8점)
- ▶ INDEX, MATCH, DMAX 함수 사용
- []

수식의 이해

1. [D3] : =IF(AND(MID(A3, 3, 1)="R", B3="세미나"), "30%", "15% ")
2. [K3] : =HLOOKUP(RANK.EQ(J3, \$J\$3:\$J\$11), \$G\$14:\$J\$15, 2)
3. [D23] : =ROUNDDOWN(DAVERAGE(A14:D22, 4, F22:F23), -2)
4. [D36] : =COUNTIF(D27:D35, ">=3000000") / COUNT(D27:D35)
5. [J36] : =INDEX(F27:I36, MATCH(DMAX(F26:I36, 3, G26:G27), H27:H36, 0), 1)

	A	B	C	D	E	F	G	H	I	J	K
1	[표1]	신제품 홍보 현황				[표2]	지점 판매현황				
2	제품코드	홍보예정일	담당자	보도자료		제품명	지점	판매량	판매총액		
3	SN-5761	2024-02-16	조경원			키보드	마포	243	3,037,500		
4	SK-3814	2024-02-27	정은경			키보드	용산	385	4,812,500		
5	SL-0336	2024-03-08	이부성	발송		키보드	서초	196	2,450,000		
6	SK-2398	2024-03-22	정은경	발송		마우스	마포	134	1,273,000		
7	SL-0409	2024-04-15	이부성			마우스	용산	310	2,945,000		
8	SN-4228	2024-04-24	조경원			마우스	서초	251	2,384,500		
9	SK-3911	2024-05-10	정은경	발송		프린터	마포	89	7,609,500		
10	SL-6570	2024-05-30	이부성	발송		프린터	용산	101	8,635,500		
11	SN-2574	2024-06-11	조경원			프린터	서초	67	5,728,500		
12	SK-7013	2024-06-28	정은경			용산점 판매량 비율			45%		
13											
14	[표3]	상공항공 예약 현황				[표4]	사원별 판매실적				
15	예약코드	예약일자	도착지	인원	좌석	사원명	부서	실적			
16	OP-1-93	2024-04-08	로마	4	일반석	김가은	영업1팀	56,986,400			
17	OP-1-57	2024-04-10	하와이	8	일반석	윤주현	영업2팀	35,470,100			
18	OP-3-22	2024-04-11	몽골	4	일등석	한원우	영업1팀	68,341,200			
19	OP-2-61	2024-04-11	두바이	2	비즈니스석	한미진	영업2팀	50,185,000			
20	OP-1-16	2024-04-16	오사카	6	일반석	이은경	영업2팀	62,734,900			
21	OP-2-29	2024-04-18	파리	5	비즈니스석	정상혁	영업1팀	39,900,000			
22	OP-3-81	2024-04-18	뉴욕	3	일등석	노경호	영업1팀	46,984,200			
23	OP-2-09	2024-04-22	사이판	4	비즈니스석	윤정희	영업2팀	55,121,000	<조건>		
24	OP-1-84	2024-04-24	다낭	8	일반석	강소정	영업1팀	43,764,900	부서	부서	
25	OP-1-70	2024-04-24	푸켓	6	일반석	최대실적평균			65,539,000	영업1팀	영업2팀
26											
27	[표5]	회원 관리 현황				<구분코드표>					
28	이름	생년월일	구분	회원코드		구분	코드				
29	송윤진	1991-03-08	B1	1991man1		D1	mas1				
30	주시아	1984-06-21	A1	1984fam1		D2	mas2				
31	강은찬	2001-04-09	C1	2001fri1		C1	fri1				
32	전주린	1995-03-25	D1	1995mas1		C2	fri2				
33	최세현	1987-07-14	C2	1987fri2		B1	man1				
34	이대로	1988-10-03	D2	1988mas2		B2	man2				
35	배상엽	1999-01-30	A2	1999fam2		A1	fam1				
36	오남규	2000-05-07	B2	2000man2		A2	fam2				
37											

1. [표1]에서 홍보예정일[B3:B12]이 3월 또는 5월이면 “발송”을, 그 외에는 공백을 보도자료[D3:D12]에 표시하시오. (8점)

▶ IF, OR, MONTH 함수 사용

[]

2. [표2]에서 지점[G3:G11]이 “용산”인 제품의 판매량[H3:H11] 비율을 [I12] 셀에 계산하시오. (8점)

▶ 판매량 비율 = 용산 판매량 / 전체 판매량

▶ SUM, SUMIF 함수 사용

[]

3. [표3]에서 예약코드[A16:A25]의 네 번째 글자가 “1”이면 “일반석”, “2”이면 “비즈니스석”, “3”이면 “일등석”을 좌석 [E16:E25]에 표시하시오. (8점)
- ▶ CHOOSE, MID 함수 사용
- []
4. [표4]에서 부서[H16:H24]가 “영업1팀”인 사원의 최대 실적[I16:I24]과 “영업2팀”인 사원의 최대 실적의 평균을 [I25] 셀에 계산하시오. (8점)
- ▶ 최대 실적 평균은 백의 자리에서 올림하여 천의 자리까지 표시 [표시 예 : 23,456 → 24,000]
 - ▶ 조건은 [J24:K25] 영역에 입력하시오.
 - ▶ ROUNDUP, AVERAGE, DMAX 함수 사용
- []
5. [표5]에서 생년월일[B29:B36], 구분[C29:C36]과 구분코드표[F29:G36]를 이용하여 회원코드[D29:D36]를 표시하시오. (8점)
- ▶ 회원코드는 생년월일의 년도와 코드를 연결하여 표시
[표기 예 : 생년월일이 ‘2000-01-01’, 코드가 ‘com1’ → 2000com1]
 - ▶ VLOOKUP, YEAR 함수와 & 연산자 사용
- []

수식의 이해

- [D3] : =IF(OR(MONTH(B3)=3, MONTH(B3)=5), "발송", " "))
- [I12] : =SUMIF(G3:G11, "용산", H3:H11) / SUM(H3:H11)
- [E16] : =CHOOSE(MID(A16, 4, 1), "일반석", "비즈니스석", "일등석")
- [I25] : =ROUNDUP(AVERAGE(DMAX(G15:I24, 3, J24:J25), DMAX(G15:I24, 3, K24:K25)), -3)
- [D29] : =YEAR(B29) & VLOOKUP(C29, \$F\$29:\$G\$36, 2, FALSE)

	A	B	C	D	E	F	G	H	I	J	K
1	[표1]	장비 대여 현황				[표2]	선수별 평점				
2	장비코드	대여일자	수량	반납예정일자		클럽	포지션	선수명	평점		
3	HER962	2024-08-05	5	2024-08-12		포항	수비수	윤보경	7.64		
4	KMD238	2024-08-06	3	2024-08-12		대전	공격수	이정환	8.91		
5	GJW183	2024-08-08	6	2024-08-13		포항	공격수	이태용	8.13		
6	PSL725	2024-08-09	4	2024-08-14		포항	수비수	강명보	6.46		
7	BWC651	2024-08-13	3	2024-08-19		대전	수비수	황재준	7.59		
8	MBA401	2024-08-16	4	2024-08-21		포항	공격수	손승민	8.28		
9	KIH390	2024-08-20	6	2024-08-26		대전	수비수	김민우	6.73	대전수비수 평점 평균	
10	SET077	2024-08-22	5	2024-08-27		대전	공격수	김지성	7.32	7.2	
11											
12	[표3]	제품 납품 현황				[표4]	1학기 성적표				
13	납품일자	거래처명	납품량	납품총액		이름	과학기술	문화예술	역사철학	학점	
14	7월23일	하나전자	480	2,640,000		김두완	3.2	3.5	3.1	B	
15	7월23일	우리전자	650	3,575,000		박영재	1.5	2.1	1.8	D	
16	7월23일	상공전자	450	2,475,000		이지환	4.1	4.2	4.1	A	
17	7월30일	하나전자	590	3,245,000		전서혜	2.5	2.7	2.8	C	
18	7월30일	우리전자	550	3,025,000		유가온	3.1	2.9	3.6	B	
19	8월3일	진성전자	620	3,410,000		권도현	2.8	2.5	2.7	C	
20	8월3일	상공전자	500	2,750,000		배재현	0.9	1.4	0.8	D	
21	8월10일	우리전자	680	3,740,000		김영택	3.1	2.7	2.4	C	
22	8월10일	진성전자	470	2,585,000		권애린	3.6	3.9	3.4	B	
23	최대-최소 납품총액 차이			1,265,000		조종연	3.1	3.2	3.5	B	
24											
25	[표5]	수험자 정보				<종목번호표>					
26	이름	수험번호	시험장	날짜-종목		번호	종목				
27	서재영	100215368	1실	1002-워드		1	워드				
28	윤은찬	101433025	2실	1014-컴활2급		2	컴활1급				
29	전태원	101029041	2실	1010-컴활1급		3	컴활2급				
30	송은주	101624257	1실	1016-컴활1급							
31	문진영	100218961	3실	1002-워드							
32	백현준	101435487	3실	1014-컴활2급							
33	고강민	101126698	1실	1011-컴활1급							
34	장세현	101125804	3실	1011-컴활1급							
35	김진아	101531523	2실	1015-컴활2급							
36	우영희	101636857	3실	1016-컴활2급							
37											

1. [표1]에서 대여일자[B3:B10]를 이용하여 반납예정일자[D3:D10]를 계산하시오. (8점)

- ▶ 반납예정일자 = 대여일자 + 5, 단 반납예정일자가 주말(토, 일요일)인 경우 다음 주 월요일로 계산
- ▶ 요일 계산 시 월요일이 1인 유형으로 지정
- ▶ IF, WEEKDAY 함수 사용

[

]

2. [표2]에서 클럽[F3:F10]이 “대전”이면서 포지션[G3:G10]이 “수비수”인 선수들의 평점[I3:I10] 평균을 [J10] 셀에 계산하시오. (8점)
- ▶ 평점 평균은 소수점 이하 둘째 자리에서 반올림하여 첫째 자리까지 표시 [표시 예 : 3.45 → 3.5]
 - ▶ ROUND, ROUNDUP, ROUNDDOWN, COUNTIFS, AVERAGEIFS 함수 중 알맞은 함수들을 사용
- []
3. [표3]에서 납품총액[D14:D22]의 최대값과 최소값의 차이를 [D23] 셀에 계산하시오. (8점)
- ▶ LARGE, SMALL 함수 사용
- []
4. [표4]에서 과학기술[G14:G23], 문화예술[H14:H23], 역사철학[I14:I23]의 평균이 10이면 “D”, 20이면 “C”, 30이면 “B”, 40이면 “A”를 학점[J14:J23]에 표시하시오. (8점)
- ▶ CHOOSE, AVERAGE, INT 함수 사용
- []
5. [표5]에서 수험번호[B27:B36]와 종목번호표[F27:G29]를 이용하여 날짜-종목[D27:D36]을 표시하시오. (8점)
- ▶ 날짜는 수험번호의 앞 4글자로 표시하고, 종목은 수험번호의 5번째 글자를 이용하여 1이면 “워드”, 2이면 “컴활1급”, 3이면 “컴활2급”으로 표시
[표시 예 : 수험번호가 ‘100123456’이면 ‘1001-컴활1급’으로 표시]
 - ▶ VLOOKUP, LEFT, MID 함수와 & 연산자 사용
- []

수식의 이해

- [D3] : =IF(WEEKDAY(B3+5, 2)=6, B3+7, IF(WEEKDAY(B3+5, 2)=7, B3+6, B3+5))
- [J10] : =ROUND(AVERAGEIFS(I3:I10, F3:F10, “대전”, G3:G10, “수비수”), 1)
- [D23] : =LARGE(D14:D22, 1) - SMALL(D14:D22, 1)
- [J14] : =CHOOSE(INT(AVERAGE(G14:I14)), “D”, “C”, “B”, “A”)
- [D27] : =LEFT(B27, 4) & “-” & VLOOKUP(MID(B27, 5, 1)*1, \$F\$27:\$G\$29, 2, FALSE)

	A	B	C	D	E	F	G	H	I	J	K
1	[표1]	사원관리				[표2]	시험 결과				
2	사원번호	사원명	부서명	소속		이름	반	점수		1, 3반 최저 점수 평균	
3	A-1529	김준용	기획부	본사		정다운	2반	92		64.5	
4	B-6011	이경실	기획부	지사		이대로	3반	89			
5	A-2768	허안주	기획부	본사		김민정	1반	59			
6	A-0644	이동준	영업부	본사		최고집	1반	95			
7	B-9537	고강민	영업부	지사		안혜경	3반	79			
8	B-2436	김민경	영업부	지사		조신수	2반	65			
9	A-8167	한상현	관리부	본사		정성민	3반	70		<조건>	
10	A-3551	김단희	관리부	본사		주지훈	2반	88		반	반
11	B-0472	이동엽	관리부	지사		박성은	1반	91		1반	3반
12											
13	[표3]	제품판매현황				[표4]	하프마라톤 결과				
14	구분	제품명	판매가	판매량	총판매액	선수번호	소속	나이	기록		
15	냉장	생쫄면	2,980	24	71,520	168001	춘천	35	1:49:27		
16	냉장	가락우동	3,690	16	59,040	168002	영월	42	1:45:51		
17	냉동	군만두	4,500	32	144,000	168003	강릉	29	2:03:26		
18	냉장	수제비	2,850	24	68,400	168004	평창	38	1:51:15		
19	냉동	핫도그	3,300	22	72,600	168005	영월	44	1:32:08		
20	냉동	피자	4,920	18	88,560	168006	고성	51	1:41:53		
21	냉동	치킨너겟	3,840	19	72,960	168007	춘천	32	2:01:17		
22	냉장	떡볶이	3,600	23	82,800	168008	강릉	40	1:39:22		
23	냉장	쌀국수	4,180	15	62,700	168009	평창	23	1:56:25		
24	판매량 10 이상 20 미만인 판매 건수 비율				44%	가장 빠른 기록		1시간32분8초			
25											
26	[표5]	회원관리현황									
27	회원코드	성별	등급	지역							
28	1-K-320	여	정회원	경기							
29	3-S-966	남	준회원	서울							
30	8-B-415	남	준회원	부산							
31	6-B-527	여	준회원	부산							
32	2-S-605	남	정회원	서울							
33	3-B-012	여	준회원	부산							
34	9-K-048	여	정회원	경기							
35	4-K-213	여	정회원	경기		<지역코드표>					
36	5-S-217	남	정회원	서울		코드	S	K	B		
37	5-B-309	남	준회원	부산		지역	서울	경기	부산		
38											

1. [표1]에서 사원번호[A3:A11]의 첫 번째 문자가 “A”이면 “본사”, “B”이면 “지사”를 소속[D3:D11]에 표시하시오. (8점)

▶ IF, LEFT 함수 사용

[

]

2. [표2]에서 반[G3:G11]이 “1반”인 학생의 최저 점수[H3:H11]와 “3반”인 학생의 최저 점수의 평균을 [J3] 셀에 계산하시오. (8점)

▶ 조건은 [J10:K11] 영역에 입력하시오.

▶ DMIN, AVERAGE 함수 사용

[

]

3. [표3]에서 판매량[D15:D23]이 10 이상 20 미만인 제품들의 판매 건수 비율을 [E24] 셀에 계산하시오. (8점)

- ▶ 판매 건수 비율 = 10 이상 20 미만 판매 건수 / 전체 판매 건수
- ▶ COUNT, COUNTIFS 함수 사용

[

]

4. [표4]에서 기록[J15:J23]이 가장 빠른 선수의 기록을 찾아 [I24] 셀에 표시하시오. (8점)

- ▶ 표시 예 : 1:23:34 → 1시간23분34초
- ▶ HOUR, MINUTE, SECOND, SMALL 함수와 & 연산자 사용

[

]

5. [표5]에서 회원코드[A28:A37]의 세 번째 문자와 지역코드표[G36:I37]를 이용하여 지역[D28:D37]을 표시하시오. (8점)

- ▶ HLOOKUP, MID 함수 사용

[

]

수식의 이해

1. [D3] : =IF(LEFT(A3, 1)="A", "본사", "지사")

2. [J3] : =AVERAGE(DMIN(F2:H11, 3, J10:J11), DMIN(F2:H11, 3, K10:K11))

3. [E24] : =COUNTIFS(D15:D23, ">=10", D15:D23, "<20") / COUNT(D15:D23)

4. [I24] : =HOUR(SMALL(J15:J23, 1)) & "시간" & MINUTE(SMALL(J15:J23, 1)) & "분" & SECOND(SMALL(J15:J23, 1)) & "초"

5. [D28] : =HLOOKUP(MID(A28, 3, 1), \$G\$36:\$I\$37, 2, FALSE)

	A	B	C	D	E	F	G	H	I	J	K	L
1	[표1]	독서실 이용현황				[표2]	부동산 투자현황					
2	고객명	입실시간	퇴실시간	총이용시간		구분	면적	보증금	월임대료	실투자금액		구분
3	김은소	9:10	11:20	2시간		내과	69	150,000,000	2,100,000	230,000,000		소아과
4	조경원	9:34	10:41	1시간		소아과	79	170,000,000	2,500,000	260,000,000		
5	이천수	10:03	11:57	2시간		안과	66	150,000,000	2,100,000	210,000,000		
6	박정연	10:29	11:36	1시간		정형외과	148	330,000,000	4,700,000	480,000,000		
7	김종현	10:31	12:09	2시간		내과	72	160,000,000	2,200,000	230,000,000		
8	권민서	10:45	12:33	2시간		소아과	105	230,000,000	3,300,000	340,000,000		
9	한상민	11:02	12:11	1시간		정형외과	135	300,000,000	4,200,000	440,000,000		
10	최연재	11:14	12:58	2시간		소아과	99	220,000,000	3,100,000	320,000,000		
11	유현진	11:25	12:46	1시간		소아과의 실투자금액 평균				306,666,000		
12												
13	[표3]	컴활2급 시험 결과				[표4]	제품구매현황					
14	수험번호	컴퓨터일반	엑셀			제품코드	주문수량	본사보유량	매장요청량	구매예정수량		
15	1270121	92	76			S-120-D	1,200	1,072	1,400	1,200		
16	1270122	38	55			G-430-F	1,500	1,138	1,300	1,500		
17	1270123	86	92			A-218-Y	1,000	943	900	1,000		
18	1270124	62	48			W-462-N	800	507	900	900		
19	1270125	91	93			C-573-B	1,200	1,138	1,000	1,200		
20	1270126	49	56			H-946-P	900	835	1,000	900		
21	1270127	89	93			F-438-E	1,000	719	1,200	1,200		
22	1270128	67	68			K-149-V	1,300	1,068	1,200	1,300		
23	합격률		63%			M-527-L	1,500	1,329	1,300	1,500		
24												
25	[표5]	재학생정보										
26	학번	입학년도	성명	학과		<학과정보표>						
27	AE1542	2020	김영철	디자인		학과코드	학과교수	학과명				
28	DE2546	2020	신서인	미디어		AE	강종원	디자인				
29	FH1095	2021	양지현	실용음악		DE	한혜진	미디어				
30	AE2428	2021	유새론	디자인		FH	김성민	실용음악				
31	BS2937	2022	이슬아	문예창작		BS	최정용	문예창작				
32	DE3810	2022	송준성	미디어								
33	FH2767	2023	박영훈	실용음악								
34	BS1571	2023	배다혜	문예창작								
35												

1. [표1]에서 입실시간[B3:B11]과 퇴실시간[C3:C11]을 이용하여 총이용시간[D3:D11]을 계산하시오. (8점)

- ▶ 총이용시간은 '퇴실시간 - 입실시간'으로 계산하되, '퇴실시간 - 입실시간'의 '분'이 30분을 초과한 경우 '퇴실시간 - 입실시간'의 '시'에 1을 더하시오. [표시 예 : 총이용시간이 6:00 → 6시간, 6:40 → 7시간]
- ▶ IF, HOUR, MINUTE 함수와 & 연산자 사용

[]

2. [표2]에서 구분[F3:F10]이 "소아과"인 자료의 실투자금액[J3:J10] 평균을 [J11] 셀에 계산하시오. (8점)

- ▶ 실투자금액의 평균은 백의 자리에서 내림하여 천의 자리까지 표시 [표시 예 : 123,456 → 123,000]
- ▶ 조건은 [L2:L3] 영역에 입력하시오.
- ▶ DAVERAGE와 ROUND, ROUNDUP, ROUNDDOWN 함수 중 알맞은 함수를 선택하여 사용

[]

3. [표3]에서 컴퓨터일반[B15:B22]과 엑셀[C15:C22]이 모두 60점 이상인 수험생의 합격률을 [C23] 셀에 계산하십시오. (8점)
- ▶ 합격률 = '컴퓨터일반'과 '엑셀'이 모두 60점 이상인 수험생 수 / 총 수험생 수
 - ▶ COUNTIFS, COUNTA 함수 사용
- []
4. [표4]에서 주문수량[G15:G23]이 본사보유량[H15:H23]보다 200 이상 많은 경우 주문수량과 매장요청량[I15:I23] 중 큰 값을 표시하고, 그 외에는 주문수량을 구매예정수량[J15:J23]에 표시하십시오. (8점)
- ▶ IF, MAX 함수 사용
- []
5. [표5]에서 학번[A27:A34]과 학과정보표[F28:H31]를 이용하여 학과[D27:D34]를 표시하십시오. (8점)
- ▶ 학번의 처음 두 글자가 학과코드임
 - ▶ VLOOKUP, LEFT 함수 사용
- []

수식의 이해

1. [D3] : $=IF(MINUTE(C3-B3)>30, HOUR(C3-B3)+1, HOUR(C3-B3))$ & "시간"
2. [J11] : $=ROUNDDOWN(DAVERAGE(F2:J10, 5, L2:L3), -3)$
3. [C23] : $=COUNTIFS(B15:B22, ">=60", C15:C22, ">=60") / COUNTA(A15:A22)$
4. [J15] : $=IF(G15-H15>200, MAX(G15, I15), G15)$
5. [D27] : $=VLOOKUP(LEFT(A27, 2), \$F\$28:\$H\$31, 3, FALSE)$